
Perl (1) - Dávka teorie na úvod

Úvodní díl seriálu nebude o ničem jiném než o vlastnostech, historii Perlu a přípravě.

8.2.2005 13:00 | Jiří Václavík | přečteno 3892x 

Co je Perl?

Perl je jednoduchý a praktický interpretační programovací jazyk, v praxi používaný zejména na WWW
serverech. Vyniká i v oblasti zpracovávání souborů - ostatně právě kvůli tomu byl Perl navržen.

Co je cílem seriálu?

Odpověď je jednoduchá - naučit se základům Perlu.

Co to programování vůbec je? Program je seznam činností, které byly nařízeny počítači k vykonání.
Programem je například tato posloupnost příkazů: požádej uživatele o 2 čísla, načti je od něj, sečti je a pošli
jejich součet na výstup. Programy nejsou jen pro počítače (stačí nahlédnot do kuchařky. S trochou nadsázky se
dá říct, že každý recept je programem). Program musí být napsán v jazyce, kterému rozumí jak autor, tak ten,
komu ho nařizuje - počítač.

Programování znamená přesvědčit počítač, aby udělal to, co chceme - tedy je to tvoření programu. Přesně
tohle bude naším úkolem.

Něco z historie

Název Perl vznikl z akronymu Practical Extraction and Report Language. Autorem je Kanaďan Larry Wall (*
10. 3. 1949, osobní stránka). Napsal jej pro svou potřebu, protože mu chybělo něco jednoduššího než Céčko,
co zároveň zvládne i větší požadavky. Potom v roce 1987 uvolnil Perl verze 1.0 pro veřejnost a sám se divil
zájmu, který tím vyvolal. Proto se začal ještě dále vyvíjet. Z nástroje pro zpracování textu se povýšil na
programovací jazyk s mnoha doplňky. Dnes se o vývoj Perlu stará skupina lidí v čele s Larrym Wallem.
Poslední stabilní verzí je verze 5.8.6. Verzi 6 lze podle vývojářů čekat v polovině roku 2006 a má se zvýšit
hlavně rychlost a efektivita (Perl 5.x je aktuální od roku 1994).

Stručně o vlastnostech

Původně byl Perl napsán pro Unixovou platformu, ale dnes běží i na mnoha dalších. Mezi nimi se dá relativně
dobře přenášet. Samozřejmě, odlišnosti jsou, avšak často do kódu nemusíte vůbec zasáhnout.

Jste-li líní, jen těžko budete hledat tak jednoduchý jazyk, ve kterém napíšete to, co právě v Perlu. Nic, co píšete,
není zbytečné. Žádné složité deklarování, vše je maximálně jednoduché. Programuje se neobyčejně rychle,
stejně tak se ho díky těmto vlastnostem dá rychle naučit. Například program vypisující text zapíšete v Céčku
takto:

#include <stdio.h>

int main(void){
    printf ("Program v Cecku\n");
}

Ale v Perlu je mnohem kratší:

print "Program v Perlu"\n;

I přesto v něm lze napsat prakticky všechno, co v jazycích jako C.

Na popularitě získal Perl i tím, že se osvědčil jako výborný nástroj pro CGI skriptování. CGI je externí
program, který je spouštěn na žádost WWW serveru. CGI skriptům se budeme věnovat později v samostatných
dílech. Dnes již nemá CGI takovou popularitu, protože ho z jeho pozice vytlačují jazyky typu PHP (PHP má
mimochodem s Perlem velmi mnoho společného).

Perl je pod GNU GPL Actistic licencí. Můžete ho používat zdarma a to i v komerčních projektech. K
programování v Perlu potřebujete jen interpret Perlu a textový editor. Není nutný ani kompilátor. Ale Perl ho
obsahuje. Umožní vám poskytovat váš program ostatním, aniž by viděli zdrojový kód.

V Perlu se dá programovat jak klasicky - strukturovaně, tak objektově.

Na síti CPAN (Comprehensive Perl Aechive Network) dále naleznete jak interpret pro různé platformy, tak i
databázi modulů a dokumentaci. Existují i české mirrory, například ftp.linux.cz/pub/perl

To byly jeho přednosti. Jaké má nevýhody? Tak především: nemusí se deklarovat proměnné, není typová
kontrola, nelze definovat vlastní datové typy, syntaxe je velmi volná - nevěřili byste, kolika způsoby se dá
napsat třeba obyčejná podmínka. Perl není určen pro mnohasetstránková díla, ale pro kratší a efektivní
programy. Říká se, že Perl není příliš vhodným jazykem pro začínající programátory. Umožňuje používat
nesrozumitelné konstrukce a člověk si může navyknout na zlozvyky typu nedeklarace proměnných. Ale zas tak
žhavé to není, většinou stačí umět psát přehledně a používat use strict; - o tom později.

Instalace a příprava

S kvalitními nástroji je práce vždy příjemnější. Předtím, než začneme programovat je tedy nutné si tyto nástroje
představit. Co bude tedy potřeba? Nějaký Linux, interpret jazyka Perl, textový editor a později, až budu psát o
databázích tak Postgres a MySQL a také k programování CGI skriptů WWW server.

Operační systém

Perl spustíte prakticky na všech linuxových distribucích i jiných operačních systémech, takže bohatě postačí
jakýkoliv linuxový systém.

Interpret Perlu

S největší pravděpodobností ho již v systému máte. Přesvědčit se o tom můžete tak, že v konzoli zadáte příkaz:

$ perl -v

This is perl, v5.8.5 built for i586-linux-thread-multi

Copyright 1987-2004, Larry Wall

Perl may be copied only under the terms of either the Artistic License or the
GNU General Public License, which may be found in the Perl 5 source kit.

Complete documentation for Perl, including FAQ lists, should be found on
this system using `man perl' or `perldoc perl'. If you have access to the
Internet, point your browser at http://www.perl.com/, the Perl Home Page.

$

Pokud dostanete podobný výstup, máte interpret instalován a o nic se dál starat nemusíte. Asi jste si všimli, že
1. řádek výstupu obsahuje číslo verze. Je dobrým zvykem mít vždy tu nejnovější stabilní verzi.

Co dělat když Perl instalovaný nemáte? Nejjednodušší je ho nainstalovat z médií, ze kterých jste instalovali
svoji distribuci.

Pokud tento způsob nemůžete z nějakého důvodu praktikovat, je nutné Perl nějak získat - nejlépe z internetu.
Ke stažení je mimo jiné zde na Linuxsoftu. Soubor se obvykle jmenuje stable.tar.gz. Stačí ho jen
rozbalit, přejít do získaného adresáře a můžete začít kompilovat.

$ ./configure
... 
$ make 
...
# make install

V případě úspěchu nyní Perl máte nainstalován, o čemž se ostatně můžete přesvědčit:

$ perl -v

Textový editor

Můžete použít kterýkoliv, jehož výstupem je neformátovaný text. Měl by umět zvýrazňovat syntaxi Perlu a
odsazovat.

Nedáte-li dopustit na Vim, klidně ho používejte. Dokáže velmi dobře rozpoznávat syntaktické prvky. K
aktivaci zvýrazňování je třeba příkaz :syntax on. Protože je Vim inteligentní, nemusíte se starat o volbu
jazyka. Vim ho rozpozná podle přípony. Totéž umí i Emacs.

Na programování všeho druhu je výborným nástrojem Quanta. Najdete ho na 
http://sourceforge.net/project/showfiles.php?group_id=4113. Naše požadavky také bez problémů splňuje.

Volba je na vás. Záleží na tom, co vám vyhovuje nejvíc.

Databáze

Nainstalujte si jak Postgres tak i Mysql. V seriálu se naučíme používat obě. Jsou-li ve vaší distribuci, máte
vystaráno, jinak je najdete na zmíněných odkazech.

WWW server

Apache je bezkonkurenční. Tady je zbytečné cokoliv vysvětlovat, snad jen odkaz.

Překladač

Je užitečný v případech, kdy chcete poskytnout program, ale zdrojový kód si chcete ponechat pro sebe. To se
hodí při psaní komerčních aplikací. Může nahradit i interpret. Ke své činnosti ho nepotřebuje. Zdrojový kód
však musí obsahovat všechny potřebné knihovny, a je tak o mnoho větší než obyčejný kód. Použitím
překladače vznikne binární soubor.

$ perlcc program.pl

Tak získáme spustitelný soubor ./a.out. Volbou -o můžeme určit získanému souboru jméno.

$ perlcc -o program program.pl

Tím byl vytvořen soubor ./program.

O čem bude seriál?

Napsal jsem předběžnou osnovu témat, které se budu do určité míry držet:

Proměnné, konstanty a operace a funkce pro práci s nimi
Pole, seznamy, hashe, kontexty a funkce pro práci s nimi
Podmínky a cykly
Podprogramy, moduly
Vstup a výstup, formátování řetězců
Práce se soubory a databázemi
Odkazy

Spolupráce s příkazovým řádkem, operačním systémem
Regulární výrazy
CGI
Objekty
A určitě i něco dalšího

Těšte se také na příklady. Už teď prozradím, že vás mimo jiné čeká pár programů na šifrování a dešifrování.

Teď jsme konečně hotovi s přípravou a příště už můžeme začít psát 1. jednoduché programy.

Online verze článku: http://www.linuxsoft.cz/article.php?id_article=675

Perl (2) - Úvod do syntaxe

Dnes se budeme zabývat nejjednodušším programům a ukážeme si i pár dalších tipů.

15.2.2005 15:00 | Jiří Václavík | přečteno 2725x 

Příklad na úvod

Do 1. řádku každého perlovského programu se píše řetězec #!/usr/bin/perl. To proto, aby systém věděl,
čím má program spouštět. Po znacích #! následuje právě cesta k interpretu, který má být použit ke spuštění
vámi napsaného programu. Máte-li interpret jinde, tak jen změňte cestu (dá se zjistit příkazem whereis nebo 
env, případně i jinak - záleží na distribuci). Takto zapsaný úvodní řádek nepoužívá jen Perl, ale interpretované
jazyky obecně. Píšete-li např. v Pythonu, stačí 1. řádek zaměnit za #!/usr/bin/python. Je třeba
podotknout, že takto napsaný 1. řádek program obsahovat nemusí. K čemu je pak ale dobrý? To si hned
ukážeme. Jsou 2 důvody.

Normálně byste spouštěli váš program takto:

  $ perl program.pl

Ale použijete-li onen magický řádek, stačí, když napíšete jen:

  $ ./program.pl

Aby to nebylo tak jednoduché, ještě předtím musíte označit soubor jako spustitelný:

  $ chmod 700 program.pl

Hlavní rozdíl je ale v tom, že když program s #!/usr/bin/perl nahrajete do adresáře /usr/bin a potom
z kteréhokoliv místa zadáte následující příkaz, spustí se váš program.

  $ program.pl

Nyní jsme připraveni a můžeme si tak ukázat 1. příklad. Vytvořte nový soubor s názvem program.pl ve
vašem textovém editoru a do něj uložte tento kód:

  #!/usr/bin/perl
  
  print "Náš 1. program v Perlu!\n";

Už ze zápisu je vám jistě zřejmé, co bude program dělat. Jak jste se v minulém díle dozvěděli, je program
posloupnost nějakých nařízení počítači. Přesně tak to platí i zde. Řádek print "Náš 1. program v
Perlu!\n"; přeložený do naší řeči, znamená Vytiskni na výstup text Náš 1. program v
Perlu! a odentruj. Print je příkaz, kterým dáváme počítači na vědomí, že chceme vypsat text, který
napíšeme za print do uvozovek. Středník dává interpretu na vědomí, že zde příkaz končí. Musí být za

každým příkazem. Až po středníku můžeme psát další příkazy. Výstup našeho 1. programu bude vypadat
takto:

  $ ./program.pl
  Náš 1. program v Perlu!
  $

Nic jiného, než výpis textu. Možná se ptáte, co znamená ta dvojice znaků \n. Vězte, že to je označení ukončení
řádku. Kdybyste tuto sekvenci znaků vynechali, dostali byste opticky nepříliš hezký výstup:

  $ ./program.pl
  Náš 1. program v Perlu!$

Už jsem nakousl téma takzvaných escape sekvencí, tak si k nim rovnou něco řekneme. Escape sekvence (v
tabulce je jejich částečný výčet) jednoduše řečeno nahrazují znaky, které nemůžete v Perlu za daných okolností
napsat.

\n nový řádek

\t tabulátor

\a pípnutí systémového zvonku

\ua Znak a - ať je jakýkoliv - bude psán velkým písmenem

\la Znak a - ať je jakýkoliv - bude psán malým písmenem

\U...\E Vše mezi escape sekvencemi \U a \E bude psané velkými písmeny

\L...\E Vše mezi escape sekvencemi \L a \E bude psané malými písmeny

\\ Zpětné lomítko

\" Uvozovka (používá se v řetězcích označených uvozovkami)

\' Apostrof (používá se v řetězcích označených apostrofy)

Například zmiňovaný nový řádek. Pokud ve zdrojovém kódu místo escape sekvenece \n jen odentrujete,
neuspějete. Perl totiž každou posloupnost tzv. bílých znaků nahrazuje právě jednou mezerou. Takže byste
klidně mohli jakkoliv rozsáhlý program napsat na 1 řádek. Stačilo by konce řádků nahradit mezerami. Bílým
znakem je mezera, tabulátor a nový řádek.

Zkuste si zaexperimentovat a v našem příkladu si do uvozovek za print přidejte nějakou další escape sekvenci -
například \a a sledujte (v tomto případě poslouchejte - tedy pokud nemáte nastaven vizuální zvonek) co se
stane.

Přípona perlovského programu

Jistě jste si všimli, že programům dávám vždy příponu .pl. Není povinná (systém typ programu stejně vždy
určuje podle řádku #!/cesta/k/interpretu), ale pro přehlednost je dobré ji používat.

Komentáře

Jsou to úseky programu, které interpret ignoruje. Na výsledný program nemají žádný efekt. K čemu jsou tedy
dobré? Zvyšují přehlednost zdrojového kódu a jsou velmi užitečné v případě, kdy chcete kód po delší době
editovat. Obzvlášť jedná-li se o složitější program. Zapisují se křížkem - od něj do konce řádku interpret vše
ignoruje. Nejlepší bude, když uvedu příklad:

  #!/usr/bin/perl

  #########################################
  # Autor:   Jiří Václavík                #
  # Verze:   20050204                     #
  # Název:   program.pl                   #
  # Činnost: Vypisuje text                #

  #########################################

  print "Náš už 2. program v Perlu. S komentáři!\n";#Další komentář

  #A ještě jeden

Jak používat komentáře? Měli by usnadňovat pochopení programu při čtení jeho zdrojového kódu (nesprávně
napsaný komentář je na obtíž) a 2. významem je informace o názvu, verzi, autorovi a činnosti popř. něčeho
dalšího na začátku rozsáhlejšího programu. Komentáře píšete pro to, aby usnadňovali pochopení. Je třeba psát
proč je ta a ta část programu napsaná takto a ne jinak, jaký význam zde má tato proměnná, co tu bude potřeba
dodělat nebo něco, co zkrátka bude při pozdějším čtení kódu k užitku. Komentář typu #vypisuje text 
Ahoj za příkazem print "Ahoj"; nám asi moc nepomůže. Ovšem setkávám se s ním poměrně často,
zejména v případech, kdy autor zkrátka komentář napsat chce. Budete-li psát do svých programů komentáře a
po nějaké době je číst, časem se to naučíte.

Potřebujete-li psát rozsáhlejší, víceřádkové komentáře, lze užít speciální syntaxi. Před řádek, kde komentář
začíná, napíšete řetězec =komentar a jeho konec označíte =cut.

  #!/usr/bin/perl

  print "Text před komentářem\n";

  =komentar
  Víceřádkový
  komentář
  =cut

  print "Text za komentářem\n";

To ale jen tak pro úplnost. Používají se totiž spíše ke speciálním účelům (dokumentace). Někdy o ní napíšu
celý díl.

Shrnutí

Soubor perlovského programu značte příponou .pl
Na 1. řádek každého programu v Perlu pište #!/usr/bin/perl
Příkaz print vypisuje na výstup určený text
Za každým příkazem musí být středník
Chcete-li vypsat na výstup konec řádku, zapište do příkazu print escape sekvenci \n
Komentáře usnadňují čitelnost zdrojového kódu

V příštím díle se dozvíte něco o základním kamenu všech programovacích jazyků - o proměnných.

Online verze článku: http://www.linuxsoft.cz/article.php?id_article=676

Perl (3) - Proměnné

Věnovat se budu proměnným a základní operaci s nimi - přiřazovaní.

22.2.2005 15:00 | Jiří Václavík | přečteno 2329x 

Proměnná

Proměnnou si můžeme představit jako oblast, v níž jsou uloženy data. Proměnnou můžeme použít kdekoliv v
programu, aniž bychom se o ní zmiňovali dříve (definovali ji).

Nejlepší bude ukázat příklad:

  #!/usr/bin/perl

  $promenna = "obsah proměnné\n";
  print $promenna;

Uložte si tento kód do souboru, nastavte práva pro spouštění a spusťte ho. Příkaz $promenna = "obsah 
promenne\n"; uloží do proměnné s názvem $promenna řetězec v uvozovkách. Tuto operaci můžeme
nazvat přiřazení (je to vlastně uložení nějaké hodnoty do proměnné). Co se děje dál už víme z minulého dílu.
Příkaz print tiskne na výstup. V tomto případě vytiskne obsah proměnné $promenna.

Název proměnné

Nejdříve si bude dobré říct něco o syntaxi. Každá proměnná se, jak jste si jistě všimli na příkladu, píše s
dolarem na začátku. Název se může skládat s písmen anglické abecedy, číslic a podtržítka. Přitom platí, že 1.
znak jména proměnné nesmí být číslice. Délka proměnných je libovolná od 1 znaku až po 255. Délce 255 se v
praxi ani zdaleka nepřiblížíte (můžete to vyzkoušet, interpret zahlásí Identifier too long).

Už když proměnnou pojmenujete názvem o 20 znacích, snižuje to čitelnost programu. Použití dlouhých
proměnných, jako například $pocetobyvatelceskerepubliky je třeba zvážit. K čitelnosti částečně
pomůže přidání podtržítek: $pocet_obyvatel_ceske_republiky, ale stále můžete udělat nějaký
překlep, navíc je psaní programů výrazně pomalejší a to se k Perlu nehodí.

Perl rozlišuje velikost písmen identifikátorů. Jen pro pořádek proměnná $promenna není totožná s
proměnnou $PROMENNA. Existuje ale nepsané pravidlo, že jména proměnných se zapisují malými písmeny. Je
to z důvodu přehlednosti. Jiné druhy identifikátorů (zatím jsme si o nich neříkali) se píší zas jiným způsobem -
viz tabulka.

Typ identifikátoru Způsob zápisu Příklad

Proměnné, podprogramy malými písmeny $promenna, &podprogram()

Moduly 1. velké, ostatní malá Modul

Návěští, konstanty, zdroje dat velkými písmeny NAVESTI:, KONSTANTA, <STDIN>

V 1 programu se nemůže vyskytovat více než 1 proměnná stejného názvu (není to úplně přesně řečeno, ale
prozatím to bude stačit).

Název proměnné volte podle toho, co hodnota v ní vyjadřuje. Jistě mi dáte za pravdu, že tento zápis:

  $promenna1 = $promenna2 * $promenna3 * $promenna4;

je oproti následujícímu takřka nicneříkající:

  $obsah_kvadru = $a * $b * $c;

Práce s proměnnými

Do proměnné můžeme přiřazovat řetězce, čísla (ať už celá nebo desetinná) nebo odkazy (na to máme čas).
Během provádění programu se může proměnná mezi číslem a řetězcem automaticky konvertovat (myslím tím,
že pokud na začátku programu použijete proměnnou jako číslo, může být v jeho průběhu měněna na řetězec a
obráceně). V dalších dílech napíšu pravidla, kterými se konverze řídí. Délka řetězce je omezená jen velikostí
paměti.

Co je to přiřazení jsem už vysvětloval. Uvedu ještě pár příkladů:

  
  $promenna = "normální řetězec";

  $promenna = "řetězec\nna 3\nřádcích";
  #řetězec je uprostřed 2× zlomen. Do proměnné se uloží i escape znaky.

  $rok = 2005;



LINUX SOFTWARE file:///C:/Ikon/Books/Perl/LinuxSoft_PERL.html

8 of 102 3/8/2006 20:54

  #$promenna obsahuje číslo

  $e = 2.71828;
  #desetinné číslo

  $x&y = 2005;
  $44 = 2006;
  #nespravné názvy proměnných. Interpret hlásí chybu.

Operátor = není jediným přiřazovacím příkazem. Další poznáte v příštích dílech.

Všechny hodnoty v uvozovkách jsou Perlem brány jako řetězec. Mimo uvozovek lze použít i apostrofy (existují
ještě jiné zápisy, ale těmi vás teď nebudu hltit). Liší se tím, že použijete-li apostrofy, ignorují se escape znaky a
proměnné v řetězci.

  $promenna = 2005;
  print 'retezec s escape sekvencemi \n\a a promennymi $promenna';

Uložte a spusťte program. Dostanete kupodivu toto:

  $./program.pl
  retezec s escape sekvencemi \n\a a promennymi $promenna\n
  $ 

Stejný příklad, jen zaměníme apostrofy za uvozovky:

  $promenna = 2005;
  print "retezec s escape sekvencemi \n\a a promennymi $promenna";

Po spuštění:

  $./program.pl
  retezec s escape sekvencemi
  a promennymi 2005
  $ 

(Pokud nejde spustit, pravděpodobně vám chybí na začátku souboru #!/usr/bin/perl nebo nemáte
správně nastavená práva.)

Kopie proměnných

Zatím jsme přiřazovali jen čísla a řetězce. Ale je možné přiřazovat i existující proměnné (ve skutečnosti ale
přiřazujete jen obsah přiřazované proměnné - takže jde také o řetězec nebo číslo).

  $pi = 3.142;
  $pi_kopie = $pi;
  print $pi_kopie;

V takovém případě se vytvoří kopie původní proměnné.

Na jednom řádku můžete přiřadit stejnou hodnotu i více proměnným.

  $a = $b = 2005;

Nyní jak hodnota $a tak i $b obsahují hodnotu 2005.

Přepsání hodnoty proměnné

Stačí, když provedete přiřazení do stejné proměnné 2krát po sobě. Při opakovaném přiřazení se původní obsah
ztratí a nahradí se novým.

  $p = 1;
  print $p; #1
  
  $p = 2;

  print $p; #2

Příště čekejte něco o operacích s čísly.

Online verze článku: http://www.linuxsoft.cz/article.php?id_article=678

Perl (4) - Čísla a řetězce

Bude řeč o řetězcích, číslech a operacích. Čekají vás mimo jiné i další operátory přiřazení, automatická
konverze mezi řetězci a čísly, příkaz use strict; a nebo priorita příkazů.

10.3.2005 15:00 | Jiří Václavík | přečteno 2233x 

Rozlišujeme datový typ skalární, pole skalárů a hash skalárů. Poslední 2 typy jsou seznamem skalárních
hodnot. Skalární hodnoty mohou být řetězcem, číslem nebo odkazem. Já se budu zabývat zatím jen řetězci a
čísly. Perl dokáže vyjádřit číslo jako řetězec a naopak, takže toto rozdělení nehraje tak velkou roli jako u
striktnějších jazyků. Více o datových typech ale až v díle o polích.

Čísla

Perl nerozlišuje mezi celými a racionálními čísly. Bere všechny čísla jako racionální - tedy jako 1 datový typ.
Dělíte-li vzájemně 2 celá čísla, která jsou nedělitelná, řešením je přesto - matematicky správně (v tomto případě
ne zas tak úplně, jak vidíte) - racionální číslo (10 / 3 = 3.33333333333333). Kdybyste to samé dělali v Céčku,
výsledkem po dělení celých čísel musí být také celé číslo (10 / 3 = 3). Ve skutečnosti se dá Perl také přepnout
na celočíselný režim pomocí příkazu use integer; a do racionálního režimu se vrátíte příkazem no 
integer. Potom Perl vrací vždy celé číslo.

  #!/usr/bin/perl
  
  print 10 / 3;#3.33333333333333
  
  use integer;#odteď celočíselný režim
  print 10 / 3;#3
  
  no integer;#a zase racionální
  print 10 / 3;#3.33333333333333

Charakteristickým znakem Perlu je možnost použít na cokoliv rozmanité způsoby zápisu. Zápis čísel není
výjimkou.

Celé číslo zapíšete jednoduše jako posloupnost číslic. Zapisujete-li desetinné číslo, desetinná část se odděluje
tečkou nikoliv čárkou. (Konvence zápisu se stát od státu liší, Perl vychází ze syntaxe v USA). Možný je zápis v

semilogaritmickém tvaru (9.4 * 103 = 9.4e3).

V zápisu čísel se nemusíte omezovat jen na desítkovou soustavu:

Soustava Zápis Použitelné znaky

desítková klasicky 0-9

osmičková před číslem je 0 0-7

šestnáctková před číslem je 0x 0-9, a-f, A-F

dvojková Před číslem je 0b 0, 1

Jestliže použijete jiné znaky než ty z uvedených intervalů, budete svědky chybové hlášky.

V zápisu lze používat takzvané unární operátory + a -. Je to znaménko + nebo - před číslem. Příkladem může
být -18 nebo +95. Unární proto, že má 1 operand. Mimo unární operátory existují v Perlu ještě binární a
ternární.

Delší čísla mohou obsahovat pro přehlednost podtržítka. Při vyhodnocení jsou odstraněny.

Toto všechno jsou zápisy stejného čísla:

2345
2345.0
2345.
2.345e3
2.345E3
2.34_5e3
2_345
23____4_5
+2345
04451
0x929
0b100100101001
0b100_100_101_001

Matematické operace

Proměnné můžete sčítat, odčítat, násobit, dělit nebo jinak s nimi operovat. Ještě před příkladem tabulka
matematických operátorů:

Operátor Operace Příklad Výsledek

+ sčítání 9 + 15 21

- odčítání 6 - 4 2

* násobení 3 * 5 15

/ dělení 9 / 4 2.25

** umocňování 3 ** 4 81

% zbytek po celočíselném dělení 9 % 4 1

U umocňování je levý operand základem a pravý mocnitelem (exponentem). Takže zápis 3 ** 4 v Perlu je

ekvivalentní matematickému zápisu 34.

Zbytek po celočíselném dělení (modulus) se dá vysvětlit takto: levý operand vydělíte pravým, vzniklé číslo
zaokrouhlíte celočíselně dolů, vynásobíte levým operandem a výsledek odečtete od levého operandu. Získaná
hodnota je modulus.

(operandem je myšlena hodnota vcházející do operace.)

Příklady:

  $a = 1;
  $b = 2;
  $c = 4;
  
  $soucet = $a + $b;    # výsledkem je 1 + 2 = 3
  $rozdil = $c / $a;    # 4 / 1 = 4
  $v1 = $b + $b ** $c;  # 2 + 2 ^ 4 = 18
  $v2 = ($b + $b) ** $c;# (2 + 2) ^ 4 = 256
  
  $v3 = 5 + 9 * 10;     # 95
  $v4 = (5 + 9) * 10;   # 140

Stejně jako v matematice lze závorkovat tam, kde potřebujete změnit prioritu operací (dále v článku).

Další operátory přiřazování

Přiřazování jsme věnovali značnou část minulého dílu. vysvětlili jsme si, jak funguje operátor =. Ten přiřadí do
proměnné určenou hodnotu. A dnes budeme pokračovat.

Tabulka přiřazovacích operátorů (poslední 2 operátory jsou pro řetězce):

Operátor Příklad Přepis Výsledek $p pro $p = 10

= $p = 3 $p = 3 3

+= $p += 3 $p = $p + 3 13

-= $p -= 3 $p = $p - 3 7

*= $p *= 3 $p = $p * 3 30

/= $p /= 3 $p = $p / 3 3.33333333333333

**= $p **= 3 $p = $p ** 3 1000

%= $p %= 3 $p = $p % 3 1

++ $p++ $p = $p + 1 11

-- $p-- $p = $p - 1 9

.= $p .= "def" $p = $p . "def" "abcdef"

x= $p x= 3 $p = $p x 3 "abcabcabc"

Příklad má v tabulce stejný význam jako přepis.

Pro názornost příklad: V proměnné $p je hodnota 10. Zdvojnásob ji, přičti 12, umocni dvěma a vyděl 4.
Vzniklou hodnotu děl 15 a zbytek po celočíselněm dělení tiskni na výstup. Napíšeme si jednoduchý program,
který to spočítá.

  #!/usr/bin/perl
  
  $p = 10; # V proměnné $p je hodnota 10
  $p *= 2; # Zdvojnásob ji,
  $p += 12;# přičti 12,
  $p **= 2;# umocni na 2.,
  $p /= 4; # a vyděl 4.
  $p %= 15;# Vzniklou hodnotu děl 15 a zbytek po celočíselném dělení
  
  print "Řešením úlohy je číslo $p.\n";# tiskni na výstup

Inkrementace, dekrementace

inkrementace - zvýšení hodnoty v proměnné o 1
dekrementace - snížení hodnoty v proměnné o 1

Inkrementace:

  $p = 10;
  $p ++;# Hodnota $p se zvýší o 1
  print $p# Vytiskne hodnotu 11 - tedy hodnotu o 1 větší než byla původní hodnota $p

Dekrementace je to samé, jen se zapisuje operátorem --.

  $p = 10;
  $p --;
  print $p#9

Tyto zápisy jsou tedy ekvivalentní:

  $p = $p + 1;
  $p += 1;
  $p ++;

Nevýhoda dlouhého zápisu tkví v tom, že je méně přehledný.

Postfixová, prefixová (a infixová) notace

Operátor ++ (resp. --) lze napsat i před proměnnou. To má význam jen v případě, kdy nepoužijete zápis
osamoceně, ale například při přiřazení. Použijete-li zápis před proměnnou (prefixový zápis), bude hodnota
nejdříve inkrementována, až poté přiřazena. Při zápisu za proměnnou (postfixový zápis), bude nejprve
přiřazena a až poté inkrementována.

Postfix:

  $a = 3;
  $b = $a++;#do $b se přiřadí hodnota proměnné $a a až potom se zvýší $a o 1. $b tedy bude
mít hodnotu 3
  
  print $a; #4
  print $b; #3

Prefix:

  $a = 3;
  $b = ++$a;#tady je to naopak. Nejdříve se $a zvýší o 1 a až potom se hodnota $a přiřadí do
$b
  
  print $a; #4
  print $b; #4

Infixová notace znamená, že je operátor mezi operandy. Například 5 + 9. U unárních operátorů nemá
význam.

Řetězce

Ve 3. díle jsme si vysvětlili rozdíl mezi označením řetězce do apostrofů a uvozovek. Ve skutečnosti jsem vám
něco zamlčel. Podle toho, co jsem napsal, by do řetězce ohraničeného apostrofy vložit právě apostrof vůbec
nešel. Existují 2 výjimky, které řetězec uvozený apostrofy zobrazí jinak, než jak je napsaný. Těmi výjimkami
jsou \' a \\, umožňující do řetězce vložit apostrof a zpětné lomítko. Obecně lze říct, že má-li být do textu vložen
znak, uvozující řetězec, musíte mu předřadit zpětné lomítko.

O escape znacích jsme již také mluvili. Ještě si zkusíme příklad.

  print "\upoužitím escape sekvencí můžeme převést \Ucelý úsek textu na velká písmena\E.\n";

Výstup je zřejmý:

  $./program.pl
  Použitím escape sekvencí můžeme převést CELÝ ÚSEK TEXTU NA VELKÁ PÍSMENA.
  $ 

Základními operacemi s textem je zřetězení a opakování.

  $a = "larry" . "wall";
  print $a; # vytiskne larrywall
  
  $jmeno = "larry";
  $prijmeni = "wall";
  
  $cele_jmeno = $jmeno . " " . $prijmeni;
  
  print $cele_jmeno; # vytiskne larry wall
  

Z příkladu jste asi pochopili, že se po použití operátoru tečky spojí operand na levé straně s pravým.

Operátor opakování se se bude hodit žákům. Chcete-li 100x vypsat nějaký řetězec, použijte zápis:

  print "Budu nosit domácí úkoly.\n" x 100;

Zkuste spustit:

  $./program.pl
  Budu nosit domácí úkoly.
  Budu nosit domácí úkoly.
  Budu nosit domácí úkoly.
  Budu nosit domácí úkoly.
  Budu nosit domácí úkoly.
  ..... (ještě 94×)
  Budu nosit domácí úkoly.
  $ 

Budou-li to po vás chtít ručně, pak neporadím:(

Konverze datových typů

Co se stane, když provádíme numerické výpočty s řetězci a naopak - řetězcové operace s čísly? Perl
automaticky převede řetězec na číslo nebo číslo na řetězec. Jestliže na začátku řetězce není číslo, převede Perl
řetězec na 0. V opačném případě je řetězec převeden na číslo, kterým řetězec začíná a ostatní znaky jsou
vypuštěny. Přičemž nebere ohled na bílé znaky na začátku.

  print 10 + "265pps";
  # vytiskne 275

  print 10 + "pps";
  # vytiskne 10

  print "2k5" + "123px";
  # vytiskne 125

  print "2e5" + "123px";

  # tady je zrada! Vytiskne 200123, protože "2e5" je konvertováno na 2e5 = 2 * 105, nikoliv 
na 2

  print " 2k5" + "123px";
  # bílý znak na začátku (mezera) jako by nebyl, Perl ho ignoruje a řetězec konvertuje na 2.
Výsledkem tedy je 125

Konverze z čísla na řetězec je jasný. Získaný řetězec bude prostě obsahovat číslo.

  print 12 . 6;
  #vytiskne 126
  
  print "retezec" . 6;
  #vytiskne retezec6

Priorita operací

K tomu, která operace se kdy vykoná, sloužítabulka priorit. Neuvádím ji zde úplnou. Ty operace, které již
známe nebo poznáme brzy (horizont 2 dílů), jsou zvýrazněny zeleně:

Operátor Operace Asociativita Arita

++, -- Inkrementace, dekrementace není 1

** Umocňování zprava 2

+, -, !, \ Plus, mínus, not (logické), odkaz zprava 1

=~, !~ Operace pro práci s regulárními výrazy zleva 2

*, /, %, x Násobení, dělení, modulus, opakování zleva 2

+, -, . Plus, mínus, spojení řetězců zleva 2

<, <=, >=, >, lt, le, ge, gt Porovnávání není 2

==, !=, <=>, eq, ne, cmp Porovnávání není 2

&& and (logické) zleva 2

|| or (logické) zleva 2

.. Operátor rozsahu není 2

?: Operátor podmínky zprava 3

=, +=, -=, *=, /=, .=, x=, **=, %= Přiřazovací operátory zprava 2

=>, , Čárka zleva 2

not not (logické) zprava 1

and and (logické) zleva 2

or or (logické) zleva 2

Čím výše je operátor, tím má vyšší prioritu - a tedy vykoná se přednostně oproti operacím s operátorem pod
ním.

Směr vykonávání (asociativita) určuje, jak se bude vykonávat operace se stejnou prioritu. Jsou 3 možnosti:

výraz se bude vyhodnocovat zleva
výraz se bude vyhodnocovat zprava
priorita není pro některé operace důležitá nebo není možná

Pochopíte to na příkladu: 8 / 4. Výsledkem je 2, nikoliv 0.5 (8 dělím čtyřmi, ne opačně - výraz se vykonává
zleva).

Vlastnosti operátorů jsou tedy obecně priorita, asociativita, počet operandů (arita), případně i typ a kontext
operandů.

Chcete-li změnit prioritu, používejte kulaté závorky. Lze je používat i tam, kde nejsou nutné (třeba když si
nejste prioritou jistí a nechce se vám hledat).

Špetka disciplíny

Dosud jsme proměnné nikdy nedeklarovali. V Perlu je to možné. A je dobré to dělat. Slouží k tomu klíčové
slovo my. Jeho parametrem je jméno proměnné, která bude platná v daném BLOKU programu (blok vysvětlím
v příštím díle o podmínkách). Příkaz use strict;, uvedený někde na začátku programu vás hlídá abyste
(mimo jiné) všechny proměnné deklarovali.

Další užitečná věc je přepínač -w. Pokud spustíte Perlu příkazem perl -w, budou se vypisovat varování,
která mohou napomoci odstranit případné chyby. Přepínač lze připsat do 1. řádku programu.

  #!/usr/bin/perl -w
  use strict;
  
  my $p;
  
  $p = 3;
  print $p;

Perl vám nedovolí použít proměnnou, aniž byste ji definovali. Odstraníme-li řádek my $p; program vypíše
chybové hlášení a skončí.

Režim strict se ruší příkazem no strict;. Pokračujme v kódu.

  no strict;
  
  $q = 3;
  print $q;

Proměnnou $q jsem teď definovat nemusel.

Více o use strict;, definici proměnných a případně jiných direktivách v díle o modulech.

Zdroje

man perlop

V dalším dílu se budu zabývat řízením cyklu podmínkami.

Online verze článku: http://www.linuxsoft.cz/article.php?id_article=737

Perl (5) - Podmínka

Dosud jsme nenapsali žádný pořádný program. Dnes si ukážeme jeden z prostředků, který je k tomu nezbytný.

24.3.2005 15:00 | Jiří Václavík | přečteno 1864x 

Podmínka rozhoduje, kudy se má skript vykonávat. Díky podmínkám nemusejí příkazy probíhat od prvního k
poslednímu, ale lze části vynechávat. Podmínka určuje, zda má být určitý příkaz nebo blok kódu (dále v
článku) vykonán.

Uvedeme si příklad takové podmínky v našem jazyce:

  #!/usr/bin/nase_rec
  
  ........ (nějaký kód) ........
  
  jestliže (bylo hlasováno){
      Tiskni "Již jste hlasoval!";
  }
  
  jestliže (nebylo hlasováno){
      Tiskni "Ok, můžete hlasovat:";
      ........ (formulář na hlasování) ........
  }
  ........ (nějaký další kód) ........

Pochopení je intuitivní. Je nutné zabezpečit, aby nehlasoval 1 člověk vícekrát. Proto, si program nejdříve musí
zkonrolovat, jestli už stejný člověk nehlasoval. Pokud zjistí že ano, vypíše program Již jste hlasoval!. V
opačném případě zobrazí hlasovací lístek.

Jak funguje podmínka je nyní jasné. Stačí se jen naučit, jak se přeloží do Perlu.

Blok příkazů

Ještě předtím ale vysvětlím pojem blok příkazů. Označuje se složenými závorkami.

  {
      příkaz1;
      příkaz2;
      příkaz3;
      ...
      příkazn;
  }

Za posledním příkazem bloku je středník nepovinný, ale je dobré ho psát pro případ, že byste chtěli někdy
editovat. Za složenou koncovou závorku se středník nepíše.

Blok je skupina příkazů, které budou podmínkou (později poznáme jiné strukturované příkazy, pro které se
blok také používá) vykonány. Bezprostředně za podmínkou (testem) jestliže (bylo hlasováno) v
příkladu můžete právě takový blok nalézt. Vše, co je mezi { a } bude vykonáno, byl-li test úspěšný. V
opačném případě bude blok přeskočen a skript pokračuje za ním.

Pravda (true) a nepravda (false)

V našem příkladu jsme za klíčové slovo jestliže napsali do jednoduchých závorek test. Nyní se zaměříme
na to, jak je vyhodnocován.

Je-li výsledek testu pravdivý (true), bude následující blok příkazů vykonán. Bude-li výsledkem false, bude
přeskočen.

Podmínka je false v případě, že je test (to není nic jiného než výraz) vyhodnocen jako:

prázdný řetězec ("")
nula (ať 0 nebo "0")
nedefinovaná hodnota (když je testem proměnná, do které jste buď nic nepřiřazovali nebo přiřazovali
nedefinovanou hodnotu)

Ve všech ostatních případech je test vyhodnocen jako true.

Podmínka if

Už víme, kdy je test vyhodnocován jako true a kdy jako false. Teď pár příkladů.

  if (0){
      print "1. blok nebude vykonán: 0 je vyhodnocena jako false\n";
  }
  
  if (""){
      print "2. blok nebude vykonán: prázdný řetězec je vyhodnocen jako false\n";
  }
  
  if (1){
      print "3. blok bude vykonán: 1 je vyhodnocena jako true\n";
  }
  
  if ("nejaky text"){
      print "4. blok bude vykonán: neprázdný (nenulový) řetězec je vyhodnocen jako true\n";
  }
  
  if (1 + 2){
      print "5. blok bude vykonán: 1 + 2 = 3 je vyhodnoceno jako true\n";
  }
  
  if ($promenna_do_ktere_jsme_nic_nepriradili){
      print "6. blok nebude vykonán: proměnná, o které jsme se nikde nezmínili má
nedefinovanou hodnotu\n";
  }
  
  $promenna = 2005;
  
  if ($promenna){
      print "7. blok bude vykonán: v proměnné je uložena hodnota 2005, která je true\n";
  }

Uložte si tento kód do souboru a zkuste spustit.

Podmínka if...else

Takto zapsaná podmínka má 2 větve - if a else.

Pokud je pomínka vyhodnocena jako false, automaticky se provede blok kódu, uvedený za else.

  $promenna = 2005;
  
  if ($promenna){
      print '1. blok bude vykonán, pokud je $promenna vyhodnocena jako true';
  }else{

      print '2. blok bude vykonán, nebude-li vykonán blok 1. (to nastane nebude-li $promenna
vyhodnocena jako true, ale jako false)';
  }

Zkuste změnit hodnotu testované proměnné na false.

Podmínka if...n*elsif...(else)

Nejdříve se testuje hodnota v testu za if. Není-li true (blok se nevykoná), vyhodnotí Perl test za elsif.
Není-li ani ten úspěšný, vyhodnocuje Perl další větev elsif a takto pokračuje, dokud není některý test
vyhodnocen jako true. Stane-li se tak, je následující blok vykonán a ostatní větve přeskočeny. Větví elsif
můžete použít libovolné množství. A jestliže testy ve všech větvích dopadnou jako false, vykoná se blok za
větví else (pokud je uvedená).

  $definovana_promenna = 2005;
  
  if ($nejaka_promenna){
      print "1. blok nebude vykonán: proměnná není definována\n";
  }elsif($dalsi_promenna){
      print "2. blok nebude vykonán: ani tato proměnná není definavána\n";
  }elsif($jeste_dalsi_nedefinovana_promenna){
      print "3. blok nebude vykonán: co dodat\n";
  }elsif($definovana_promenna){
      print "4. blok bude vykonán: test byl vyhodnocen jako true\n";
  }elsif(0){
      print "5. blok nebude vykonán: všechny větve za úspěšně vykonaným 4. blokem budou
předskočeny, navíc 0 je false\n";
  }elsif(3.1416){
      print "6. blok nebude vykonán: i přesto, že test by byl true. Je totiž také
přeskočen\n";
  }

Ještě příklad s větví else:

  $definovana_promenna = 2005;
  
  if ($nejaka_promenna){
      print "1. blok nebude vykonán: proměnná není definována\n";
  }elsif($dalsi_promenna){
      print "2. blok nebude vykonán: ani tato proměnná není definavána\n";
  }elsif(""){
      print "3. blok nebude vykonán: ještě jednou false\n";
  }else{
      print "4. blok bude vykonán: to proto, že doteď žádný test nebyl vyhodnocen jako
true\n";
  }

Podmínkový operátor ?...:

Zejména u krátkých podmínek nebo uvnitř výrazu se tento zápis často používá. Obecný zápis vypadá takto:

  test ? příkaz_v_případě_true : příkaz_v_případě_false;

Znamená to samé, co zápis:

  if (test){
      příkaz_v_případě_true;
  }else{
      příkaz_v_případě_false;
  }

Konkrétní příklad:

  1 ? print "TRUE\n" : print "FALSE\n";

Protože 1 je true vytiskne TRUE.

Alternativní podmínky

To nejsou všechny podmínky, které Perl obsahuje. Mimo výše zmiňovaných existuje podmínka s opačným
významem než má if a příkazy s podmínkou na konci. Pokud s programováním začínáte, tak se zbytečně
nehlťte a následující část zatím přeskočte.

Podmínka unless...(n*elsif)...(else)

Už z názvu je patrné, co bude unless dělat. Česky se dá if vyjádřit slovem jestliže, unless slovy 
jestliže ne. Je to protiklad příkazu if. Kód ve větvi unless se provede, vyhodnotí-li se test jako
false. Používat ho klidně můžete, ale dá se převést na if.

V těle unless lze použít jak elsif, tak else. Použití elsif může být poněkud matoucí. Unless totiž
provede blok, je-li test false, elsif je-li test true.

Teď příklad na unless (ovšem dá se jednoduše převést na if...else):

  $definovana_promenna = 2005;
  
  unless ($definovana_promenna){
      print "1. blok nebude vykonán: test vrátí true (proměnná je definována) a vykonává se
větev else\n";
  }else{
      print "2. blok bude vykonán\n";
  }

Podmínka za příkazem

  $bylo_hlasovano = 1;
  
  print "Díky za hlas!" if ($bylo_hlasovano);
  print "Nehlasoval jste!" unless ($bylo_hlasovano);

Příkaz se provede, je-li test true (if) resp. false (unless).

Podmínka za blokem

Stejné jako podmínka za příkazem až na to, že se místo 1 příkazu vykoná celý blok. Před blok musíte uvést
klíčové slovo do. Lze použít jak konstrukci do...if, tak i do...unless.

  do{
      $x = 2 ** 10;
      print $x;
  }if (1);

Za testem musí být středník. To proto, že za ním už není blok, ale následují další příkazy.

Podmínka case a switch

To je jedna z mála běžných konstrukcí, které Perl nenabízí. Lze ji ale napsat pomocí bloků. Dá se použít ve
speciálních případech místo konstrukce if...mnoho*elsif...(else).

  $p = 2;
  
  CASE:{
      $p == 1 and do{print '$p je jedna!'; last CASE;};
      $p == 2 and do{print '$p je dva!'; last CASE;};
      $p == 3 and do{print '$p je tři!'; last CASE;};
      $p == 4 and do{print '$p je čtyři!'; last CASE;};
      do{print '$p není 1, 2, 3 ani 4!';}; #Tento do blok bude proveden v případě, že
nevyhověla žádná předešlá podmínka

  }

Platí-li pravdivostní výraz na začátku řádku, vykoná se blok za klíčovým slovem do a dál program pokračuje
za blokem označeným CASE (značení bloků a klíčové slovo last probereme až v některém z dalších dílů,
pokud nerozumíte, nelámejte si s tím hlavu).

Online verze článku: http://www.linuxsoft.cz/article.php?id_article=727

Perl (6) - Pravdivostní výrazy

Výraz je něco, bez čeho nelze napsat ani ten nejkratší program. Je totiž součástí každého příkazu.

11.4.2005 06:00 | Jiří Václavík | přečteno 1816x 

Výrazy jsou důležitým kamenem programovacích jazyků. Každý zápis, jehož vyhodnocením získáme nějakou
hodnotu, je výrazem. S jednoduchými výrazy už jsme se setkali. Mám tím na mysli konstanty (tedy samotná
čísla), proměnné, příkaz print a další. Dnes arzenál výrazů značně rozšíříme.

Výraz se středníkem na konci se stává příkazem. Proto, pokud jakémukoliv příkazu odejmeme středník,
získáme výraz. Příkaz tedy má také nějakou hodnotu - a to i když není potřebná (k čemu nám je, že příkaz
print vrátí hodnotu 1?). Odtud plyne i to, že program je posloupností výrazů a středníků.

Ve skutečnosti je každý výraz pravdivostní. Vyhodnocením libovolného výrazu je totiž možné získat pravdu
nebo nepravdu. I přesto se teď zaměřím jen na výrazy tvořené relačními, logickými a bitovými operátory.

Začneme ukázkou:

  print 9 > 6;
  print 9 > 18;

V 1. případě bylo vytisknuto 1, ve 2. případě "" (tedy prázdný řetězec). Jistě tušíte proč. 9 je skutečně větší
než 6 a protože Perl neobsahuje žádný booleovský datový typ, uchovávající false nebo true, tiskne místo
true hodnotu 1. Naopak prázdný řetězec značí false. Tento test můžeme užít v podmínce:

  $a = 3;
  $b = 2;
  
  if ($a > $b){# $a je skutečně větší než $b (3 > 2). Výsledkem tohoto výrazu je 1 - tedy
true
      print '$a je větší než $b';
      print "\n";
  }else{
      print '$b je větší než $a';
      print "\n";
  }

Všimněte si, že jsem použil apostrofy jako označení řetězce. To proto, aby se místo $a a $b nevytiskly jejich
hodnoty. (i když by to v tomto případě zas tolik nevadilo)

Zkusme ještě další příklad:

  if ($zisk < 0){
      if ($zakazky == 0){
          $bankrot = 1;
      }else{
          $zamestnancu *= 3/4;
      }
  }elsif ($zisk > 1e7){
      $plat *= 1.05;
  }

Operátory pro porovnávání (relační operátory):



LINUX SOFTWARE file:///C:/Ikon/Books/Perl/LinuxSoft_PERL.html

20 of 102 3/8/2006 20:54

Operace Operátor pro čísla Operátor pro řetězce Celý název

je rovno == eq EQual

není rovno != ne Not Equal

je menší než < lt Less Than

je větší než > gt Greater Than

je menší nebo rovno než <= le Less than or Equal

je větší nebo rovno než >= ge Greater than or Equal

je menší, rovno nebo menší než <=> cmp CoMParison

Porovnávání čísel je jasné - řídí se zákony matematiky.

Ale jak se porovnávají řetězce? Perl vezme 1. znak levého operandu a porovná jej s pravým. To samé s druhým
atd. Každý znak bere jako číslo, určené jeho ASCII hodnotou. 1. odlišnost, kterou Perl objeví rozhodne o
výsledku porovnání. Není-li žádná odlišnost, výrazy jsou si rovny.

  $a = 4;
  $b = 8;
  
  #výraz $a == $b bude vyhodnocen jako false. 4 není rovno 8
  if ($a == $b){ 
      print "$a == $b\n";
  }
  
  #výraz $a != $b bude vyhodnocen jako true. 4 není rovno 8.
  if ($a != $b){ 
      print "$a != $b\n";
  }
  
  #výraz $a > $b bude vyhodnocen jako false. 4 není větší než 8.
  if ($a > $b){ 
      print "$a > $b\n";
  }
  
  #výraz $a != $b bude vyhodnocen jako true. 5 * 4 je větší než 2 * 8.
  if (5 * $a >= 2 * $b){ 
      print "5 * $a >= 2 * $b\n";
  }
  
  #'a' je v ASCII tabulce paradoxně až za 'A', takže je větší. Platí, že jakékoliv malé
písmeno je vždy větší než jakékoliv velké písmeno.
  if ("a" gt "A"){ 
      print "a > A\n";
  }
  
  #psal jsem, že u číselného porovnávání se Perl řídí vždy matematicky správně. Zde to ale
zdánlivě neplatí:) "1slon" bude vyhodnocen jako 1, "slon" jako 0. 1 > 0
  if ("1slon" == "slon"){ 
      print "1slon > slon\n";
  }
  
  #slon sice není 1slon, ale je to myš. Oba výrazy budou vyhodnoceny jako 0, tudíž se
rovnají.
  if ("slon" == "myš"){ 
      print "slon == myš\n";
  }

Operátor <=> vrací hodnotu 1, je-li pravý výraz větší než levý, hodnotu -1, je-li levý výraz větší než pravý
nebo hodnotu 0, jsou-li výrazy na obou stranách stejné.

  $a = 1;
  $b = 2;
  
  print $a <=> $b; #-1
  print $b <=> $a; #1
  print $a <=> 1;  #0



LINUX SOFTWARE file:///C:/Ikon/Books/Perl/LinuxSoft_PERL.html

21 of 102 3/8/2006 20:54

Logické operátory

Obyčejné testy většinou nestačí. Občas potřebujeme podmínku typu je-li a > b a zároveň a > d. 
Od toho jsou logické operace.

Operace Operátor Céčkovský zápis

Konjunkce (logické a) and &&

Disjunkce (logické nebo) or ||

Negace not !

Céčkovský zápis operátorů má stejný význam jako operátory, liší se jen v prioritě.

Konjunkce je splněna, jsou-li oba její operandy true. Disjunkce je splněna, je-li alespoň 1 její operand true
(je-li true 1. operand, 2. je přeskočen a už se nevyhodnocuje - výsledek by to neovlivnilo).

a b a and b a or b

1 1 1 1

1 0 0 1

0 1 0 1

0 0 0 0

Negace je unární, vrací opačnou hodnotu než její operand.

  $a = 4;
  
  #1 je true, $a také, proto je test vyhodnocen jako true
  if (1 and $a){ 
      print "1. test\n";
  }
  
  #1 je true, ale prázdný řetězec ne, proto je test vyhodnocen jako false
  if (1 and ""){ 
      print "2. test\n";
  }
  
  #1 je true, výsledek je taky true
  if (1 or ""){ 
      print "3. test\n";
  }
  
  #1 je true, výsledek je taky true
  if (1 or 2){ 
      print "4. test\n";
  }
  
  #1 je true, negace obrací výraz, takže test je false
  if (!1){ 
      print "5. test\n";
  }
  
  #1 and "MP" je true, potom je true i 0 or (1 and "MP"). Tento celý výraz je operandem
druhého or a protože je true, je i tato operace or true. Výsledek negujeme a dostáváme
false.
  if (!(0 or (1 and "MP") or ($a and ""))){ 
      print "6. test\n";
  }

Přiřazení jako test podmínky

Jako test se hojně používá i přiřazení. Přiřazujete-li hodnotu true, je test true. Přiřazujete-li hodnotu
false, je test také false.

  if ($a = 0){ #0 je false
      print "TRUE\n";
  }else{
      print "FALSE\n";
  }

Jen připomínám, že to není porovnávání.

Absolutní hodnota

Zkusme napsat program, který vypíše absolutní hodnotu čísla.

Stačí vynásobit hodnotu číslem -1, pokud je záporná.

  #!/usr/bin/perl
  use strict;
  
  my $hodnota = -3;
  
  if ($hodnota < 0){
      $hodnota *= -1;
  }
  
  print "$hodnota\n";

Chcete-li, dá se podmínka pomocí podmínkového operátoru zahrnout přímo do funkce print.

  #!/usr/bin/perl
  use strict;
  
  my $hodnota = 3;
  
  print $hodnota < 0 ? $hodnota * -1 : $hodnota . "\n";

V tomto případě se nejdříve vyhodnotí výraz $hodnota < 0 ? $hodnota * -1 : $hodnota a poté
se tiskne.

Bitové operátory

Ještě se zmíním o dalším typu operací. Pokud vás nezajímají, klidně tuto část zatím přeskočte. Bitové operace
mají význam jen v případě celočíselných operandů (zadáte-li desetinné číslo, Perl si desetinnou část odřízne).

Operátor Operace Počet operandů

~ bitová negace (bitové not) 1

& bitový součin (bitové and) 2

| bitový součet (bitové or) 2

^ exkluzivní bitový součet (bitové xor - výlučné nebo) 2

>> bitový posun doprava 2

<< bitový posun doleva 2

Bitové operátory fungují tak, že je operand(y) převeden do dvojkové soustavy. U operátorů &, | a ^ Perl
postupně porovnává bity na stejných pozicích.

bitové not

Každý výsledný bit je doplňkem bitu operandu do 1. Funguje tudíž na různých počítačích jinak (podle rozsahu
čísla). Zkusíme negaci pro 110101110000101:

      0b11111111111111111111111111111111
  $a  0b00000000000000000110101110000101

      ----------------------------------
  ~$a 0b11111111111111111001010001111010

bitové and

Pokud jsou v pravém i levém operandu bity na stejných pozících 1, výsledkem na daném bitu je 1. V ostatních
případech 0.

  $a = 1631;
  $b = 55;
  print $a & $b; #23

1631 a 55 vyjádříme ve dvojkové soustavě. Odtud je zřejmé, jak se k výsledku došlo.

  $a      0b11001011111    1631
  $b      0b     110111      55
          -------------    ----
  $a & $b 0b00000010111      23

bitové or

Pokud jsou v operandech oba bity na stejných pozicích 0, výsledkem bitu na dané pozici ve výsledku je 0.
Vyskytne-li se v jednom z nich 1 (tedy všechny ostatní případy), výsledkem je 1.

  $a = 1322;
  $b = 986;
  print $a | $b; #2042

To je zřetelné odtud:

  $a      0b10100101010    1322
  $b      0b 1111011010     986
          -------------    ----
  $a | $b 0b11111111010    2042

bitové xor

Číslice 1 a 0 nebo 0 a 1 vyhodnotí jako 1, v ostatních případech 0.

  $a = 30549;
  $b = 6806;
  print $a ^ $b; #28099

Bitově:

  $a      0b111011101010101    30549
  $b      0b001101010010110     6806
          -----------------    -----
  $a ^ $b 0b110110111000011    28099

bitový posun

Operátor >> umazává zprava bity, operátor << (zleva) přidává nuly.

  $a = 61;
  $b = 2;
  print $a >> $b; #15
  print $a << $b; #244

Bitově:

  $a        0b  111101     61
  $b                 2      2
            ----------    ---
  $a >> $b  0b    1111     15
  $a << $b  0b11110100    244

Nakonec ještě přehled bitových operátorů:

bit a bit b a & b a | b a ^ b

0 0 0 0 0

0 1 1 1 1

1 0 1 1 1

1 1 1 1 0

Zdroje

man perlop

Online verze článku: http://www.linuxsoft.cz/article.php?id_article=755

Perl (7) - Vstup poprvé

Dosud jsme psali jen programy, které s uživatelem komunikovaly jednostranně. Tento díl vás naučí pracovat s
ovladači umožňujícími i standartní vstup.

20.4.2005 15:00 | Jiří Václavík | přečteno 1870x 

Ovladač (manipulátor) slouží ke čtení dat ze zdroje (zpřístupňuje zdroj dat) nebo naopak zasílání dat. V Perlu
jsou některé ovladače už vestavěné. Mezi ně patří mimo jiné STDIN (čte ze standartního vstupu - klávesnice) a
STDOUT (tiskne na standartní výstup). Ovladače zobecňují práci se zdroji dat. Ze souboru tak čtete stejným
způsobem jako ze standartního vstupu.

Čtení dat z klávesnice

Práce s ovladačem je velmi jednoduchá. Zkuste si spustit následující program:

  print "Jak se jmenuješ?\n";
  
  $jmeno = <STDIN>;
  chomp $jmeno;
  
  print "Jsi $jmeno!\n";

Do proměnné $jmeno přiřazujeme hodnotu získanou ze standartního vstupu - STDIN. Chceme-li data
získávat, musíme dát STDIN mezi < a >. Program čeká, až něco napíšete a odentrujete. V té chvíli se do
proměnné $jmeno uloží získaná hodnota a skript normálně pokračuje dál. Stojí za to připomenout, že do
proměnné $jmeno se ukládá i ENTER v podobě znaku konce řádku.

Na dalším řádku voláme funkci chomp (o funkcích bude příští díl), které předáváme jako parametr proměnnou
$jmeno. Funkce chomp odstraňuje z konce řetězce znak konce řádku, pokud tam je. Zkuste tento řádek
smazat a rozdíl bude ihned zřejmý.

Perl umožňuje tyto 2 řádky zkrátit do jednoho:

  chomp ($jmeno = <STDIN>);

Poznámka: o operátoru <> se někdy mluví jako o diamantovém operátoru nebo operátoru dvojité šipky.

Standartní výstup

Funkci print používáme, aniž bychom se starali o to, kam posílá data. Vždy se objevily na standartním

výstupu. Tak se print chová implicitně. My ale můžeme ovladač změnit. Ve skutečnosti print přijímá ještě
jako další parametr ovladač výstupu. Následující zápisy jsou ekvivalentní.

  print "Toto se tiskne na výstup\n";
  print STDOUT "Toto se tiskne na výstup\n";

Jak vidíte, stačí před tisknutý řetězec uvést ovladač. Od řetězce je oddělen jen bílým znakem. Protože nevrací
žádná data, nepíše se mezi <>.

Chcete-li tisknout text do souboru, je postup úplně stejný.

  print OVLADACSOUBORU "Toto se tiskne do souboru\n";

Jak vytvořit ovladač souboru bude vysvětleno až v díle o práci se soubory, takže se tím dnes více zabývat
nebudu.

V minulém díle jsme napsali program, který počítal absolutní hodnotu čísla. Dnes ho vylepšíme tak, aby počítal
absolutní hodnotu čísla, které bude přečteno ze standartního vstupu.

  #!/usr/bin/perl
  use strict;
  
  my $hodnota;
  print "Zadejte hodnotu:";
  chomp ($hodnota = <STDIN>);
  
  if ($hodnota < 0){
      $hodnota *= -1;
  }
  
  print "Absolutní hodnota zadaného čísla je $hodnota\n";

V další ukázce načte program ze standartního vstupu tentokrát ne 1, ale 3 čísla. Výsledkem bude jejich průměr,
součet a největší a nejmenší ze zadaných čísel. Algoritmus není nijak složitý, takže ho není třeba podrobně
popisovat.

  #!/usr/bin/perl
  use strict;
  
  my ($a, $b, $c);                             #hodnoty, zadané ze vstupu
  my ($soucet, $prumer, $nejvetsi, $nejmensi); #proměnné, ve kterých budou výsledky
  #proměnné lze definovat i takto. my přijímá jako argument seznam skalárních hodnot
  
  print "Zadejte 1. hodnotu: "; chomp ($a = <STDIN>);
  print "Zadejte 2. hodnotu: "; chomp ($b = <STDIN>);
  print "Zadejte 3. hodnotu: "; chomp ($c = <STDIN>);
  
  $soucet = $a + $b + $c;
  $prumer = $soucet / 3;
  
  if ($a > $b and $a > $c){
      $nejvetsi = $a;
  }elsif($b > $c){
      $nejvetsi = $b;
  }else{
      $nejvetsi = $c;
  }
  
  if ($a < $b and $a < $c){
      $nejmensi = $a;
  }elsif($b < $c){
      $nejmensi = $b;
  }else{
      $nejmensi = $c;
  }
  
  print "Součet zadaných čísel je $soucet\n";
  print "Průměr zadaných čísel je $prumer\n";

  print "Největší ze zadaných čísel je číslo $nejvetsi\n";
  print "Nejmenší ze zadaných čísel je číslo $nejmensi\n";

S pomocí cyklů, polí a funkcí by se dal předchozí příklad řešit lépe. Nám toto řešení zatím bude stačit.

Ovladač <>

Mezi < a > být ve skutečnosti žádný identifikátor nemusí. V tom případě bere Perl data ze souboru, který jste
přidali programu jako argument. Pokud jste argument nepřidali, má <> stejný význam jako <STDIN>.
Následující příklad to ilustruje.

Máme soubor kralove s obsahem:

  Přemysl Otakar I. 1197-1230
  Václav I. 1230-1253
  Přemysl Otakar II. 1253-1278

A vlastní program kralove.pl:

  $radek = <>;
  print "$radek\n";

Spusťte program bez argumentu:

  $ perl kralove.pl
  Perl ceka na standartni vstup - neuvedli jsme parametr [ENTER]
  Perl ceka na standartni vstup - neuvedli jsme parametr
  $

A s názvem souboru jako argumentem:

  $ perl kralove.pl kralove
  Přemysl Otakar I. 1197-1230
  $

V 1. případě program žádný argument nedostal, takže měl zápis <> stejný význam jako <STDIN> - zdrojem
dat byl standartní vstup. V 2. případě byl argumentem soubor s 1. českými krály a Perl jako zdroj dat určil
právě soubor kralove. Proto načítal z něj.

Ještě podotknu, že Perl načte ze zdroje dat vždy jen 1 řádek. Proč, to bude blíže vysvětleno v dílu o kontextech.

Počet čtení ze zdroje dat

Ve speciální proměnné $. je uložen počet přístupů k ovladači. Každý ovladač má své počítadlo. $. obsahuje
hodnotu aktuálního (tedy většinou toho posledně použitého) ovladače. Chování této proměnné vysvětluje
následující kód. Nejprve čteme několikkrát ze STDIN, přičemž se $. inkrementuje. Poté čteme z jiného
ovladače a $. se inkrementuje opět od 0. Nakonec zavoláme ještě STDIN, proměnná $. se nenuluje, ale
pokračuje tam, kde u STDIN skončila.

  $a = <STDIN>; print $.;#1
  $b = <STDIN>; print $.;#2
  $c = <STDIN>; print $.;#3
  $d = <>;      print $.;# 1
  $e = <>;      print $.;# 2
  $f = <STDIN>; print $.;#4
  

I přesto je možné získat počet čtení z jiného než posledně použitého ovladače. Funkcí seek se nastavuje
aktuální ovladač.

Tématem 8. dílu budou funkce.

Online verze článku: http://www.linuxsoft.cz/article.php?id_article=804

Perl (8) - Funkce

Další ze základních kamenů programovacích jazyků jsou bezesporu funkce.

2.5.2005 07:00 | Jiří Václavík | přečteno 1931x 

Jako snad v každém programovacím jazyce jsou i součástí Perlu funkce. Funkce je nějaký program uvnitř
programu, kterému je obvykle předáván parametr nebo parametry (pravidlem to ale není) a na základě něj vrátí
funkce určitou hodnotu, vytiskne něco na výstup apod. Protože mají funkce návratovou hodnotu, jsou to
vlastně výrazy. 2 funkce jsme již poznali: print a chomp.

Parametr funkce se zapisuje do závorek. Ale jestliže závorky neuvedete, Perl většinou sám pozná, co
argumentem je a co ne. Proto můžete závorky téměř vždy vynechat (tuto vlastnost ostatně u print používáme
stále).

Matematické funkce

Absolutní hodnota potřetí

Určitě pamatujete, jak jsme psali program na výpočet absolutní hodnoty. To je příklad jako kovaný pro použití
funkce. Místo zdlouhavého testování, zda je vstupní číslo záporné nebo není, použijeme předdefinovanou
funkci abs:

  #!/usr/bin/perl
  use strict;
  
  my $hodnota;
  print "Zadejte hodnotu:";
  chomp ($hodnota = <STDIN>);
  
  $hodnota = abs $hodnota;
  print "Absolutní hodnota zadaného čísla je $hodnota\n";

Funkce abs přijímá jako parametr číslo a vrací jeho absolutní hodnotu. Tu lze získat přiřazením této funkce do
proměnné. Pokud chcete návratovou hodnotu funkce přímo vytisknout, uveďte ji jako parametr funkce print.

Goniometrické funkce

Funkce Popis Příklad Výsledek

sin(hodnota_v_rad) Vrací sínus parametru sin 3.141592 0.000...

cos(hodnota_v_rad) Vrací kosínus parametru cos 3.141592 -0.999...

atan2(x, y) Vrací arkus tangens podílu x / y atan2 (1, 1) 0.785...

Víme, že tan-1 (1) = pi / 4. Po vynásobení rovnice číslem 4 tak získáme Ludolfovo číslo. Příklad také
ilustruje nepovinnost psát závorky i v případech, kdy má funkce více argumentů.

  print 4 * atan2 1, 1; #3.14159265358979

Další matematické funkce

Funkce Popis Příklad Výsledek

log(hodnota) Přirozený logaritmus (základem je Eulerovo číslo) log 1 0

exp(exponent) Vrací Eulerovo číslo umocněné exponentem exp 1 2.718...

sqrt(číslo) 2. odmocnina čísla sqrt 4 2

int(číslo) Vrací celou část čísla (parametru) int 4.7 4

oct(číslo_v_8kové_soustavě) Převádí z 8kové do 10kové soustavy oct "777" 511

hex(číslo_v_16kové_soustavě) Převádí z 16kové do 10kové soustavy hex "ff" 255

Funkce pro práci s řetězci

Funkce Popis Příklad Výsledek

chomp(řetězec) Odstraní případný znak \n./td> $p = "abc\n";chomp $p abc

chop(řetězec) Odstraní poslední znak řetězce $p = "abc";chop $p ab

uc(řetězec) Převede v řetězci malá písmena na velká uc "AbCdE" ABCDE

lc(řetězec) Převede v řetězci velká písmena na malá lc "AbCdE" abcde

ucfirst(řetězec) Převede 1. písmeno na velké uc "abcde" Abcde

lcfirst(řetězec) Převede 1. písmeno na malé lc "ABCDE" aBCDE

length(řetězec) Počet znaků v řetězci length "abcde" 5

chr(ASCII_kód) Vrátí znak náležící ASCII kódu chr 100 d

ord(znak) Vrátí ASCII kód ord "d" 100

Funkce substr

substr je užitečná a často používaná funkce. Pracuje s podřetězci - vyhledává je v řetězcích, případně je
nahrazuje jinými.

Má celkem 4 parametry:

  substr(řetězec, od, kolik, náhrada);

1. parametr určuje řetězec, se kterým bude substr pracovat. 2. parametrem je pozice, která určuje začátek
podřetězce. Další parametr definuje, kolik znaků bude obsahovat podřetězec. Posledním parametrem dáváme
na vědomí, čím chceme vybraný podřetězec nahradit. Povinný je jen 1. a 2. argument.

  print substr "Pchjongjang", 3;
  #Vráceno je vše od 3. pozice napravo - tedy jongjang
  
  print substr "Pchjongjang", 3, 4;
  #Vráceny jsou 4 po sobě jdoucí znaky, z nichž 1. je na pravo od 3. pozice - tedy jong
  
  $slovo = "Pchjongjang";
  substr $slovo, 3, 4, "jing";
  print $slovo;
  #Pchjingjang
  #Jestliže chcete nahrazovat, je nutné předat funkci řetězec v proměnné - ne jen řetězec
(který je vlastně neměnnou konstanou).
  
  $slovo = "Pchjongjang";
  substr $slovo, 3, 4, "";
  print $slovo;
  #Pchjang

Hodnoty od a kolik mohou být také záporné. U od je 1. pozice na konci atd., u kolik určujeme, kolik znaků má
zůstat do konce řetězce.

  $slovo = "Pchjongjang";
  print substr $slovo, -4, 3;
  #jan
  
  $slovo = "Pchjongjang";
  print substr $slovo, -4, -3;
  #j

  $slovo = "Pchjongjang";
  substr $slovo, -4, 4, "jing";
  print $slovo;
  #Pchjongjing
  #Jsou nahrazeny 4 znaky vpravo za 4. pozicí od konce
  
  $slovo = "Pchjongjang";
  substr $slovo, -11, 9, "Beiji";
  print $slovo;
  #Beijing

substr má ještě jednu možnost zápisu. Paradoxem je, že se v něm funkce vyskytuje nalevo od operátoru
přiřazení. To je naprosto unikátní jev (alespoň já jsem se s ním nikde jinde nesetkal). Tyto zápisy mají na
řetězec stejný efekt:

  substr(retezec, od, kolik, náhrada);
  substr(retezec, od, kolik) = náhrada;

Takhle to pak vypadá:

  $slovo = "Pchjongjang";
  substr($slovo, 3, 4) = "XXX";
  print $slovo; #PchXXXjang
  
  $slovo = "Pchjongjang";
  $f = (substr($slovo, 3, 4) = "XXX");
  print $slovo;#PchXXXjang
  print $f;#XXXj

Funkce index a rindex

Obě funkce mají 3 parametry:

  index(řetězec, hledaný_podřetězec, od);
  rindex(řetězec, hledaný_podřetězec, od);

Tyto funkce vyhledávají podřetězce uvnitř řetězců. Vrací pozici 1. znaku hledaného podřetězce v řetězci.
Poslední parametr je nepovinný a specifikuje pozici, od které začíná hledání. Není-li v řetězci podřetězec
nalezen, vrací funkce -1. index hledá 1. výskyt, rindex poslední výskyt.

  print index "Zimbabwe", "w";    #6
  print rindex "Zimbabwe", "w";   #6
  print index "Zimbabwe", "b";    #3
  print rindex "Zimbabwe", "b";   #5
  print index "Zimbabwe", "b", 4; #5
  print rindex "Zimbabwe", "b", 3;#3
  print rindex "Zimbabwe", "x";   #-1
  print rindex "Zimbabwe", "Z";   #0

Funkce die

Uvedení die znamená konec programu. Ty příkazy, které jsou za die, se neprovádí. Jako parametr je možné
uvést hlášku, která se zobrazí na výstupu.

  if ($selhani == 1){
      die "Nepovedlo se zapsat data!\n";
  }

Pokud parametr neukončíte koncem řádku, automaticky se vypíše i místo, kde byla funkce die volána.

Funkce defined

Rozlišuje nedefinovanou hodnotu od všech ostatních - včetně 0 nebo prázdného řetězce. V případě
nedefinované hodnoty vrací false. Lze ji užít i na detekci podprogramů.

Příklad - řešení rovnice 2. stupně

Programovat se nenaučíte čtením manuálů nebo opisováním kódu z učebnic, ale psaním svých programů. Proto
doporučuji, abyste si každý příklad, který v seriálu uvedu, nejdříve zkusili napsat sami.

Napište program, který na základě zadání koeficientů ze standartního vstupu spočítá kořeny kvadratické
rovnice v množině reálných čísel a vypíše je na výstup (nezapomeňte na všechny podmínky).

Řešení

Takto rozsáhlý příklad jsme zatím nedělali, ale sám algoritmus je velmi jednoduchý.

U všech větších příkladů se budu snažit postupovat tak, jak byste ho pravděpodobně řešili. Nemám rád zvyk,
kdy je v učebnicích pro začátečníky zdrojový kód vysvětlován řádek po řádku, aniž by byly přiblíženy
myšlenky, kterými se k výslednému kódu dospělo.

Tento příklad je ale jedním z těch jednodušších, které lze řešit i metodou od 1. řádku k poslednímu. Takže
nejdříve načteme data ze vstupu:

  #!/usr/bin/perl
  use strict;

  my ($a, $b, $c, $x1, $x2);
  
  print "Zadejte koeficienty kvadratické rovnice ax^2 + bx + c = 0\n";
  print "a = "; $a = <STDIN>;
  print "b = "; $b = <STDIN>;
  print "c = "; $c = <STDIN>;

a poté už můžeme začít s výpočtem. Vzorec x1 = (-b + odmocnina(b^2 - 4*a*c)) / (2*a)
popř. x2 = (-b - odmocnina(b^2 - 4*a*c)) / (2*a) jistě znáte. Je nutné si uvědomit
podmínky řešitelnosti - tj. dávat si pozor jestli neodmocňujeme záporné číslo, nedělíme 0 nebo se nemění počet
řešení. Nejprve ošetříme podmínkami speciální případy a poté dopíšeme obecné řešení.

Ošetřme tedy případ, kdy je diskriminant záporný (pro větší názornost používám zápis $b**2 - 4 * $a * 
$c < 0, ale nic vám nebrání nerovnici upravit a použít $b**2 < 4 * $a * $c):

  if ($b**2 - 4 * $a * $c < 0){ 
      die "V množině reálných čísel nemá rovnice řešení.\n";
  }

Pokud je diskriminant roven 0, řešení je jen jedno:

  if ($b**2 - 4 * $a * $c == 0 and $a != 0){ 
      $x1 = $b / (2 * $a);
      die "Řešením dané rovnice je $x1.\n";
  }

Další speciální situace nastane, je-li koeficient a = 0. V tom případě jde o lineární rovnici. Do podmínky
zahrňme i test koeficientu b, protože to je další speciální případ.

  if ($a == 0 and $b != 0){
      $x1 = -$c / $b;
      die "Řešením dané rovnice je $x1.\n";
  }

Je-li totiž i koeficient b = 0, jedná se o rovnici nultého stupně. Ta má jako řešení buď všechna reálná čísla a
nebo řešení nemá.

  if ($a == 0 and $b == 0 and $c != 0){ 
      die "V množině reálných čísel nemá rovnice řešení.\n";
  }
  
  if ($a == 0 and $b == 0 and $c == 0){ 

      die "Řešením rovnice jsou všechna reálná čísla.\n";
  }

Všechny speciální případy jsou vyřešeny, tak se můžeme pustit do obecného řešení. Sem se program dostane
pouze pokud nevyhověly vstupní hodnoty některému předchozímu případu.

  $x1 = ($b + sqrt($b**2 - 4 * $a * $c)) / (2 * $a);
  $x2 = ($b - sqrt($b**2 - 4 * $a * $c)) / (2 * $a);
  
  print "Řešením dané rovnice je $x1 a $x2.\n";
  
  

Poznámka: Samozřejmě to není jediné řešení problému. Další možností by mohla být například podmínka if 
($a == 0){ ... }else{ ... }, v jejíchž větvích by byly další podmínky. Toto řešení by bylo možná

dokonce lepší, ale chtěl jsem vám ukázat mimo sqrt i funkci die.

Zde máme výsledek:

  $ ./kvadraticka_rovnice.pl
  Zadejte koeficienty kvadratické rovnice ax^2 + bx + c = 0
  a = 5
  b = 2
  c = 5
  V množině reálných čísel nemá rovnice řešení
  $ ./kvadraticka_rovnice.pl
  Zadejte koeficienty kvadratické rovnice ax^2 + bx + c = 0
  a = 2
  b = 5
  c = 3
  Řešením dané rovnice je 1.5 a 1.
  $ ./kvadraticka_rovnice.pl
  Zadejte koeficienty kvadratické rovnice ax^2 + bx + c = 0
  a = 5
  b = 10
  c = 5
  Řešením dané rovnice je 1.
  $ ./kvadraticka_rovnice.pl
  Zadejte koeficienty kvadratické rovnice ax^2 + bx + c = 0
  a = sh
  b = hgf
  c = shgfd
  Řešením rovnice jsou všechna reálná čísla.
  $

V posledním případě byly řetězce automaticky konvertovány na čísla.

To je pro tento díl vše, příště začneme s cykly.

Zdroje

man perlfunc

Online verze článku: http://www.linuxsoft.cz/article.php?id_article=811

Perl (9) - Cyklus

Další hojně užívanou řídící strukturou, bez které se neobejde žádný větší program, je cyklus.

23.5.2005 06:00 | Jiří Václavík | přečteno 1805x 

Cyklus umožňuje vykonávat určitou část programu opakovaně. Funguje v podstatě podobně jako podmínka. Na
základě vyhodnocení testu se provede blok kódu. Rozdíl mezi podmínkou a cyklem je v tom, že u cyklu se
vyhodnocuje test opakovaně. Dokud není test vyhodnocen jako false (nebo v některých speciálních cyklech
true), stále se znovu a znovu opakuje blok kódu. Existuje hned několik druhů cyklů.

Cyklus while

Dokud platí test, provádí se blok kódu.

  while (test){
      příkazy
  }

Cyklus by měl mít možnost být ukončen - tj. vepsat nějaký kód, který ho bude přibližovat ke konci, až nakonec
také skončí. Lze to vyřešit například takto:

  $i = 0;
  while ($i < 10){
      print "Probíhá cyklus $i.\n";
      $i++;
  }

10× je proveden kód uvnitř cyklu. Je to stejné, jako kdybychom za sebe nakopírovali 10× blok.

  $ perl while1.pl
  Probíhá cyklus 0.
  Probíhá cyklus 1.
  Probíhá cyklus 2.
  Probíhá cyklus 3.
  Probíhá cyklus 4.
  Probíhá cyklus 5.
  Probíhá cyklus 6.
  Probíhá cyklus 7.
  Probíhá cyklus 8.
  Probíhá cyklus 9.
  $

Je to dost netypická situace pro použití while. V praxi byste bez váhání sáshli po cyklu for. Uvedl jsem ho
proto, že se dá tento zápis velmi snadno pochopit. Teď už ale přijde na řadu skutečné while. Další příklad
totiž do nekonečna vypisuje to, co napíšete. V praxi má velmi časté využití.

  while ($radek = <STDIN>){
      chomp $radek;
      print "Napsal jste: $radek\n";
  }

Po spuštění kódu získáváme výstup:

  $ perl while2.pl
  Tento cyklus lze ukončit jen násilně. [ENTER]
  Tento cyklus lze ukončit jen násilně.
  Sám program, pokud není přerušen, by neskončil nikdy. [ENTER]
  Sám program, pokud není přerušen, by neskončil nikdy.
  [čeká na další vstup]

Zkusme kód poupravit tak, aby se při zadání q nebo exit ukončil. Dosáhneme toho přidáním podmínky
(funkce die často nebude stačit. Příští díl osvětlí lepší řešení).

  while ($radek = <STDIN>){
      chomp $radek;
      if ($radek eq "q" or $radek eq "exit"){
          die "Konec\n";
      }
      print "Napsal jste: $radek\n";
  }

Cyklus for

Cyklus for se od while liší tím, že je předem znám počet iterací (jde sice změnit v bloku, ale potom je často
lepší užít while). for a while jsou vzájemně zaměnitelné (for lze napsat jako while a naopak). Ale vždy
je některý vhodnější, některý méně.

Syntaxe je následující:

  for (počáteční_hodnota_počítadla; konečná_hodnota_počítadla; krok){
      příkazy
  }

Počáteční_hodnota_počítadla spouští počítadlo. Konečná_hodnota_počítadla je vlastně testem. Dokud platí, je
cyklus opakován (tj. proveden blok kódu). A krokem je výraz, který změní hodnotu počítadla.

  for ($i=1; $i<10; $i++){
      print "Probíhá cyklus $i.\n";
  }

Výstup je stejný jako u 1. příkladu na while. Nejprve se přiřadí do proměnné $i hodnota 1. Je-li test true,
provede se blok, proměnná se inkrementuje (krok) a pokud opět platí test, opakuje se vše znovu. Ve chvíli, kdy
test neplatí, je blok přeskočen a program pokračuje za ním.

Složitost výrazů v hlavičce for není omezena jen na takto jednoduché záležitosti. Operátor , (čárka) odděluje
se 2 výrazy, přičemž 1. se vyhodnotí a následně zapomene. 2. výraz se také vyhodnotí a je výsledkem
celkového vyhodnocení obou výrazů.

  for ($i=0,$j=0; $i<10,$j<10; $i++,$j+=2){
      print "Probíhá cyklus $i. J: $j.\n";
  }

Výstup:

  $ perl for.pl
  Probíhá cyklus 0. J: 0.
  Probíhá cyklus 1. J: 2.
  Probíhá cyklus 2. J: 4.
  Probíhá cyklus 3. J: 6.
  Probíhá cyklus 4. J: 8.
  $

Je vidět, že počítadlem řídící cyklus je až 2. část výrazu - za čárkou (s proměnnou $j). Výraz s $i je sice
každou iteraci zapomenut, ale hodnota je vždy změněna.

počáteční_hodnotu_počítadla, konečnou_hodnotu_počítadla  ani krok není povinné uvést. Vynecháte-li
počáteční_hodnotu_počítadla, zůstane v $i hodnota, která tam dosud byla. Pokud je nedefinovaná, v 1. iteraci
bude stále proměnná $i nedefinovaná, ale v dalších už nemusí. Je-li totiž v kroku výraz $i++, v 2. iteraci bude 
hodnota $i = undef + 1 = 1.

Vynecháním konečné_hodnoty_počítadla se cyklus stává nekonečným. Nemá se co vyhodnocovat, a proto je za
všech okolností true.

Neuvedení kroku má za následek, že se počítadlo (aspoň v hlavičce cyklu) nebude iterovat.

Tímto způsobem lze vynechat třeba i všechny 3 výrazy. Zápis for (;;){...} je ekvivalentní while 
(1){...}.

Odtud plyne, že počáteční_hodnota_počítadla není nic jiného, než výraz, který se vyhodnotí před započetím
cyklu, konečná_hodnota_počítadla je testem a krok výrazem, který se vyhodnocuje na konci každé iterace.

Cyklus for podle posledního tvrzení můžeme jednoduše přepsat do while (není to úplně přesné, krok by
správně měl být až v bloku continue, který představím příště. Toto řešení neplatí pro případy, kdy je konec
bloku vynechán):

  počáteční_hodnota_počítadla;
  while (konečná_hodnota_počítadla){
      příkazy
      krok;
  }

for má ještě další možnost zápisu, která je shodná s cyklem foreach. Budeme se jím zabývat v souvislosti s

poli.

Alternativní cyklus until

Stejně jako jsou ve vztahu if a unless je ve vztahu i while a until. Cyklus je prováděn, dokud je test
false.

Cyklus za příkazem

Cyklus while se dá v Perlu napsat i za příkaz. Ukažme si alespoň pro představu zdrojový kód:

  $i = 1;
  
  $i++ while ($i < 10); # k $i se bude tak dlouho přičítat jednička, dokud bude $i nižší než
10
  print $i;#10
  
  $i-- until ($i < 0); # nyní se od $i bude pro změnu jednička odečítat a to tak dlouho,
dokud nebude $i nižší než 0
  print $i;#-1

Cyklus za blokem

Cykly do...while a do...until jsou dalšími variacemi while. Nejprve se provede blok příkazů a až
poté se vyhodnotí test (výsledný efekt je stejný, jako kdybyste blok v cyklu zkopírovali před klasický while). 
Je-li true (while) resp. false (until), vykonává se blok znovu. A stále dokola...

  $i = 0;
  
  do{ #vykoná se i přesto, že $i má hodnotu 0. Testuje se až za blokem.
      print "Vykonáno\n";
  }while ($i == 1);

Za testem musí být stejně jako u konstrukcí do...if a do...unless středník.

I cyklus for lze napsat za příkaz nebo blok. O tom se zmíním až budu rozebírat další cyklus foreach.

Příklad

Napišme program, který bude ze vstupu přijímat kladná čísla. Zadáním nekladného čísla skončí zadávání (toto
číslo se již nepočítá) a program vytiskne počet zadaných čísel, jejich součet a průměr.

Měly bychom získat přibližně takový výstup:

  $ perl cisla.pl
  Zadávejte kladná čísla, 0 pro konec.
  12
  994
  328
  666
  666
  9004
  0
  Počet: 6
  Součet: 11670
  Průměr: 1945
  $

Kostra programu bude vypadat velmi podobně jako dnes již zmíněná ukázka kódu k while. Lišit se bude tím,
že v každé iteraci se bude aktualizovat součet a počet zadaných čísel a před ukončením se ještě vytisknou
výsledky.

  #!/usr/bin/perl
  use strict;
  
  my $cislo;
  my ($pocet, $prumer, $soucet);
  
  print "Zadávejte čísla, 0 pro konec.";
  while (chomp($cislo = <STDIN>)){
      if (int $cislo <= 0){
          ...vypsání statistik a konec...
      }
  
  ...mezivýpočet...
  }

V mezivýpočtu stačí navyšovat součet a počet.

      $soucet += $cislo;
      $pocet++;

Ani výpis výsledků není nic složitého. Spočítáme průměr, vypíšeme výsledky a ukončíme.

          $prumer = $soucet / $pocet;
          print "Počet: $pocet\n";
          print "Součet: $soucet\n";
          print "Průměr: $prumer\n";
          die "\n"; #Pokud bychom neuvedli konec řádku, vypsala by automatická hláška (Tento
řádek by šel lépe řešit pomocí funkce exit, která zatím nebyla probrána)

Když dopíšete program, je vždy nutné otestovat jeho reakce v závislosti na podmínkách. Všechny případy musí
být ošetřeny. To my nemáme, protože program hlásí chybu, pokud neuvedeme žádný parametr. Takže na
začátek bloku v podmínce připišme:

          if ($pocet == 0){
              die "Nezadal jste žádné číslo!\n";
          }

A tím je úloha splněna. Zdrojový kód příkladu.

Příště o dalších klíčových slovech a napíšeme si i jednoduchou hru.

Online verze článku: http://www.linuxsoft.cz/article.php?id_article=828

Perl (10) - Další řídící struktury

Dnes mimo jiné o klíčových slovech, používaných v cyklech, návěštích a bloku continue.

9.6.2005 07:00 | Jiří Václavík | přečteno 1765x 

last, next, redo

Existuje i jiný způsob jak řídit cykly, než pomocí testu. Ve skutečnosti můžete napsat nekonečný cyklus a
ukončit ho až za určitých podmínek v jeho bloku. Cyklus tak vůbec nemusí celý řádně proběhnout. V minulém
dílu jsme k tomuto účelu použili funkci die. Ta však nevyskočí jen z cyklu, ale přímo ukončí program.
Následující informace platí pro všechny cykly, které zatím známe.

Klíčové slovo last uvedené v těle cyklu cyklus okamžitě ukončuje. Provádění programu pokračuje
bezprostředně za cyklem (za uzavírací složenou závorkou).

Oproti tomu next přeskočí pouze zbytek bloku a provádění pokračuje další iterací (má stejnou funkci jako
céčkovské continue).

U obou těchto klíčových slov je možný parametr, kterým je jméno návěští. To použijete, pokud se provádí

cyklus uvnitř jiného cyklu. Potom vyskočíte z obou.

Teď uvedu několik ukázek kódu. Cyklus while zde používám jen kvůli lepší názornosti. V praxi byste použili
u většiny nejspíše for. (Praktická ukázka bude na konci článku)

  $i = 0;
  while ($i < 9){
      $i++;
      if ($i == 5) {
          last;
      }
      print "Probíhá cyklus $i.\n";
  }

Proběhne pouze 5 cyklů, protože v 5. cyklu je klíčovým slovem last cyklus ukončen a program pokračuje za
blokem:

  $ perl last.pl
  Probíhá cyklus 1.
  Probíhá cyklus 2.
  Probíhá cyklus 3.
  Probíhá cyklus 4.
  $

Podobný program, jen s drobnou modifikací - vyměnímě last za next:

  $i = 0;
  while ($i < 9){
      $i++;
      if ($i == 5) {
          next;
      }
      print "Probíhá cyklus $i.\n";
  }

Na začátku 5. cyklu je cyklus přerušen a pokračuje dalším cyklem. Proto není u 5. cyklu zaznamenán výpis:

  $ perl while1.pl
  Probíhá cyklus 1.
  Probíhá cyklus 2.
  Probíhá cyklus 3.
  Probíhá cyklus 4.
  Probíhá cyklus 6.
  Probíhá cyklus 7.
  Probíhá cyklus 8.
  Probíhá cyklus 9.
  $

Příkaz redo má stejný účinek jako next, jen nedojde ke změně počítadla.

  for ($i=1; $i<10; $i++){
      $i++;
      if ($i == 4) {
          redo;
      }
      print "Probíhá cyklus $i.\n";
  }

Při 1. iteraci je v bloku zvýšena hodnota $i na 2. V další iteraci je nejprve zvýšena v počítadle, dále v bloku a
má hodnotu 4. To je důvod pro volání redo. Program přeskočí opět na začátek bloku, přičemž počítadlo
hodnotu $i nemění. Ta se změní až v bloku. Aktuální hodnota je tedy 5 a od tohoto okamžiku už pokračuje
skript, aniž by kdykoliv testu podmínky vyhověl.

  $ perl while1.pl
  Probíhá cyklus 2.
  Probíhá cyklus 5.
  Probíhá cyklus 7.
  Probíhá cyklus 9.
  $

Návěští

Další možností řídit cykly je pojmenovat blok a umístit za klíčové slovo last, next nebo redo právě název
bloku (návěští). To musí být definováno. Před cyklus, pro který má návěští platit, se uvádí jeho název
následovaný dvojtečkou. Návěští se pojmenovává velkými písmeny.

  $i = 0;
  NAVESTI:
  while (){ #nekonečný cyklus
      while ($i < 10){
          print "Probíhá cyklus $i.\n";
          if ($i == 3){
              last NAVESTI;
          }
  
          $i++;
      }
  }

last zde nevyskakuje z vnitřního cyklu, ale z cyklu označeného jako NAVESTI.

  $ perl while1.pl
  Probíhá cyklus 0.
  Probíhá cyklus 1.
  Probíhá cyklus 2.
  Probíhá cyklus 3.
  $

Blok continue

continue je nepovinným blokem příkazů na konci cyklu. Je spuštěn při každé iteraci mimo případů, kdy je
cyklus přerušen klíčovým slovem redo nebo last. V následujícím příkladu je proměnná $i iterována i v
případě, kdy je klíčovým slovem next přeskočeno na další iteraci. continue se proto hodí pro opakování
kódu, které by muselo být vyvoláno v každé iteraci. Ovšem často místo použití continue stačí vyměnit
cyklus while za for.

  while ($i < 10){
      if ($i == 5){
          next;
      }
  }continue{
      $i++; #blok continue je proveden vždy. I v případě, že $i == 5.
  }

Klíčové slovo goto

goto funguje podobně jako návěští. Parametrem goto je opět návěští, které ale nepojmenovává blok. Může
být uvedeno před libovolný příkaz. Po goto se bude dále vykonávat část programu, začínající návěštím.

  print "1\n";
  
  goto GOTO;
  
  print "2\n";
  print "3\n";
  
  GOTO:
  
  print "4\n";
  print "5\n";

2. a 3. print nebude vykonán. Když program narazí na goto, bude vše až po návěští přeskočeno:

  $ perl goto.pl
  1
  4

  5
  $

Nic vám nebrání umístit návěští na goto před samotné goto.

  print "1\n";
  $i = 0;
  GOTO:
  
  print "2\n";
  print "3\n";
  
  $i++;
  if ($i != 2){
      goto GOTO;
  }
  print "4\n";
  print "5\n";

Nastává zajímavá situace. Po 2. inkrementaci je goto vynecháno, aby cyklus nebyl nekonečný. Všimněte si, že
mluvím o cyklu, aniž bych použil while nebo for. Slovo cyklus je skutečně na místě, protože je část kódu
vykonávána opakovaně. Pokud se trochu zamyslíte, zjistíte, že jde vlastně logicky o cyklus do...while,
který je jen převlečený do jiného kabátu.

  $ perl goto.pl
  1
  2
  3
  2
  3
  4
  5
  $

Přes tento hezký efekt ale doporučují programátoři goto používat jen ve výjimečných situacích, kdy by bylo
jiné řešení značně složité. Program tím ztrácí na přehlednosti (řekli byste na 1. pohled, že jde v předchozím
úseku kódu o do...while?).

exit, die

Tyto 2 funkce mají podobný význam. die, jak už víme z dílu o funkcích, ukončí okamžitě program. exit
také. Rozdíl je v tom, že exit nevydá žádné hlášení o chybě (nemá žádný výstup).

(Pseudo)náhoda

A již ta slíbená hra. Tedy za chvíli. Budeme k ní totiž potřebovat vygenerovat náhodné číslo.

Musíme inicializovat generátor náhodných čísel. K tomu máme příkaz srand bez parametrů.

  srand;

V tuto chvíli spustil Perl stopky. Zbývá z nich zachytit nějakou hodnotu.

  $i = rand 983;

Do proměnné $i bylo přiřazeno náhodné desetinné číslo z intervalu <0; 983>. Neuvedete-li mez, vrací funkce
true/false.

Hádání myšleného čísla

Ujasňeme si, co má program vlastně dělat. Nejprve si po nás vyžádá číslo, které bude určovat horní hranici
intervalu, ve kterém leží myšlené číslo. Potom bude opakovaně vyzývat k zadání čísla a prozradí vždy, zda je
myšlené číslo větší nebo menší než hádané. Přitom bude počítat pokusy. V případě uhádnutí skončí cyklus.

Na začátku programu je nutné aktivovat generátor:

  #!/usr/bin/perl 
  
  srand;

Dále získáme ze vstupu číslo:

  print "Zadej největší možné číslo: ";
  chomp($max = <STDIN>);

Musíme podmínkou ošetřit případ, kdy je zadané číslo nekladné.

  if (int $max <= 0){
      die "Toto není správně zadané číslo.\n";
  }

Ještě stále nemáme myšlené číslo. Vygenerujeme ho funkcí rand. Je zde ale problém a možná tušíte jaký.
rand vrací desetinné číslo a my chceme celé. Desetinnou část odřízneme funkcí int. To má za vedlejší efekt
posunutí intervalu o jedničku dolů. Nemáme zájem, aby mohlo být tajné číslo 0, proto přičteme jedničku.

  $tajne_cislo = int(rand($max)) + 1;

Teď na řadu přijde samotný cyklus. V něm musí být výzva k zadání čísla a vyhodnocení - zda je menší, větší
nebo rovno myšlenému číslu. Pokud je rovno, vypíšeme hlášku a ukončíme cyklus pomocí klíčového slova
last.

  while (1){
      print "Hádej číslo mezi 1 a $max.\n";
      chomp($hadane_cislo = <STDIN>);
  
      if ($hadane_cislo == $tajne_cislo){
          print "Gratuluji!\n";
          last;
      }elsif($hadane_cislo < $tajne_cislo){
          print "To je málo. ";
      }else{
          print "Moc. ";
      }
  }

Na něco jsme přece jen zapomněli. Počítat pokusy. S výhodou použijeme blok continue. Před cyklem
nastavíme číslo pokusu na 1

  $pokus = 1;

a bezprostředně za cyklus přidáme právě blok continue. V něm budeme přičítat pokusy.

  continue{
      $pokus++;
  }

Protože s počítáním pokusů dříve nepočítali, musíme upravit hlášku v případě uhodnutí tajného čísla:

print "Gratuluji, uhodl jste na $pokus. pokus!\n";

Nakonec ještě přidáme use strict; a definice proměnných. Jsme hotovi, teď spusťte program.

  $ ./cislo.pl
  Zadej největší možné číslo: 100
  Hádej číslo mezi 1 a 100.
  50
  To je málo. Hádej číslo mezi 1 a 100.
  75
  Moc. Hádej číslo mezi 1 a 100.
  63
  To je málo. Hádej číslo mezi 1 a 100.
  69
  To je málo. Hádej číslo mezi 1 a 100.

  72
  Moc. Hádej číslo mezi 1 a 100.
  70
  To je málo. Hádej číslo mezi 1 a 100.
  71
  Gratuluji, uhodl jste na 7. pokus!
  $

Když jsem to zkoušel poprvé a chtěl zkopírovat výstup sem, vyšlo mi z intervalu <0; 100> právě 100;-)

Zdrojový kód příkladu.

Příště začneme s poli.

Online verze článku: http://www.linuxsoft.cz/article.php?id_article=851

Perl (11) - Úvod do polí

Představíme si další datovou strukturu - pole.

21.7.2005 06:00 | Jiří Václavík | přečteno 1653x 

Skaláry a seznamy

Známe zatím 2 druhy skalárních dat. Řetězce a čísla. Skalár (proměnná se označuje dolarem ($) před názvem)
obsahuje jen 1 věc - tedy 1 řetězec nebo 1 číslo. Dosud jsme nepracovali s jinými proměnnými než skalárními.

Seznamy oproti tomu obsahují více položek. V Perlu se zapisují oddělené čárkami v kulatých závorkách.
Proměnnou, která uchovává seznam, je pole. To se značí zavináčem (@) před názvem.

Kontext

Kontext běžně používáte i v řeči. Vezměme si slovo síť. Když zmíníte síť samostatně, nikdo nemůže vědět,
kterou síť myslíte. Použijete-li ji například v souvislosti elektřinou, pravděpodobně půjde o elektrickou síť.
Podobně můžete sítí myslet pavoučí síť, počítačovou síť nebo nějakou další. Záleží na tom, v jakém kontextu
síť užijete. Přesně to je kontext v programování.

Uvedu ještě jeden a možná lépe vypovídající příklad. V matematice nemůžete násobit podtržítka s tečkami
nebo umocňovat tučňáka na kružnici. Operace násobení a umocňování potřebují dostat oba operandy jako
číslo. Pro ostatní případy není výsledek operací definován.

Funkce se chovají k datům podle toho, v jakém jim je předáváte kontextu. Zda jde o seznam nebo skalár (jestli
jde o číslo nebo řetězec - případně logický výraz - nehraje v Perlu příliš velkou roli). Stejná funkce může
fungovat s oběma možnostmi. Pak se ale obvykle liší ve zpracování a výsledku. Předáte-li nějaké funkci skalár,
může třeba vrátit jeho 2. mocninu. Pokud stejná funkce dostane jako parametr seznam, může vracet pole 2.
mocnin původního seznamu, 2. mocninu 1. prvku původního seznamu, nebo klidně něco jiného. Právě proto je
vždy dobré vědet, co děláte (a to platí obecně:).

Pole (array)

Pole je, jak už jsem zmiňoval, identifikátor pro nějaké množství skalárů (čísel, řetězců).

Jde opět o oblast v paměti, do které ale, narozdíl od skalárních proměnných, můžeme uložit celou řadu hodnot.
Pole je seznamem skalárních hodnot.

Skalárním proměnným v poli se říká prvky. Prvek je určen indexem a obsahuje hodnotu. Index je číslo. 1.
indexem je nula, dalším 1 atd. Skalární proměnné se značí dolarem, ale pole zavináčem. Přesto se dolar používá

pro označení jednotlivých prvků pole. Je to logické - prvek pole uchovává sám o sobě jen skalární hodnotu -
například číslo - nikoliv seznam hodnot. Zavináč se používá i v případě pole nebo řezu polem. Řez polem je
totiž opět seznamem.

Práce s poli je v Perlu narozdíl striktních jazyků jako Céčko velmi jednoduchá. Žádná deklarace ani alokace.
Platí to, co u proměnných - prostě můžete pole kdykoliv začít používat.

Vyvstává otázka: Máme přeci proměnné, tak proč používat pole? Oproti proměnným mají pole řadu výhod.
Představte si, že máte databázi jmen lidí. Použijete-li pole, nemusíte vymýšlet jména tolika proměnných, kolik
lidí máte. Nabízejí se vám i další možnosti. Můžete získat počet jmen jediným příkazem nebo jednoduchým
cyklem smazat ty, jejichž příjmení začíná na S. Kdykoliv můžete velmi jednoduše přidat další jméno. Zkrátka
pole jsou skvělým pomocníkem a větší aplikace se bez nich neobejdou.

Operace s poli

Pole vytvoříte, stejně jako skalární proměnnou - přiřazením. Levým operandem je identifikátor, začínající
zavináčem a na pravo se uvádí v kulaté závorce hodnoty jednotlivých prvků, oddělené čárkami (tedy vlastně
seznam):

  @pole = ("hodnota1", "další", "a poslední");

Tím jsme vytvořili 3prvkové pole. Zde je tabulka určující jeho strukturu:

Index Hodnota

0 "hodnota1"

1 "další"

2 "a poslední"

Pole máme, jak z něj číst? Chcete-li získat hodnotu prvku @pole s indexem x, zápis je následující:

  $pole[x]

Pozor na to, prvek pole je skalárem (nikoliv seznamem) a označuje se, jako jsme byli u skalárů zvyklí, dolarem.

Takže, chceme-li získat data z našeho @pole:

  print "Prvek 0: $pole[0]\n";
  print "Prvek 1: $pole[1]\n";
  print "Prvek 2: $pole[2]\n";

Výstup:

  $ perl pole.pl
  Prvek 0: hodnota1
  Prvek 1: dalsi
  Prvek 2: a posledni
  $

Pole můžete editovat i přiřazením do $pole[x]:

  $pole[0] = "Perl";
  $pole[4] = "Linux";

Za předchozí kód přidejte ještě tento:

  print "Prvek 0: $pole[0]\n";
  print "Prvek 1: $pole[1]\n";
  print "Prvek 2: $pole[2]\n";
  print "Prvek 3: $pole[3]\n";
  print "Prvek 4: $pole[4]\n";

Po spuštění dostanete:

  $ perl pole.pl
  Prvek 0: Perl
  Prvek 1: dalsi
  Prvek 2: a posledni
  Prvek 3:
  Prvek 4: Linux
  $

Výstup ukazuje hned několik skutečností. Přiřazením hodnoty do prvku pole přemažete původní obsah (0). Lze
přiřadit hodnotu i do prvku, který bezprostředně nenavazuje na předchozí (4). Prvek (3) existuje, ale zůstal
prázdný. I přesto, že prvek (4) je definován.

Následující 2 zápisy si jsou ekvivalentní:

  @pole = ("a", "b", "c");
  
  $pole[0] = "a";
  $pole[1] = "b";
  $pole[2] = "c";

Užitečná je také funkce qw, která vám šetří ruce. Umožňuje položky v uvozovkách, oddělené čárkou, oddělovat
jen mezerou (obecně bílým místem). Řetězec není v uvozovkách ani apostrofech. Užijete ji hlavně při delších
výčtech. Tyto příkazy mají stejný význam:

  @pole = ("a", "b", "c");
  
  @pole = qw(a b c);

Zkopírujme obsah jednoho prvku pole do jiného. Je to stejné, jako u obyčejných proměnných.

  @pole = qw(a b c);
  
  $pole[3] = $pole[0]; #do prvku s indexem 3 bude přiřazena hodnota, která je v prvku s
indexem 0

Kopírovat můžete i celá pole.

  @pole = qw(a b c);
  
  @pole2 = @pole;#@pole2 je kopií @pole

Jak @pole, tak @pole2 mají stejný obsah.

Stejně tak můžete kopírovat část pole:

  @pole = qw(a b c d e);
  
  @pole2 = ($pole[2], $pole[3], $pole[0]);#@pole2 nyní obsahuje prvky 0, 1, 2 s obsahem po
řadě c, d, a.

Chcete-li přiřadit prvek z @pole s indexem 7 do @pole2 (index 4) a zároveň prvek @pole s indexem 3 do 
@pole2 (index 11), lze použít zápis:

  @pole2[7, 3] = @pole[4, 11];

Ekvivalentním zápisem je:

  $pole2[7] = @pole[4];
  $pole2[3] = @pole[11];

Pokud bude málo prvků v hranatých závorkách (ale platí to obecně, pokud přiřazujete seznam) na pravé straně
přiřazení, doplní se nedefinovanými hodnotami. Přebývají-li, jsou ignorovány prvky navíc zprava.

Podobný příklad jako předchozí:

  @pole2 = @pole[4, 11, 25, 94];#@pole2 nyní obsahuje prvky 0, 1, 2, 3 s obsahem stejným
jako prvky s hodnotami 4, 11, 25, 94 v @poli.

Pole můžete složit i z několika jiných polí.

  @jazyky = qw(Perl Python);
  @cisla = qw(68, 3.14);
  
  @mix = (@jazyky, @cisla); # obsahuje prvky s hodnotami Perl, Python, 68, 3.14

Přidat na konec pole nějakou hodnotu se dá i následovně (později poznáme lepší způsob - funkcí):

  @jazyky = qw(Perl Python);
  @jazyky = (@jazyky, "Ruby");

Seznam proměnných

Perl umožňuje mnoho zvláštních (ale když se zamyslíte, logiku najdete) konstrukcí. Pro zajímavost příklad:

  $p = ("a", "b", "c")[1];
  print $p;

("a", "b", "c") je vlastně seznamem (polem). Z něj se do $p přiřadí prvek s indexem 1.

Stejně jako je seznamem ("a", "b", "c"), může jím být i ($p1, $p2, $p3). Seznamem je dokonce i
($p). Takto zapsané hodnoty v kulatých závorkách jsou seznamem vždy. Poslední případ je sice
jednoprvkovým seznamem, ale stále seznamem. Pokud do pole přiřadíte prázdné závorky, pole bude dokonce
prázdné.

V další ukázce přiřazuji 1 seznam do jiného.

  ($a, $b) = ("a", "b");
  
  print $a; #a
  print $b; #b

Existuje řada variací tohoto příkladu. Pokračujme v kódu.

  ($a, $b) = ($b, $a); #prohození hodnot v proměnných
  
  print $a; #b
  print $b; #a
  
  @mix = ($a, 1, 2, $b, "Perl"); #b, 1, 2, a, Perl
  @mix2 = ($a, @mix, 4, "Linux") #b, b, 1, 2, a, Perl, 4, Linux

Jak vidíte, možnosti jsou obrovské.

Záporné indexy

Zatím víte, že index označuje pořadí prvku. Mohou být i záporné. V tom případě je hodnota pořadím od konce
pole. Pozor na to, že 1. prvek od konce se značí $pole[-1], nikoliv $pole[-0]. To proto, že $pole[-0]
je stejné jako $pole[0].

Příště budu pokračovat v polích a snad si na ně napíšeme i nějaký rozsáhlejší příklad.

Online verze článku: http://www.linuxsoft.cz/article.php?id_article=875

Perl (12) - Pole podruhé

Navázujeme na poslední díl. Ten dnešní se bude věnovat operacím Perlu, typickým pro práci se seznamy jako
je třídění polí, operátor rozsahu a také chování funkce print.

1.8.2005 06:00 | Jiří Václavík | přečteno 1747x 

Seznam ve skalárním kontextu

Asi se právě ptáte - co je to za nesmysl?! Nadpis je ale opravdu správně. Jak jsem psal v minulém díle, různé
funkce a operátory se chovají ke skalárům a seznamům jinak. Pracujete-li se seznamem jako se skalární
hodnotou (za chvíli vysvětlení), Perl si s tím šalamounsky poradí. Takovýto seznam ve skalárním kontextu se
tváří jako proměnná, jejíž hodnotou je počet prvků seznamu. Možná se vám to zdá trochu zamotané. Vyjasníme
si to na těchto řádcích kódu:

  @pole = qw(a b c d e f);
  $skalar = @pole;
  print $skalar;

Sledujme 2. řádek. Protože je levý operand skalár, musí do něj být uložena skalární hodnota. Pravý operand
tedy bude vyhodnocován také jako skalár, ať je čímkoliv. V našem případě se do proměnné $skalar přiřadí
hodnota 6, což je počet prvků @pole.

  $skalar = (10, 20, 30); # POZOR!!!
  print $skalar;

U seznamů, které nejsou poli, je to však jinak, než by se dalo podle dosavadních tvrzení čekat! Do $skalar je
přiřazena poslední hodnota seznamu, tedy 30. Zápis má stejný význam jako s operátorem čárky mezi
hodnotami prvků.

Perl má k tomuto účelu i funkci, která vyjadřuje seznam ve skalárním kontextu. Funkce scalar převádí
hodnotu v parametru na skalární. Je-li skalární, nic se neděje. Ale jde-li o pole, funkce vrátí počet jeho prvků.

  @pole = qw(a b c d e f);
  print scalar @pole; # vytisknuto je opět 6.

Teď si vezměme opačný případ. Přiřazení skalární proměnné do seznamu (ono je to ve skutečnosti stejné jako
bychom přiřazovali do seznamu přímo hodnotu...). Přiřazením vznikne jednoprvkové pole, jehož nultou
hodnotou je hodnota na pravé straně přiřazení. Skalární hodnota se konvertuje v seznam.

  $skalar = 666;
  @pole = $skalar; # index 0: hodnota 666

Funkce print s parametrem v seznamovém kontextu

Předáte-li funkci print pole, vytiskne hodnoty všech prvků pole.

  @pole = qw(a b c d e f);
  print @pole; # vytiskne abcdef.

Nemusíte předávat už vytvořené pole, ale třeba jen seznam hodnot, oddělených čárkami

  print "a", "b", "c"; # vytiskne abc.

Samozřejmě i v print můžete použít uzavření do uvozovek pomocí qw:

  print qw(a b c); # vytiskne abc.

Speciální proměnné pro pole

Perl obsahuje desítky speciálních proměnných, jež mají speciální význam. My si představíme 3 z nich, se
kterými se často setkáte při práci s poli.

Proměnná Význam Implicitně

$, řetězec, který bude tištěn mezi jednotlivými prvky seznamu (ne v řetězci) ""

$"
řetězec, který bude tištěn mezi jednotlivými prvky seznamu uvedeném v řetězci
označeným uvozovkami.

" "

$\ řetězec, který bude vytištěn na konci seznamu ""

Sledujte, jak se změní poslední příklad použitím proměnných $, a $\:

  $, = ", ";
  $" = "-"; #v tomto případě neovlivní výsledek
  $\ = "\n";
  
  print "a", "b", "c";

Vytiskne se řetězec "a, b, c\n". Mezi jednotlivými prvky pole je vždy čárka a mezera a za poslední
položkou je znak nového řádku.

$" působí na tisknuté pole, vložené do řetězce:

  $, = ", "; #v tomto případě neovlivní výsledek
  $" = "-"; #zkuste zakomentovat tento řádek, implicitním obsahem je mezera
  
  @p = ("a", "b", "c");
  print "@p";

Vytiskne se řetězec "a-b-c". Ale zkuste odstranit z příkazu print uvozovky. Pole by potom nebylo v
řetězci, místo $" by se mezi jednotlivé prvky tiskla hodnota v proměnné $, a výsledkem by bylo "a, b, 
c".

Poslední prvek pole

Zápis $#pole vrací poslední (nejvyšší) index pole. Je ekvivalentní zápisu $pole[-1].

Operátor rozsahu

Přiřaďme nyní do prvků s indexy 5-14 pole @pole2 hodnoty prvků s indexy 2-11 pole @pole:

  @pole2[5 .. 14] = @pole[2 .. 11];

Výraz 5 .. 14 má stejný význam jako 5, 6, 7, 8, 9, 10, 11, 12, 13, 14. Jde o řez polem.

Na základě této informace vytvořme pole s hodnotami 1 až 100:

  @pole = (1 .. 100);

Operátor rozsahu funguje i na malá a velká písmena anglické abecedy:

  @pole = ("c" .. "z"); # výsledkem je pole obsahující znaky c až z

Rozsah toho umí ještě o něco víc:

  $, = ", ";
  $\ = "\n";

  print(1.88 .. 2.21);                        #stejné jako 1 .. 2. Desetinná část se odřízne
  print("001" .. "010");                      #001, 002, 003 ... 010\n

  print("abc" .. "abx");                      #abc, abd, abe ... abx\n
  print("bc" .. "de");                        #bc, bd, be ... bz, ca, cb ... dd, de\n

  print(0 .. 9, "A" .. "F");                  #nic nového, pouze se tiskne seznam složený ze 2 seznamů

  print(0 .. 3, "a" .. "c", $promenna, "XXX");#prakticky totéž

  print(-10 .. 10);                           #záporná čísla nejsou překážkou
  print(10 .. 1);                             #ale pozpátku neumí

Setřídění pole - funkce sort

Řazení polí podle abecedy je v Perlu velmi jednoduché. Funkce sort řadí (implicitně) podle ASCII tabulky.
Další funkce - reverse - obrací pořadí prvků pole. Poslední prvek prohodí s prvním, předposlední s druhým
atd.

  @p = qw(a g d f e b c); #spřeházený seznam
  print @p;               #agdfebc
  @p = sort @p;           #setřídí podle abecedy

  print @p;               #abcdefg
  @p = reverse @p;        #obrátí pořadí prvků
  print @p;               #gfedcba

Tento systém má háček. Porovnává po znaku, zleva doprava, takže 6 je větší než 44. Na čísla je předchozí zápis
nepoužitelný.

Řazení čísel ale zvládne sort také. Slouží k tomu speciální konstrukce {$a <=> $b}. Porovnává vždy 2
prvky pole, určuje, který je větší a takto je řadí. Proměnné $a a $b jsou přednastaveny. Když prohodíte jejich
pořadí, setřídí se pole sestupně.

  @p = qw(6 1 3 5 4 2);     #přeházený seznam

  print @p;                 #613542
  @p2 = sort {$a <=> $b} @p;#setřídí podle velikosti vzestupně

  print @p2;                #123456
  @p3 = sort {$b <=> $a} @p;#setřídí podle velikosti sestupně

  print @p3;                #654321

Nastavení jazyka

Dalším problémem je čeština. Implicitně Perl třídí písmena s háčky až na konec a písmeno ch za c. To se dá
vyřešit pomocí locales. V systému nastavte hodnotu proměnné LANG na czech:

  $ export LANG=czech

Je dobré si tento řádek připsat do nějakého konfiguračního souboru, který se spouští při zapnutí systému.

Dalším krokem je přidat do samotného programu příkaz use locale; (vysvětlím ho později, až se budeme
zabývat moduly). Mějme kód:

  #!/usr/bin/perl
  use locale;
  
  $, = ", ";
  $\ = "\n";
  
  @p = qw(a d f č ď e c ch u); #přeházený seznam
  print sort @p;
  

Pokud máte správně nastavenou proměnnou prostředí LANG, měli byste spatřit výstup:

  $ ./sort.pl
  a, c, č, d, ď, e, f, ch, u
  $

Teď smažte řádek use locale;:

  $ ./sort.pl
  a, c, ch, d, e, f, u, č, š
  $

Funkce grep

grep vyhledává prvky v seznamu, ve kterých byl nalezen určitý podřetězec a ty vrací jako seznam. Existují 2
možnosti zápisu grepu:

  @vyhovujici_prvky = grep {test} seznam
  @vyhovujici_prvky = grep /regulární_výraz/, seznam

V 1. řádku grep postupně přiřazuje každý prvek seznamu do výchozí proměnné $_ (o té se podrobněji
zmíním v někerém brzkém díle), ověří, zda vyhovuje výrazu ve složených závorkách a pokud ano, prvek se
ocitne i ve výsledném poli.

  $, = ", ";
  @pole = (8, 2, 4, 6, 9, 5);
  print grep {$_ < 7} @pole; #vytiskne 2, 4, 6, 5 - tedy ty prvky, které jsou menší než 7

V 2. syntaktickém zápisu je namísto testovacího výrazu použit vzor. To je záležitost regulárních výrazů. grep
vrací seznam prvků z původního pole, které danému vzoru vyhovují

  $, = ", ";
  @pole = ("chomp", "print", "grep", "sin", "index");
  print grep /^.+in/, @pole;

Vypsáno bude "print, sin" - tedy řetězce, které obsahují podřetězec in, ale nezačínají jím. Regulární
výrazy budeme již brzy řešit v samostatných dílech.

Příklad - výčet potřebných platidel

Zadání: Napište program, který načte ze vstupu číslo, vyjadřující peněžitý obnos. Definujte pole, ve kterém
budete mít seznam hodnot českých platidel (haléřové mince zanedbejte - tedy v poli budou prvky pro 1 Kč, 2
Kč, 5 Kč, 10 Kč ... 2000 Kč, 5000 Kč). Obnos poté rozdělte na hodnoty jednotlivých platidel tak, aby byl
rozložen do co nejvyšších platidel. Přitom se snažte o obecnost - tedy například přibude-li platidlo, musí stačit
jeho hodnotu přiřadit do pole platidel. Ať pracuje program. Výsledky (kolik kterých platidel) tiskněte na
výstup. (A nezapomeňte vše ošetřit.)

Řešení

Začátek je jasný, definujeme pole platidel:

  #!/usr/bin/perl
  
  @platidla = qw(1 2 5 10 20 50 100 200 500 1000 2000 5000);

Načteme částku. Tady se na chvilku zastavíme. Musíme ošetřit hned 3 nestandartní případy. Uživatel vůbec
nemusí zadat číslo nebo může být záporné. Oba problémy vyřešíme tak, že otestujeme, zda je částka větší než 0
(poznámka: pokud tedy uživatel zadá "506xyz", bude částka zkonvertována na 506). Dalším problémem
jsou desetinná čísla. Je zadáno, že haléře máme zanedbat, takže funkcí int zadanou haléřovou částku
ořízneme.

  print "Zadej částku v Kč: ";
  $castka = int <STDIN>;

  if ($castka < 1){
      die "Toto není regulérní častka.\n";
  }

Zamysleme se nad tím, jak budeme postupovat dál. Určitě budeme mít nějaký cyklus. V každé jeho iteraci
budeme odečítat příslušný počet 1 druhu platidla. Protože mají být platidla maximálně vysoké, bude výhodné v
1. iteraci odečítat nejvyšší platidlo, v 2. iteraci 2. nejvyšší atd. Proto nejprve seřaďme pole sestupně:

  @platidla = sort {$b <=> $a} @platidla;

Zbývá najít nějaký vhodný test, který by rozhodoval o pokračování cyklu. Iterovat se bude tolikkrát, kolik
prvků má pole @platidla. Zaveďme ještě před cyklem proměnnou $pozice, kterou nastavíme na 0.
Každou iterací se inkrementuje. Až nabude hodnoty stejné jako počet prvků pole @platidla, cyklus se
zastaví. A protože předem známe počet iterací, použijeme cyklus for. Nic nám nebrání začít se samotným
cyklem.

  for($pozice=0; $pozice<@platidla; $pozice++){
  
      ... n krát odečtení hodnoty platidla ...
  
  }

Teď je třeba se zamyslet, jak to bude vypadat v cyklu. Zjistíme počet kusů daného platidla, které lze maximálně
odečíst, aby byl výsledek kladný. Není to až tak složité. Stačí od zbývající částky odečíst zbytek, který by zbyl
po dělení hodnotou aktuálního platidla - výsledkem bude nejvyšší částka, dělitelná hodnotou platidla, ale stále
nižší než zbývající částka. Tuto vzniklou částku vydělíme aktuální hodnotou platidla:

      $pocet_kusu = ($castka - ($castka % $platidla[$pozice])) / $platidla[$pozice];

Získali jsme počet kusů aktuálního platidla, který lze odečíst od zbývající částky. Tak to udělejme:

      $castka -= $pocet_kusu * $platidla[$pozice];

Poznámka: Možná jste si všimli, že pokud bychom tyto 2 řádky spojili do jednoho (za $pocet_kusu v

druhém dosadili hodnotu prvního), získali bychom zlomek, který obsahuje v čitateli i jmenovateli hodnotu

$platidla[$pozice], a tudíž jde zkrátit. $pocet_kusu ale ještě budeme potřebovat.

To je téměř celé. Zbývá zařídit nějaký výstup. Využijme cyklu a výsledky budeme vypisovat přímo v něm.
Ještě před cyklus ale vložme pro pořádek velikost původního obnosu:

  print "K vyplacení $castka Kč bude potřeba:\n";

A v cyklu bude:

      print "$pocet_kusu kusů platidla $platidla[$pozice].\n";

Jsme hotovi, nyní se pokochejte pohledem na chod právě vzniklého programu:

  $ ./vycet.pl
  Zadej částku v Kč: 19749
  K vyplacení 19749 Kč bude potřeba:
  3 kusů platidla 5000.
  2 kusů platidla 2000.
  0 kusů platidla 1000.
  1 kusů platidla 500.
  1 kusů platidla 200.
  0 kusů platidla 100.
  0 kusů platidla 50.
  2 kusů platidla 20.
  0 kusů platidla 10.
  1 kusů platidla 5.
  2 kusů platidla 2.
  0 kusů platidla 1.
  $

Můžete si to přepočítat, skutečně to platí. Zkusme do pole @platidla přidat ještě bankovku v hodnotě

10000. Mělo by to teoreticky také fungovat.

  $ ./vycet.pl
  Zadej částku v Kč: 19749
  K vyplacení 19749 Kč bude potřeba:
  1 kusů platidla 10000.
  1 kusů platidla 5000.
  2 kusů platidla 2000.
  0 kusů platidla 1000.
  1 kusů platidla 500.
  1 kusů platidla 200.
  0 kusů platidla 100.
  0 kusů platidla 50.
  2 kusů platidla 20.
  0 kusů platidla 10.
  1 kusů platidla 5.
  2 kusů platidla 2.
  0 kusů platidla 1.
  $

Možná jste si také uvědomili, že není jednoduché se ve výstupu zorientovat okamžitě. Je tam moc řádků. Ještě
program trochu pozměníme. Není-li od nějakého platidla ani kus, nebudeme řádek výbec vypisovat. Použijme
konstrukci podmínka za příkazem. Vykonání příkazu print v cyklu podmíníme nenulovým počtem kusů
daného platidla:

      print "$pocet_kusu kusů platidla $platidla[$pozice].\n" if $pocet_kusu;

Ještě odstraníme platidlo 10000 a program opět vyzkoušíme:

  $ ./vycet.pl
  Zadej částku v Kč: 19749
  K vyplacení 19749 Kč bude potřeba:
  3 kusů platidla 5000.
  2 kusů platidla 2000.
  1 kusů platidla 500.
  1 kusů platidla 200.
  2 kusů platidla 20.
  1 kusů platidla 5.
  2 kusů platidla 2.
  $

Je to hned o něco lepší. Celý zdrojový kód si můžete stáhnout. K dokonalosti chybí zformátovat výstup.
Formátu někdy v budoucnu budu věnovat celý díl.

A přece ještě něco - jako bonus si zkuste program upravit tak, aby správně česky skloňoval slovo kus:)

Zdroje

man perlvar - předdefinované proměnné
man perlfunc - spousta dalších informací o funkcích print, grep a hlavně sort
man perllocale - více o locales
man perlop, oddíl Range Operators - operátory rozsahu

Online verze článku: http://www.linuxsoft.cz/article.php?id_article=891

Perl (13) - Hashe

I dnes pokračujeme v seznamech. Přesněji asociativními poli, což je další typ seznamové proměnné.

12.9.2005 06:00 | Jiří Václavík | přečteno 1523x 

Při práci s poli jsme byli dosud omezeni na číselné indexy. Lze použít i řetězce. V tom případě nejde o klasické
pole, ale o asociativní pole (hash). Každý prvek obsahuje 2 skalární hodnoty. Můžeme říct, že jde o seznam
uspořádaných dvojic.

S hashi se pracuje podobně jako s poli. Na 1. pohled jsou zřejmé 2 odlišnosti. Hash se označuje procentem a
klíč se píše do složených (nikoliv hranatých) závorek. Stejně jako u polí platí, že podle klíče prvku lze
jednoznačně zjistit hodnotu, ale nikoliv naopak.

Hash plníme hodnotami takto:

  %hash = (klic1, hodnota1, klic2, hodnota2, ..., klicn, hodnotan);

Konkrétně:

  %vzdalenosti = ("Amsterodam", 970, "Moskva", 1900, "Kodaň", 750, "Řím", 1300, "Varšava", 
630);

Na uvození parametrů můžeme samozřejmě použít i nám již dobře známou funkci qw. Následující zápis je
ekvivalentní poslednímu uvedenému příkazu:

  %vzdalenosti = qw(Amsterodam 970 Moskva 1900 Kodaň 750 Řím 1300 Varšava 630);

Další synonymní zápis získáme nahrazením každého lichého operátoru čárky operátorem =>. Operátory , a =>
mají stejný význam, takže teoreticky by měly jít nahradit i sudé čárky. K tomu ale není důvod. => se užívá pro
lepší přehlednost. (V našem případě je sice jasné, co je klíčem a co hodnotou. Ale představte si, že by obojí
byly na 1. pohled od sebe nerozeznatelné řetězce.)

  %vzdalenosti = ("Amsterodam" => 970, "Moskva" => 1900, "Kodaň" => 750, "Řím" => 1300, 
"Varšava" => 630);

A nakonec úplně nejpřehlednější zápis získáte rozepsáním kódu do více řádků. Každý pár klíč => 
hodnota bude na jednom.

  %vzdalenosti = (

    "Amsterodam" =>  970,
    "Moskva"     => 1900,
    "Kodaň"      =>  750,

    "Řím"        => 1300,
    "Varšava"    =>  630
  );

Každý z předcházejících kódů vytvořil hash s těmito prvky:

Klíč Hodnota

Amsterodam 970

Moskva 1900

Kodaň 750

Řím 1300

Varšava 630

Jednotlivé hodnoty získáme obdobně jako u polí. Jak daleko je to z Prahy do Moskvy?

  print $vzdalenosti{"Moskva"};

Vidíme, že je to stejné jako s poli, jen se, jak už bylo zmíněno, klíč píše do složených závorek.

Konverze pole - hash

Přiřadíte-li pole do hashe, je to jako byste přiřazovali seznam (tedy hodnoty prvků pole).

  @pole = qw(1 2 3 4 5 6);
  %hash = @pole;

Zápis má (vzhledem k obsahu hashe %hash) stejný význam jako tento:

  %hash = qw(1 2 3 4 5 6);

a proto %hash bude obsahovat prvky:

Klíč Hodnota

1 2

3 4

5 6

Konverze hash - pole

Konverze opačným směrem je samozřejmě také možná. Hash je rozložen na hodnoty klic1, hodnota1, klic2, 
hodnota2... a ty jsou přiřazeny do pole:

  @pole = %vzdalenosti;

@pole potom obsahuje prvky:

Klíč Hodnota

0 Varšava

1 630

2 Amsterodam

3 970

4 Kodaň

5 750

6 Moskva

7 1900

8 Řím

9 1300

Chování funkce print

Uvedete-li jako argument hash, nejdříve se zkonvertuje na pole a až vzniklé pole se tiskne.

  $, = ", ";
  print %vzdalenosti;

Výstupem tedy bude text

  Varšava, 630, Amsterodam, 970, Kodaň, 750, Moskva, 1900, Řím, 1300

Funkce exists

Testuje, zda existuje klíč pole. Přičemž nezáleží na tom, zda má definovanou hodnotu. Funkce exists se
často používá v podmínkách.

  print exists($vzdalenosti{"Amsterodam"});#1
  print exists($vzdalenosti{"Tokyo"});     #""

Funkce delete

Maže klíč a vrací jeho hodnotu.

  print exists($vzdalenosti{"Řím"});#1
  print delete($vzdalenosti{"Řím"});#130
  print exists($vzdalenosti{"Řím"});#""

Funkce each

  while (($mesto, $vzdalenost) = each(%vzdalenosti)){
      print "$mesto - $vzdalenost\n";
  }

Příklad vypíše všechny klíče a hodnoty pole. Funkce each vrací klíč a hodnotu prvku. Přitom si pamatuje,
které klíče prvky již vrátila a při opakovaném volání v cyklu vrací jiný prvek.

  $ perl each.pl
  Varšava - 630
  Amsterodam - 970
  Kodaň - 750
  Moskva - 1900
  Řím - 1300
  $

Funkce keys

keys vrací pole prvků s hodnotami, které jsou klíči hashe.

  print keys %vzdalenosti;#tiskne klíče všech prvků hashe %vzdalenosti

Vytiskněme ještě klíče v abecedním pořádku:

  print sort keys %vzdalenosti;

Funkce values

Vrací hodnoty všech prvků hashe.

  print values %vzdalenosti;#tiskne hodnoty všech prvků hashe %vzdalenosti

Počet prvků hashe

Zjistíte ho sečtením klíčů hashe:

  print scalar keys %vzdalenosti;

Píšeme si slovník

Dnešní díl zakončíme vytvořením programu, který načte ze vstupu slovo a pokud ho má v databázi pojmů,
vytiskne jeho význam. Zatím budeme pojmy definovat přímo v programu, což není dobré řešení. Někdy později
si předvedeme, jak k tomuto účelu použít externí soubor nebo, ještě lépe, databázi.

Řešení

Ze všeho nejdříve bude nutné definovat pár pojmů.

  #!/usr/bin/perl 
  use strict;
  
  my $slovo;#bude obsahovat hledaný pojem
  my %pojmy = (

      "perl" => "programovací jazyk",
      "ankara" => "turecké hlavní město",
      "uran" => "chemický prvek",
      "Uran" => "planeta",
      "klaus" => "2. prezident České republiky",
      "dioda" => "elektrotechnická součástka"
  );

Dále načteme pojem:

  print "Zadej pojem: ";
  chomp($slovo = <STDIN>);
  

Hash necháme prozkoumat cyklem foreach (probereme ho sice až v příštím díle, ale vězte, že do zadané
proměnné přiřazuje postupně každý prvek zadaného pole) a pokud se hledané slovo shoduje s názvem pojmu
vypíšeme informace a ukončíme. Přitom převedeme všechna písmena na malá, aby nezáleželo na velikosti
písmen načteného řetězce. Z tohoto důvodu dále musíme zajistit, aby se tiskly všechny pojmy i v případě, že
existuje více pojmů, lišících se jen ve velikosti (tedy konkrétněji - pokud bude dotaz znít URAN, ve výsledcích
se musí objevit Uran i uran).

Poznámka - právě jste si možná uvědomili, že nebudeme schopni definovat 2 stejné pojmy, které se neliší ani ve

velikosti (kupříkladu Uran - planeta a Uran - bůh). Pouze můžeme do 1 hodnoty prvku vepsat oba významy.

  foreach my $pojem (keys %pojmy){
      if (lc $pojem eq lc $slovo){
          print "POJEM $pojem --- $pojmy{$pojem}\n";
      }
  }

To je téměř vše. Zbývá nám vytisknout nějakou hlášku v případě, že ve slovníku pojem odpovídající zadanému
není. K tomu si definujeme další proměnnou $uspech,

  my $uspech = 0;

ve které bude pravdivá hodnota, pokud byl hledaný pojem nalezen. V opačném případě zůstane hodnota
nepravdivá. Pokud byl pojem nalezen, přiřadíme do ní pravdivou hodnotu. To zapíšeme do těla podmínky
uvnitř cyklu.

          $uspech = 1;

A v případě neúspěchu tiskneme chybové hlášení.

  print "Zadaný pojem není ve slovníku :(\n" unless $uspech;

Zkusme spustit náš program (zdrojový kód):

  $ ./slovnik.pl
  Zadej pojem: uran
  POJEM Uran --- planeta
  POJEM uran --- chemický prvek
  $ ./slovnik.pl
  Zadej pojem: zinek
  Zadaný pojem není ve slovníku :(
  $

Až budeme znát regulární výrazy, budeme umět napsat slovník, který hledá i podle části názvu. Další věcí,
která by našemu slovníku slušela, by bylo předání hledaného pojmu jako argumentu příkazové řádky.

Ve 14. díle konečně dokončíme blok dílů o seznamových proměnných.

Online verze článku: http://www.linuxsoft.cz/article.php?id_article=911

Perl (14) - Funkce pro seznamy

Cyklus foreach, několik nových funkcí a vstup v souvislosti se seznamy - to jsou témata 14. dílu.

29.9.2005 06:00 | Jiří Václavík | přečteno 1277x 

Cyklus foreach

foreach je další do sbírky cyklů. Používá se v souvislosti s poli, protože umožňuje jejich snadné procházení.
Název se skládá ze slov for a each, což už mnoho napovídá o jeho významu.

  @ptaci = ("kos", "vrabec", "papoušek", "datel");
  
  foreach $klic (@ptaci){
      print $klic, "\n";
  }

Do proměnné $klic se každou iteraci cyklu přiřadí jedna hodnota pole. Tedy postupně hodnoty
$ptaci[0], $ptaci[1], $ptaci[2], $ptaci[3]. Toho užijeme pro výpis dat z pole:

  $ perl foreach.pl
  kos
  vrabec
  papoušek
  datel
  $

Chcete-li vypisovat hashe, bude se hodit funkce keys, která vrací pole klíčů hashe. Pomocí klíčů totiž lze
získat i hodnoty.

  %ptaci = (
      "kos"      => "slunečnice",
      "vrabec"   => "slunečnice",

      "papoušek" => "proso",
      "datel"    => "hmyz"

  );
  
  foreach $klic (keys %ptaci){
      print "$ptaci{$klic} je potrava pro $klic\n";

  }

Doufám, že jsem nějakého opeřence výběrem jídla neurazil. Každopádně odtud vidíte, že to skutečně funguje:

  $ perl foreach.pl
  slunečnice je potrava pro kos
  slunečnice je potrava pro vrabec
  proso je potrava pro papoušek
  hmyz je potrava pro datel
  $

Cyklus by měl fungovat stejně i v případě, že přepíšete foreach na for. Perl sám pozná, o který cyklus jde.

foreach (tedy i for) lze napsat i za příkaz

  print "$_, " foreach (1, 2, 3, 4, 5);
  print "$_, " for (1, 2, 3, 4, 5);

případně za blok.

  @p = (1, 2, 3, 4, 5);
  do {$_ += 10;
      print "$_, ";
  } for @p;

Proměnná $_ zatím v tomto seriálu nebyla blíže rozebírána. Bude tak učiněno v příštím díle. Jde o implicitní
proměnnou, do které se hodnoty pole postupně přiřazují.

chomp v seznamovém kontextu

Uvedete-li jako parametr pole, je to jako byste použili funkci chomp na všechny skaláry, které pole obsahuje.
Je-li parametrem hash, aplikuje se chomp pouze na hodnoty - tedy klíče zůstanou nezměněny.

Funkce pop a shift

Jak z názvu vyplývá, pop odstraňuje prvek pole z konce seznamu (ten s nejvyšším indexem) a shift ze
začátku. Obě funkce rovněž odstraněnou hodnotu vracejí.

  @p = 1 .. 5;
  
  print pop @p;#5
  print @p;#1234
  
  shift @p;
  print @p;#234

Funkce push a unshift

push přidává prvek na konec seznamu, unshift na začátek. Obě funkce vrací jako návratovou hodnotou
hodnotu přidaného prvku.

  @p = 1 .. 5;
  
  push @p, 6;
  print @p;#123456
  
  unshift @p, 7;
  print @p;#7123456

Funkce splice

Odstraňuje, přidává a nahrazuje prvky v poli. Má 4 parametry.

název proměnné typu pole, na kterou se bude příkaz vztahovat
pozice, za kterou splice dojde k manipulaci s polem
počet znaků, se kterými bude manipulováno. Nezadáteli ho, je koncová pozice automaticky nastavena na
konec řetězce
seznam prvků, které budou přidány

Mějme pole čísel:

  $, = ", ";
  @p = 1 .. 20;

Odstraňme prvky 8 - 11:

  @odstranene = splice(@p, 7, 4);

Jak vidíte, vrací funkce jako návratovou hodnotu pole odstraněných prvků:

  print @odstranene;#8, 9, 10, 11

Samotné pole @p obsahuje:

  print @p;#1, 2, 3, 4, 5, 6, 7, 12, 13, 14, 15, 16, 17, 18, 19, 20

Odstraňme všechny prvky od prvku s hodnotou 13:

  splice(@p, 8);

  print @p;#1, 2, 3, 4, 5, 6, 7, 12

Prvky s indexy 1 a 2 nahradíme jinými. Lze přidat jiný počet než prvků, než kolik je odebráno. Též je možné
zadat jako 3. parametr 0 - v tom případě se budou prvky jen přidávat.

  splice(@p, 1, 1, 2000, 2002, 2004, 2006);
  print @p;#1, 2000, 2002, 2004, 2006, 3, 4, 5, 6, 7, 12

A nakonec smažme všechny prvky:

  splice @p;
  print @p;#""

Pole a vstup

Když jsme načítali vstup ve skalárním kontextu, vždy jsme získali jen 1 řádek ze zdroje dat. Přiřazujete-li zdroj
dat do pole, vytvoří se tolik prvků, kolik mají data řádků. Hodnotami prvků jsou právě řádky.

Modifikujme náš příklad s českými králi ze 7. dílu tak, aby se vypsali všichni, kteří jsou v souboru uloženi.
Připomeňme si datový soubor kralove:

Přemysl Otakar I. 1197-1230
Václav I. 1230-1253
Přemysl Otakar II. 1253-1278

Jen zaměníme skalární proměnnou za pole:

  #!/usr/bin/perl
  use strict;
  
  my @radky = <>;
  print @radky;

A dostaneme všechny 3 řádky.

  $./kralove.pl kralove
  Přemysl Otakar I. 1197-1230
  Václav I. 1230-1253
  Přemysl Otakar II. 1253-1278
  $

Funkce split

split rozdělí řetězec na několik podřetězců a ty uloží do seznamu. Funkce hledá v řetězci výskyty
oddělovače (1. parametr). Oddělovačem je normální řetězec. Výskyty oddělovačů určují právě začátek a konec
výsledných podřetězců. Oddělovač se ale v podřetězcích neobjevuje. 2. parametrem je samotný řetězec. Lze
uvést i 3. parametr, který určuje, kolik nejvíce může podřetězců být. Po vyhledání daného počtu oddělovačů je
posledním podřetězcem celý text za posledním výskytem, přičemž nezáleží, zda tam nějaký výskyt ještě není.

  $veta = "Perl letos slaví 18. narozeniny!";
  
  ($slovo1, $slovo2, $slovo3) = split(" ", $veta);
  print $slovo1;#Perl
  print $slovo2;#letos
  print $slovo3;#slaví

Oddělovačem byla mezera - tímto způsobem získáme (ve většině případů) jednotlivá slova. Ve výsledných
podřetězcích již mezery nebudou.

  @vsechna_slova = split(" ", $veta, 4);

Na posledním řádku byl text rozdělen na 4 podřetězce a ty jsou následně přiřazeny do pole. Každý podřetězec
bude hodnotou prvku. Pokračujme ale v kódu:

  foreach $slovo (@vsechna_slova){
      print "$slovo, ";
  }

Každou iteraci bylo vytisknuto 1 slovo věty a za ním čárka. Při 5. iteraci se vytiskl zbytek textu.

Ještě jednou dnes dostanou slovo králové. Uděláme z řádků věty tak, že před období vlády přidáme text.
Předpokládejme, že období je vždy za poslední mezerou v řádku.

  while ($radek = <>){
      chomp $radek;
      @p = split(" ", $radek);
      $obdobi = pop @p;
      print "@p vládl v letech $obdobi.\n";
  }

Sám algoritmus je velmi jednoduchý. rozdělíme řádek mezerami, poslední prvek (období) přesuneme do jiné
proměnné a můžeme tisknout.

$ perl kralove.pl kralove
Přemysl Otakar I. vládl v letech 1197-1230.
Václav I. vládl v letech 1230-1253.
Přemysl Otakar II. vládl v letech 1253-1278.
$

Funkce join

Zatímco split řetězce rozděluje, join je naopak spojuje. Jeho použití velmi podobné jako u split. 1.
parametrem je spojovač, který bude vložen do výsledného řetězce mezi každé 2 hodnoty seznamu.

  $rada[0] = 0;
  for ($i=1; $i<10; $i++){
      $rada[$i] = $i + $rada[$i-1];
  }
  
  print join(" ", @rada); #0 1 3 6 10 15 21 28 36 45
  print join(", ", @rada); #0, 1, 3, 6, 10, 15, 21, 28, 36, 45
  print join(" *** ", @rada); #0 *** 1 *** 3 *** 6 *** 10 *** 15 *** 21 *** 28 *** 36 *** 45
  print join("", @rada); #0136101521283645

Tím jsme dokončili blok o seznamových proměnných. Příští díl bude směsí různých témat. To proto, abychom
byli připraveni na miniseriál o regulárních výrazech, který bude následovat.

Zdroje

man perlfunc - popis dnes zmíněných funkcí

Online verze článku: http://www.linuxsoft.cz/article.php?id_article=930

Perl (15) - Co se jinam nevešlo

15. díl nemá jednotné téma. Půjde vesměs o věci, které jsou na celý díl krátké, dosud se nikam výrazněji
nehodily a přesto je užitečné něco o nich tušit.

11.10.2005 07:00 | Jiří Václavík | přečteno 1284x 

Ještě dříve, než se pustíme do tajemných hlubin regulárních výrazů, bychom se měli seznámit s několika
tématy. Proto vznikl tento díl.

Výchozí proměnná $_



LINUX SOFTWARE file:///C:/Ikon/Books/Perl/LinuxSoft_PERL.html

58 of 102 3/8/2006 20:54

$_ je proměnná, kam se implicitně ukládají skalární hodnoty. Typicky - nezadáme-li název proměnné někde,
kde je vyžadován, použije se právě $_. Význam je pak stejný, jako bychom použili název $_. Toto pravidlo
neplatí všude, ale podporují ho pouze některé konstrukce nebo funkce.

  while ($radek = <>){
      if ($radek == 0){
          exit;
      }
      print $radek;
  }

Tento kód můžeme přepsat pomocí $_:

  while (<>){
      if ($_ == 0){
          exit;
      }
      print $_;
  }

Uvedete-li jako test jen <>, bude vstup implicitně uložen do proměnné $_ (tedy stejné jako $_ = <>).
Příklad můžeme ještě zkrátit. Pokud není funkci print předán žádný parametr, automaticky tiskne hodnota
$_:

  while (<>){
      if ($_ == 0){
          exit;
      }
      print;
  }

Takovým způsobem se nechová jen print, ale řada dalších funkcí. To, jestli funkce pracuje s výchozí
proměnnou zjistíte v manuálové stránce perlfunc.

Velmi často se výchozí proměnná používá u cyklu foreach. Pokud neuvedete v hlavičce cyklu skalární
proměnnou, použije se právě $_. Poměrně snadno můžeme psát i takové programy jako je výpis třetích mocnin
čísel od 1 do 10:

  print $_**3, "\n" for (1..10);

Proměnná s názvem určeným vyhodnocením výrazu

Za proměnnou v řetězci musí být vždy mezera. 1 z možností, jak toto "pravidlo" obejít, je uzavření názvu
proměnné do složených závorek.

  $slovo = "TEXT";
  print "zhusteny${slovo}bezMEZER.";

Složené závorky umí ještě obecnější věc. Název proměnné lze získat vyhodnocením výrazu ve složených
závorkách.

  $p = 1;
  ${"p" x 3} = 2; #"p" x 3 je vyhodnoceno jako ppp, což je název proměnné
  ${if($ppp){$p?"a":"b"}else{$p?"c":"d"}} = 3;
  
  print $ppp;#2
  print $a;  #3
  print $b;  #nedefinováno
  print $c;  #nedefinováno
  print $d;  #nedefinováno

Jak vidíte, můžete zakomponovat i celé konstrukce jako je if. Jedinou podmínkou je, aby byl úsek kódu ve
složených závorkách vyhodnocen jako řetězec, který je platným identifikátorem. Předchozí výraz je
vyhodnocen jako "a". Proto bude hodnota přiřazena do proměnné $a.



LINUX SOFTWARE file:///C:/Ikon/Books/Perl/LinuxSoft_PERL.html

59 of 102 3/8/2006 20:54

Nicméně tuto vlastnost uvádím spíše pro úplnost než z praktického hlediska. Mohou s ní být problémy (nejen
ve strict režimu) a v praxi ji moc neužijete.

Funkce undef

Funkce undef vrací nedefinovanou hodnotu. Takže, například, potřebujeme-li vytvořit pole, v němž budou
mít indexy 1, 3, 4 nedefinované hodnoty, můžeme udělat toto:

  @p = ("a", undef, "b", undef, undef, "c");

Speciální syntaxe zápisu řetězce

Syntaxe here-document

Potřebujete-li vytisknout nějaký delší text, jistě přivítáte tzv. here-document možnost zápisu.

  $zaznamu = 91;
  print << 'KONEC';
  Vyberte z těchto voleb:
  +----------------------+
  | 1 ... Přidat záznam  |
  | 2 ... Odebrat záznam |
  | 3 ... Konec          |
  +----------------------+
  Záznamů: $zaznamu\n
  KONEC

Tisknuto je vše, dokud není nalezeno slovo KONEC na samostatném řádku (avšak který nesmí být zároveň
posledním v souboru). Můžete zvolit libovolné slovo. Je to stejné jako v shellech (jistě si vybavíte příkazy jako
cat > soubor <<EOF).

Protože se k uvození nepoužívají ani apostrofy ani uvozovky, lze je normálně používat v tisknutém řetězci.

  $ perl print.pl
  Vyberte z těchto voleb:
  +----------------------+
  | 1 ... Přidat záznam  |
  | 2 ... Odebrat záznam |
  | 3 ... Konec          |
  +----------------------+
  Záznamů: $zaznamu\n
  $

Z předchozího výpisu je zřejmé, že nebyly nahrazeny proměnné a escape znaky. To je samozřejmě možné
ovlivnit. Nyní je nahradíme jejich obsahem. Změňme apostrofy kolem slova KONEC na uvozovky:

  $zaznamu = 91;
  print << "KONEC";
  Vyberte z těchto voleb:
  +----------------------+
  | 1 ... Přidat záznam  |
  | 2 ... Odebrat záznam |
  | 3 ... Konec          |
  +----------------------+
  Záznamů: $zaznamu\n
  KONEC

Potom se text chová jako řetězec v uvozovkách. Místo řetězce "$zaznamu" se oproti minulému příkladu
vytiskla hodnota této proměnné.

  $ perl print.pl
  Vyberte z těchto voleb:
  +----------------------+
  | 1 ... Přidat záznam  |
  | 2 ... Odebrat záznam |
  | 3 ... Konec          |
  +----------------------+

  Záznamů: 91

  $

Tato syntaxe není svázaná s příkazem print, jak by se doposud mohlo zdát. Lze ji použít prakticky kdekoliv
místo řetězce. Tedy například při přiřazení do proměnné.

  $x = << 'KONEC';
  +----------------------+
  | 1 ... Přidat záznam  |
  | 2 ... Odebrat záznam |
  | 3 ... Konec          |
  +----------------------+
  KONEC

Jistou nevýhodou syntaxe here-document v Perlu je, že nelze vložený řetězec odsazovat. Vzniká tím poměrně
nečitelný kód. Pokud se o odsazení pokusíte, budou mezery, kterými odsadíte, součástí řetězce. Nicméně v
Perlu 6 bude tato věc řešena.

Pokud trváte na odsazení, existuje částečné řešení už nyní. Pomocí regulárních výrazů lze smazat všechna bílá
místa na začátku řádku.

  ($x = << 'KONEC') =~ s/^\s+//gm;
      radek 1
      radek 2
  KONEC

Předchozí kód je také ukázkou dalšího způsobu použití této syntaxe.

To není o here-document zdaleka vše. Dává řadu jiných možností. Koho to zajímá, tak si může najít další
informace v manuálové stránce perlop v části Regexp Quote-Like Operators.

Syntaxe q a qq

Další možností zápisu řetězce je použít uvození pomocí q a qq.

Zápis pomocí q napodobuje řetězec v apostrofech. Liší se tím, že apostrofy v řetězci uvozeném pomocí q
můžete normálně používat. Nahrazuje se zpětné lomítko a uvozovací znak - v našem případě lomítko.

  $x = 99;
  print q/řetězec jako v apostrofech: ' \n " $x \\ \//;

Vytiskne:

  $ perl q.pl
  řetězec jako v apostrofech: ' \n " $x \ /
  $

Podobně můžeme simulovat uvozovky. Stačí zaměnit q za qq:

  $x = 99;
  print qq/řetězec jako v uvozovkách: ' \n " $x \\ \//;

Escape znaky i proměnné se nyní normálně nahrazují hodnotami.

  $ perl q.pl
  řetězec jako v uvozovkách: '
   " 99 \ /
  $

Není nutné uvozovat lomítkem. Použijte cokoliv, co se, pokud možno, v textu minimálně objevuje. Uvozovací
znak v textu předřadíte zpětným lomítkem. Použijete-li některou ze závorek, je počátečním znakem otevření
koncovým uzavření (to platí pro (), {}, [] a <>). Používáte-li křížek jako uvozovací znak, nesmí před ním být
bílé místo. V tom případě bude totiž brán jako komentář.

  print q{řetězec jako v apostrofech: ' \n " $x \\ \/};

  print q#řetězec jako v apostrofech: ' \n " $x \\ \/#;
  print q #komentář#;

To samé lze dělat s qq.

  print qq"řetězec jako v uvozovkách: ' \n \" $x \\ \/";

Nyní nám nic nebrání, abychom se pustili do regulárních výrazů.

Zdroje

man perlvar - proměnná $_ - kde se používá atd.
man perlop, část Regexp Quote-Like Operators - podrobnější informace o uvozování
řetězců

Online verze článku: http://www.linuxsoft.cz/article.php?id_article=938

Perl (16) - Regulární výrazy - začínáme

Dneškem začíná v rámci seriálu Perl miniseriál o nesmírně mocném nástroji - regulárních výrazech.

19.10.2005 07:00 | Jiří Václavík | přečteno 1547x 

Regulární výrazy (regular expressions, někdy zkráceně označované jako regexp, regex nebo RE) pocházejí z
unixových nástrojů pro editaci a vyhledávání v řetězcích - tedy sed, vi, grep, awk a další (více o historii
například na Wikipedii). Nejsou mezi sebou ale úplně kompatibilní. To, co funguje awku, nemusí fungovat v
grepu nebo může mít jinou syntaxi apod. Mezi nástroje, které umí regulární výrazy se samozřejmě řadí i Perl.
Jen těžko byste hledali jiný jazyk, který by podporoval regulární výrazy tak komplexně. Tímto spojením
kvalitního skriptovacího jazyka s regulárními výrazy získáváme do svého arzenálu nesmírně silný nástroj. To je
jeden z hlavních důvodů, proč se Perl těší oblibě, kterou má. Proto ani já nebudu regulárními výrazy šetřit a
pokusím se je přiblížit opravdu podrobně.

Regulární výraz není nic jiného než speciální řetězec, který je šablonou, vystihující určitou množinu textových
řetězců. Každý textový řetězec takové šabloně buď vyhovuje nebo ne. S touto množinou vyhovujících řetězců
můžeme manipulovat - hledat její prvky v textu nebo je nahrazovat jiným řetězcem.

Pokud si vzpomínáte, tak jsme na regulární výrazy již natrefili při popisu funkce grep. Tehdy jsem ale jen
uvedl příklady a dál je nerozebíral.

Regulární výrazy verzus žolíkové znaky

Regulární výrazy se často zaměňují s žolíkovými znaky. Je zde ale poměrně jasný rozdíl. Obecně - žolíkový
znak může vyhovovat jednomu nebo více znakům, oproti tomu regulární výraz je řetězec s jasně určenými
podmínkami a žádný samotný znak zde nezastupuje více znaků. Například * v žolíkových znacích znamená
libovolný řetězec, ale v regulárních výrazech libovolný počet opakování. Žolíkové znaky se používají téměř
výhradně k hledání v názvech souborů.

Regulární výrazy v Perlu

Základní syntaxe

Začneme příkladem. Následující zápis vrací 1, pokud se v řetězci vyskytuje podřetězec "vzor", v opačném
případě prázdný řetězec.

  "řetězec" =~ m/vzor/

Regulárnímu výrazu m/vzor/ vyhoví takové řetězce jako "vzory", "re vzor" nebo "vzor".

Operátor =~ slouží k porovnávání vzorů. S přiřazováním má společný opravdu jen ten znak rovnítka, jinak jde
o odlišné operace. "Písmeno" m není povinné, navíc můžete stejně jako u nám už známých q a qq lomítka
nahradit jiným znakem. Negovaný operátor =~ se zapisuje jako !~.

  print "řetězec" =~ m"vzor";
  print "řetězec" =~ m!vzor!;
  print "řetězec" =~ /vzor/; #nejběžnější užití
  print "řetězec" !~ /vzor/;

Existuje několik speciálních ohraničení. Použijí-li se apostrofy, neprobíhá vkládání obsahu proměnných. Nelze
tedy psát:

  $vzor = "[abc]{2}";
  "aa" =~ m'$vzor'; #$vzor není nahrazen obsahem proměnné, ale je brán jako řetězec '$vzor'

Další možností je ohraničení otazníky (potom uvedení m není povinné). V takovém případě dojde k použití
vzoru pouze jednou (i v testu cyklu s přepínačem g). Teprve až je zavolána funkce reset, může se porovnání
opakovat. Tato konstrukce se příliš nedoporučuje.

Porovnávání řetězců se užívá jako test při rozhodování.

  $veta = "Existuje 10 druhů lidí - ti, kteří znají binární kód, a ti, kteří ne.";
  
  if ($veta =~ m/10 druhů/){
      print "'10 druhů' je ve větě!\n"
  }
  
  if ($veta =~ m/777/){
      print "'777' je ve větě!\n"
  }

"10 druhů" se v řetězci vyskytuje, proto byl v tomto případě výraz vyhodnocen jako true. Nezáleží na tom
na jaké pozici nebo co je okolo.

Použití výchozí proměnné

Je také možné uvést jako test podmínky jen čistě regulární výraz v uvozovacích znacích. Implicitně je
porovnáván s výchozí proměnnou. Příklad má stejný účinek jako předchozí, ale využívá této vlastnosti:

  $_ = "Existuje 10 druhů lidí - ti, kteří znají binární kód, a ti, kteří ne.";
  
  if (/existuje/){
      print "'existuje' je ve větě!\n"
  }
  
  if (/Existuje/){
      print "'Existuje' je ve větě!\n"
  }

Poslední ukázka tiskne řetězec "'Existuje' je ve větě!". Na velikosti písmen tedy implicitně záleží.
Způsob, kterým lze toto chování změnit, si představíme v 1 z příštích dílů.

V případě zkráceného zápisu pomocí výchozí proměnné je též možné negovat. Stačí přidat vykřičník před
úvodní /.

  if (!/2 druhy/){
      print "'2 druhy' není ve větě!\n"
  }

Metaznaky

Mezi lomítka lze napsat všechny znaky znaky kromě takzvaných metaznaků. Mají totiž nějakou jinou funkci.
Patří mezi ně *, +, ., ?, ^, $, (, ), {, }, [, ], |, /. Lze je nahradit předřazením zpětného lomítka. Pokud si

nejste jisti, zda má znak speciální význam a není alfanumerický (tedy pokud je z množiny \W), můžete mu
předřadit zpětné lomítko vždy. Sekvence zpětného lomítka a nealfanumerického znaku se chová vždy jako znak
bez speciálního významu.

Proměnné v regulárních výrazech

Než se začne regulární výraz vyhodnocovat, jsou nahrazeny proměnné ve vzoru svým obsahem (což ale
samozřejmě neplatí u ohraničení pomocí apostrofů).

  $cisla = "2005";
  print "0002005000" =~ /$cisla/;

2. řádek předchozího kódu bude fungovat takto:

  print "0002005000" =~ /2005/;

Návratová hodnota výrazu řetězec =~ vzor

Tuto část můžete zatím klidně přeskočit, protože je zde řada věcí, které jsme zatím neprobírali. Je to ale téma,
které do 1. dílu o regulárních výrazech bezesporu patří.

skalární kontext

Ve skalárním kontextu vrací v případě úspěchu 1, v případě neúspěchu nepravdivou hodnotu.

seznamový kontext

O něco složitější je to v seznamovém kontextu. Při neúspěchu bez přepínače g vrací prázdný seznam. V případě
úspěchu vrací seznam zapamatovaných hodnot, pokud jsou, jinak vrací seznam (1). S přepínačem g vrací
výraz pole vyhovujících podřetězců.

  $retezec = "12345";
  @return1 = $retezec =~ /\d\d\d/;
  @return2 = $retezec =~ /(\d)(\d)(\d)/;
  @return3 = $retezec =~ /(\d)/g;

Pole @return1 obsahuje prvek (1). Žádná hodnota nebyla zapamatovaná. Oproti tomu v poli @return2 už
takové hodnoty byly: jeho prvky (1, 2, 3) jsou právě zapamatovanými hodnotami. Poslední pole obsahuje
(1, 2, 3, 4, 5), protože se vzor aplikoval celkem 5x.

Pokračování seriálu přiblíží ukotvení vzoru v textu.

Online verze článku: http://www.linuxsoft.cz/article.php?id_article=947

Perl (17) - Regulární výrazy - kotvy

2. díl o regulárních výrazech věnujeme kotvám.

1.11.2005 07:00 | Jiří Václavík | přečteno 1264x 

Kotvy jsou speciálními znaky, které se vyznačují nulovou délkou. Jsou totiž určeny svým okolím. Nejde tedy o
znak ve svém smyslu, ale spíše o pozici. V regulárním výrazu se označuje buď speciálním metaznakem nebo 
escape znakem.

Určení začátku a konce řetězce

Dosud jsme vytvářeli vzory, které se mohli vyskytovat na libovolném místě řetězce. Nezřídka ale potřebujeme
specifikovat, jakým podřetězcem má řetězec začínat nebo jakým končit. To je asi nejčastější užití kotev.

Začátek řetězce je reprezentován znakem ^ a konec znakem $. V praxi vypadá jejich použití následovně:

  $retezec = "text, který končí\n slovem xxx";
  print $retezec =~ /xxx$/;   #true - řetězec skutečně končí na xxx
  print $retezec =~ /^text/;  #true - řetězec začíná na text
  print $retezec =~ /končí\n/;#true - řetězec obsahuje daný vzor
  print $retezec =~ /končí$/; #false - řetězec nekončí vzorem

Samozřejmě nám nic nebrání určit začátek i konec řetězce zároveň. Toto jsou zápisy, které dělají totéž, ale
každý jinak:

  $retezec = "regexp";
  print $retezec eq "regexp";  #tiskne 1 - $retezec má stejnou hodnotu jako "regexp"
  print $retezec =~ /^regexp$/;#tiskne 1 - to samé s využitím regulárních výrazů

Hranice slova

Escape sekvence \b je nalezena na místech, kde končí nebo začíná slovo - tedy v místě, kde se stýká znak slova
(\w) s jiným znakem (\W). Doplňkem pozice, určené znakem \b je \B. Tomu vyhovují všechny pozice mimo
hranici slova.

  print "slovo" =~ /\blov\b/; #nevyhovuje
  print "slov" =~ /\blov\b/;  #nevyhovuje
  print "lov," =~ /\blov\b/;  #vyhovuje; čárka není znakem slova
  print "lovec" =~ /\blov\B/; #vyhovuje

První 3 příkazy hledají v řetězci slovo "lov" a poslední příkaz slovo, které na "lov" začíná.

Začátek řádku, začátek řetězce

Dalšími znaky se speciálním významem jsou \A a ^. ^ znamená začátek řádku a \A začátek řetězce. Abychom
mohli tyto 2 znaky rozlišit, musíme ale v regulárním výrazu aktivovat režim více řádků. K tomu slouží přepínač
m (regulární výraz pak bude vypadat takto: m//m). V tomto režimu bude ^ nalezeno na začátku řetězce, ale
také za každým znakem nového řádku. \A se vždy vztahuje jen na začátek řetězce - je tedy v řetězci právě
jednou.

O přepínačích ještě nějakou dobu mluvit nebudeme. Pro pochopení stačí vědět, že se uvádějí za koncové

lomítko regulárního výrazu. Pokud k m/vzor/ chceme přidat přepínače x a y, bude zápis vypadat takto:

m/vzor/xy.

Mějme řetězec:

  radek 1\nradek 2\nradek 3

Teď do něj znázorněme znaky \A a ^. Bez použití přepínače m:

  <\A><^>radek 1\nradek 2\nradek 3

A pokud je přepínač m nastaven na "zapnuto", je na začátku každého řádku ještě ^:

  <\A><^>radek 1\n<^>radek 2\n<^>radek 3

Pokud nemáte představu, jak by se začátek řádku použil, zde je ještě konkrétní příklad:

  print "MATCHED" if "radek 1\nradek 2\nradek 3" =~ /^\Aradek 1\n^/m; #vyhovuje
  print "MATCHED" if "radek 1\nradek 2\nradek 3" =~ /\A^radek 1\n^/m; #vyhovuje - navíc je
vidět, že nezáleží na pořadí ^ a \A
  print "MATCHED" if "radek 1\nradek 2\nradek 3" =~ /^radek 1\n^/;   #nevyhovuje - není
přepínač a ^ tedy funguje stejně jako začátek řetězce

Konec řádku, konec řetězce

Mechanizmus funguje podobně jako začátek řetězce nebo řádku. Je též nutný přepínač m.

$ - odpovídá konci řetězci a je-li uveden přepínač m, je také před každým znakem nového řádku
\Z - konec řetězce

S přepínačem m vypadá řetězec a neviditelné znaky v něm následovně:

  <\A><^>radek 1<$>\n<^>radek 2<$>\n<^>radek 3<$><\Z>

Variantou znaku \Z je \z. Jejich význam se liší v tom, že u \z nesmí být na konci řetězce znak nového řádku.

  $x = "...XXX\n";
  print "MATCHED" if $x =~ /XXX\Z/; #vyhovuje
  print "MATCHED" if $x =~ /XXX\z/; #nevyhovuje

Poznámka: Existuje speciální proměnná $*, která umí nahradit přepínač m. Pokud $* == 1, fungují vzory jako

s m, v případě, že $* == 0 ne. Tato proměnná se ale nedoporučuje používat. Dávejte raději přednost uvedení

přepínače.

Poslední nalezení vzoru

Znak \G je na pozici, kterou lze určit funkcí pos, což je místo posledního úspěšného nalezení vzoru.

Funkce pos

pos vrací pozici, na které byl nalezen poslední hledaný výskyt. Je tedy nutný přepínač g, který ale teprve
budeme rozebírat. Teď napišme program, který vypíše počet výskytů určeného znaku a jeho pozice. V každé
iteraci cyklu budeme hledat v daném řetězci další pozici hledaného znaku, a právě tu přidáme do pole
@pozice.

  $" = ", ";
  $retezec = "matematika";
  while ($retezec =~ /a/g){
      push(@pozice, pos $retezec);
  }
  print "Písmeno a se v řetězci vyskytuje ", scalar @pozice, "x a to na pozicích
@pozice.\n";

Vyjádření "NEBO"

Představte si následující kód:

  if ($volba =~ /Linux/ or $volba =~ /Perl/ or $volba =~ /C/){
      print "Správná volba!";
  }

Zápis je dost nepohodlný a nepřehledný. Logické or lze naštěstí přesunout z podmínky přímo do regulárního
výrazu. Zapisuje se znakem | (někdy se nazývá alternace). Předchozí kód tak velmi zjednodušíme:

  if ($volba =~ /Linux|Perl|C/){
      print "Správná volba!";
  }

Tento kód uzná každý řetězec, ve kterém se vyskytuje 1 z hledaných podřetězců. Ale takové C se může
vyskytovat v leckterém řetězci. Proto bude vhodné specifikovat začátek a konec řetězce. K tomu nestačí uvést
jen znaky ^ a $, ale je nutné ozávorkovat vše, co patří k OR. Závorky sdružují skupinu znaků.

Poznámka - problematice závorek se ještě budeme podrobně věnovat.

  if ($volba =~ /^(Linux|Perl|C)$/){
      print "Správná volba!";
  }

Podobným způsobem je výhodné dělit slova. Kupříkladu

  /Perl(5|6)/

bude rychlejší než

  /Perl5|Perl6/

Řetězec "Perl" je totiž v 1. případě hledán pouze jednou.

Příště se konečně podíváme na třídy znaků.

Online verze článku: http://www.linuxsoft.cz/article.php?id_article=981

Perl (18) - Regulární výrazy - množiny znaků

Ve 3. pokračování o regulárních výrazech jsou představeny třídy znaků.

3.11.2005 13:00 | Jiří Václavík | přečteno 1217x 

Množina znaků je skupina obyčejných znaků. Pokud ji uvedeme ve vzoru, potom se taková množina chová
jako 1 ze znaků, které obsahuje. Máme několik možností, jak množinu definovat.

libovolný znak

Začneme tím nejjednodušším. Znak tečka zastupuje 1 libovolný znak mimo konce řádků.

Poznámka - Pokud chcete zahrnout do množiny znaků, které tečka zastupuje i znak konce řádku, použijte

přepínač /s.

Příklad vypisuje slova s názvem o délce 4:

  @slova = qw(C++ Perl Python Ruby Java);
  foreach $slovo (@slova){
      print $slovo, "\n" if ($slovo =~ /^....$/); #vypíší se slova, které zabírají 4 znaky
  }

Poznámka - Nutno ale poznamenat, že místo cyklu foreach bychom mohli jednoduše použít funkci grep:

  $, = "\n";
  print grep /^....$/, @slova;

Výčet znaků

Je možné specifikovat užší množinu znaků než libovolný znak. K tomu slouží hranaté závorky. Skupina znaků,
které jsou v hranatých závorkách, tvoří tuto užší množinu. Znaky nejsou nijak speciálně odděleny, protože to
není potřeba.

  print "MATCHED" if "SUSE Linux 9.2" =~ /S[Uu]SE Linux 9\.[123]/;
  print "MATCHED" if "SuSE Linux 9.3" =~ /S[Uu]SE Linux 9\.[123]/;
  print "MATCHED" if "SUSE Linux 9.4" =~ /S[Uu]SE Linux 9\.[123]/;

1. a 2. příklad vrací 1, poslední příklad prázdný řetězec. Číslo 4 totiž není v dané množině povolených znaků.

Pomocí alternace můžeme vytvořit vzor, kterému vyhoví i "SUSE Linux 10.0".

  print "MATCHED" if "SUSE Linux 10.0" =~ /S[Uu]SE Linux (9\.[123]|10\.0)/;

Poznámka - Zde návratová hodnota obsahuje zapamatované hodnoty. Pamatování jsme zatím neřešili.

Zkusme si spočítat, kolika různým řetězcům náš poslední vzor vyhoví. Aby to nebylo nekonečno, vzor nejdříve
ukotvíme:

  /^S[Uu]SE Linux (9\.[123]|10\.0)$/

Nyní pojďme znak po znaku (resp. množinu po množině), u každého zjistíme počet možností, a tyto počty mezi
sebou vynásobíme. 1. znak, S, je samostatný znak - tedy množina jediného znaku. [Uu] je množina 2 znaků a
jsou tedy 2 možnosti. Dále máme až k závorce vždy 1 možnost. Potom následuje 1 z řetězců 9.1, 9.2, 9.3
nebo 10.0. Tedy 4 možnosti. Velikosti množin mezi sebou vynásobíme, a tím získáváme
1*2*1*1*1*1*1*1*1*1*1*4 = 8. Naší šabloně tak vyhoví 8 řetězců.

Pro výčet znaků existuje několik usnadnění. Představte si, že chcete definovat množinu, které vyhovuje
libovolné anglické písmeno. Řešení může vypadat například takto:
[abcdefghijklmnopqrstuvwxyzABCDEFGHIJKLMNOPQRSTUVWXYZ]. Jistě uznáte, že to není to
pravé. Je možné si ale nadefinovat proměnné, kde budou skupinu znaků obsahovat a následně je použít asi
takto ve vzoru:

  $pismeno = "[abcdefghijklmnopqrstuvwxyzABCDEFGHIJKLMNOPQRSTUVWXYZ]";
  print "r" =~ /$pismeno/;

Znaky v množinách se speciálním významem

Stále to je příliš těžkopádné. Regulární výrazy nabízejí mnohem lepší řešení. Existuje totiž speciální operátor
rozsahu. Ten se zapisuje v syntaxi regulárních výrazů pomlčkou. Předchozí problém s množinou všech písmen
se tímto způsobem řeší mnohem pohodlněji. Do vzoru stačí uvést [a-zA-Z]. To samé lze dělat s čísly.
[2-5] je totéž, co [2345]. Množninu znaků, které vyhoví číslice šetnáckového zápisu čísla, tak zapíšeme
jako [0-9a-fA-F].

Zkuste si tipnout, co všechno vyhoví zápisu ^[x|y]$. Ač to možná intuitivně vypadá jako znak x nebo znak 
y, má tento zápis úplně jiný význam. Znak | nemá na tomto místě žádnou speciální funkci, proto celému
regulárnímu výrazu vyhovuje znak x, y nebo |.

Jedním z těchto speciálních znaků je ^ (negace - pozor!, se začátkem řetězce má společné jen to, že používá
stejný symbol). Užijete ji, pokud například chcete definovat množinu, které mají vyhovovat všechny znaky
kromě velkých písmen. Píše se bezprostředně za úvodní hranatou závorku (pokud je jinde, svoji funkci ztrácí).
Má ten efekt, že to, co je obvykle množinou znaků, je jejím doplňkem. Náš problém s vynecháním velkých
písmen vyřešíme takto: [^A-Z] (oproti tomu v zápisu [a-c^d-f] stříška speciální význam nemá, protože
již není na 1. pozici).

Stejně jako stříška ztrácí uvnitř hranatých závorek svoji funkci na jiném než 1. místě, ztrácí ji i uzavírací
závorka na 1. místě. Pokud je tam uvedete, bere se jen jako znak z množiny. Je-li umístěna jinam, musíte jí
předřadit zpětné lomítko. To samé platí pro pomlčku - speciální funkci má pouze, pokud dostane z obou stran
nějaké smysluplné hodnoty. V opačném případě totiž buď hlásí Invalid range (je-li na jedné straně
písmeno, na druhé číslo, nebo když jsou strany prohozeny), a nebo bere pomlčku jako znak bez speciálního
významu (pokud je uvedena na začátku nebo na konci řetězce v hranatých závorkách).

Nejen na alfanumerické znaky se může rozsah vztahovat. Lze zadat osmičkové rozpětí znaků z ASCII tabulky.
Například znaky ( (osmičkový kód 050), ) (051), * (052), + (053). Následující 2 zápisy jsou ekvivalentní:

  /^[()*+]$/
  /^[\050-\053]$/

čeština

Operátor rozsahu se na české znaky nevztahuje. Jinak je to ale s předdefinovanými množinami znaků.
Použijete-li příkaz use locale; a máte-li správně nastavené proměnné prostředí, vztahují se množiny i na
národní znaky.

  use locale;
  print "MATCHED" if "č" =~ /\w/;

Je vytisknuto MATCHED, protože znak s háčkem množině vyhovuje. Zkuste nyní odstranit use locale;. 

Potom toto tvrzení platit nebude.

Předdefinované množiny

Další elegantní řešení nabízejí předdefinované množiny znaků. Existují speciální escape sekvence, které je
zastupují. Přitom se nemusejí uzavírat hranatými závorkami. Pro srovnání do tabulky zahrnuji i tečku.

Escape 
znak

Ekvivalentní zápis Význam

\d [0-9] Číslice

\D [^0-9] nebo [^\d] Cokoliv mimo číslici

\w [_0-9a-zA-Z] Znaky identifikátorů

\W
[^_0-9a-zA-Z] nebo 
[^\w]

Cokoliv mimo znaků identifikátorů

\s [\ \n\r\t\f] Bílý znak

\S [^ \n\r\t\f] nebo [^\s] Cokoliv mimo bílého znaku

. [^\n]
Libovolný znak mimo znaku nového řádku (chování se dá pozměnit
přepínačem s)

Poznámka - Ekvivalenty \w a [_0-9a-zA-Z]) resp. \W a [^_0-9a-zA-Z] v poslední tabulce nejsou

skutečně 100% ekvivalenty. Vzory [_0-9a-zA-Z] a \w se mají chovat stejně, ale v případě použití locales
tomu tak není. Předveďme si důkaz:

  use locale;
  print "MATCHED" if "č" =~ /\w/;           #true
  print "MATCHED" if "č" =~ /[_0-9a-zA-Z]/; #false

Nyní zapíšeme regulární výraz pro formát 12hodinového zápisu času.

  /^[01]\d[:\.][0-5]\d[:\.][0-5]\d[ ]?[pPaA][mM]$/;

Vysvětleme si, jak se tento regulární výraz vyhodnocuje:

^ - začátek řetězce
[01] - znak 0 nebo 1
\d - znak z číslic 0 až 9
[:\.] - Znak : nebo . (tečka). Oddělovač hodin a minut
[0-5]\d - číslo mezi 00 a 59. Počet minut
poslední 2 body se opakují pro vyjádření sekund
[ ]? - nepovinná mezera
[pPaA][mM] - určuje, zda se jedná o dopoledne nebo odpoledne (povinné).
$ - konec řetězce

Je tu problém. Našemu výrazu vyhoví i čas 13 - 19 hodin. Musíme použít znak |. 1. operand bude hledat, zda
jsou hodiny mezi 00 a 09, pokud ne, 2.operand zjistí zda nejsou ještě mezi 10 až 12.

  /^((0\d)|(1[0-2]))[:\.][0-5]\d[:\.][0-5]\d[ ]?[pPaA][mM]$/;

Regulárnímu výrazu ((0\d)|(1[0-2])) vyhovují čísla 01 až 12, zbytek je stejný jako v minulém příkladu.
Celý výraz ještě o něco zjednoduší přepínač /i. Zápis [pPaA][mM] pak může vypadat o něco pohodlněji.

  /^((0\d)|(1[0-2]))[:\.][0-5]\d[:\.][0-5]\d[ ]?[pa]m$/i;

Sice ani toto není dokonalé řešení (vyhoví i takové věci jako "10.10:40pM"), ale pro představu nám
posloužilo.

Unicode třídy

Toto asi nebudete nikdy potřebovat, ale koho to zajímá ať čte dál. Třída znaků se zadává pomocí zápisu
\p{třída}. Pokud chceme doplněk třídy (tedy vše mimo toho, co do třídy patří), použijeme zápis
\P{třída}. 

V 1. tabulce jsou jen pro představu 2 Unicode třídy. Existuje jich ale mnohem více (man perlretut).

Třída Význam

IsSm matematický symbol

IsSc znaky měny

Znakem měny je například dolar:

  print "MATCHED" if "r" =~ /^\p{IsSc}$/;  #nevyhovuje
  print "MATCHED" if "\$" =~ /^\p{IsSc}$/; #vyhovuje

Dále si Perl některé třídy sám definuje.

Mají 2 možnosti zápisu. Buď \p{třída} nebo [:třída:]. Zápis [:třída:] je možný jen uvnitř dalších
[ ]. Samotné [:třída:] by totiž znamenalo množinu znaků :, t, ř, í, d, a. Je tedy nutný zápis
[[:třída:]].

UNICODE \p{třída} POSIX [:třída:] Význam

IsAlpha alpha Anglická písmena

IsAlnum alnum Anglická písmena a číslice

IsASCII ascii Znaky ASCII 0-127

IsSpace blank Mezera, tabulátor

IsCntrl cntrl Řídící znaky

IsDigit digit Číslice

IsGraph graph Grafické

IsLower lower Malá anglická písmena

IsPrint print Tisknutelné znaky a mezera

IsPunct punct Interpunkce a pomocné znaky

IsSpacePerl space Bílý znak

IsUpper upper Velká anglická písmena

IsWord word Alfanumerický znak nebo podtržítko

IsXDigit xdigit Šestnáckové číslo

Použití ukazuje následující kód. Všimněte si odlišných barev hranatých závorek. Vnější definují množinu
znaků, jejíž obsahem je třída xdigit.

  print "MATCHED" if "5" =~ /^[[:xdigit:]]$/; #vyhovuje - 5 je hexadecimální
  print "MATCHED" if "a" =~ /^[[:xdigit:]]$/; #vyhovuje - a je hexadecimální
  print "MATCHED" if "g" =~ /^[[:xdigit:]]$/; #nevyhovuje - g není hexadecimální

Perl umožňuje POSIX třídy negovat. Negovaná [:třída:] je [:^třída:].

Na závěr ještě něco k definici unicode tříd. Je nutné vytvořit podprogram se stejným jménem jako třída. Jeho
návratovou hodnotou jsou řádky. Je-li na řádku 1 číslo, určuje ASCII znak, určený 16kově. Jsou-li na řádku 2
čísla, oddělená mezerou, jde o rozsah. Definujeme si třídu IsZavorka, které vyhoví 1 ze znaků (, ), {, }, [, 
], <, > - tedy závorka kulatá, složená, hranatá nebo lomená.

  print "MATCHED" if "(" =~ /^\p{IsZavorka}$/; #vyhovuje
  
  sub IsZavorka {
      return <<END;
  28 29
  3C
  3E
  5B
  5D
  7B
  7D
  END
  }

Třídy lze definovat i pomocí již definovaných tříd. V takovém případě je do návratové hodnoty podprogramu
třeba uvést řádek [+!-&]balík::Trida. Implicitní unicode třídy jsou v balíku utf8. + (sjednocení)
funguje pro přidání znaků z příslušné třídy, ! (rozdíl) pro odebrání znaků, - (doplněk) pro všechny znaky
mimo znaky z této třídy a & (průnik) vybere pouze znaky, které jsou v této třídě a třídě jiné. Pro definici třídy
IsBracket, která má stejný význam jako IsZavorka stačí psát toto:

  sub IsBracket {
      return <<END
  +main::IsZavorka
  END
  }

Další možností definice unicode tříd je podle typu písma. Existují například třídy InHebrew, 
InMathematicalOperators. Celý seznam je na manuálové stránce perlunicode.

Zdroje

Toto téma by zabralo několik samostatných článků. Mnohem více o unicode třídách a jejich definicích na
manuálových stránkách

man perlretut
man perlunicode.

Online verze článku: http://www.linuxsoft.cz/article.php?id_article=1000

Perl (19) - Regulární výrazy - opakování a kvantifikátory

Jedním ze základních kamenů regulárních výrazů jsou kvantifikátory, které umožňují aplikovat část regulárního
výrazu vícekrát.

18.11.2005 07:00 | Jiří Václavík | přečteno 1094x 

Z předchozích dílů o regulárních výrazech zatím nejsme schopni (opomineme alternaci, ta má v regulárních
výrazech jiný účel) vytvořit šablonu pro řetězce různé délky. Právě toto mají na starosti kvantifikátory.
Kvantifikátorem určujeme počet opakování části regulárního výrazu. To znamená, že určitý počet znaků za
sebou můžeme porovnat s jednou množinou znaků (šablonou pro 1 znak). Dosud jsme takto porovnávali vždy
pouze 1 znak.

Kvantifikátor se vždy uvádí za množinu znaků, jejíž četnost specifikuje. Množinou může opět být jak výčet
znaků pomocí hranatých závorek, předdefinovaná množina tak i samotný znak.

Libovolný počet opakování

Uvedením hvězdičky dáváme najevo, že množina znaků před ní se může opakovat libovolněkrát.

Regulárnímu výrazu x* (x je množina znaků, * je počet opakování) potom vyhovuje libovolný řetězec,
skládající se z písmen x nebo prázdný řetězec. V prázdném řetězci je 0 písmen x, což hvězdička též zahrnuje.

Jak bude vypadat podmínka, která určuje, zda bylo zadáno číslo?

  print "Zadejte číslo:";
  $cislo = <STDIN>;
  if ($cislo !~ /^[0-9]*$/){
      print "Nezadal jste číslo!\n";
  }

Problémem je, že regulárnímu výrazu vyhoví i prázdný řetězec. Tomu se dá zamezit například přidáním další
číselné množiny před výraz. Potom musí být 1. znak číslem a další, pokud jsou, také:

  /^[0-9][0-9]*$/

Nejméně 1 výskyt

Asi nejlepším řešením posledního problému by bylo použití znaku +, který provádí opakování minimálně
jednou, a prázdný řetězec tudíž nevyhoví. Tyto zápisy mají stejný význam:

  /^[0-9][0-9]*$/
  /^[0-9]+$/

Nejvíce 1 výskyt

Použijete-li jako kvantifikátor otazník, má téměř stejný význam kdyby tam žádný kvantifikátor nebyl. Tyto 2
možnosti se liší v tom, že otazníku vyhoví ještě navíc prázdný řetězec. Jedna číslice nebo prázdný řetězec
vyhovuje výrazu

  /^[0-9]?$/

Jiný počet opakování

Pokud se vám nehodí žádná z dosud nabízených možností, nabízejí regulární výrazy zápis počtu opakování
pomocí složených závorek. Ty vymezují spodní a horní hranici počtu opakování. Existuje několik možností
syntaxe.

  {minimum,maximum}
  {minimum,}
  {počet}

Čísla vyjadřující minimum a maximum nejsou neomezené. Záleží na nastavení při kompilaci. Obvykle nelze
používat větší čísla než 32766, což dokazuje hláška Perlu Quantifier in {,} bigger than 32766 
in regex, která na vás vyskočí po překročení definované meze.

Vezměmě si výraz:

  /^ab{2,4}c/

Vyhoví mu řetězce abbc (2 výskyty), abbbc (3 výskyty), abbbc (4 výskyty), ale už ne abc (1 výskyt) nebo
abbbbbc (5 výskytů). Pokud je uveden jen 1 argument a není použita čárka, vyhoví vzoru pouze řetězce,
obsahující danou množinu znaků právě tolikkrát, kolik je uvedeno v argumentu. Další možnost nabízí uvedení
čárky, ale bez maxima. V takovém případě je maximum nekonečno.

Potřebujete-li zapsat do regulárního výrazu nějaký rozsah opakování s mezerou (například 2, 3 nebo 5 a více
opakování, ale ne 4 opakování), musíte si vypomoci alternací.

Metaznaky +, * a ? pro počet opakování jsou tedy jen speciálními případy, které lze zapsat i pomocí složených
závorek:

Metaznak Ekvivalent

+ {1,}

* {0,}

? {0,1}

Ukažme si nyní program, který ověřuje bezpečnost hesla. Heslo budeme považovat za bezpečné, bude-li
splňovat tyto 4 podmínky:

obsahuje alespoň 1 číslici
obsahuje alespoň 1 písmeno
obsahuje alespoň 1 znak, který není znakem slova
je delší než 5 znaků

Pro každou podmínku napíšeme regulární výraz a pokud vyhoví heslo všem těmto vzorům, máme jistotu, že je
heslo bezpečné.

  print "Zadejte heslo: ";
  chomp($_ = <STDIN>);
  if (/[0-9]/ and /[a-zA-Z]/ and /\W/ and /.{6,}/){
      print "Heslo je bezpečné\n";
  }else{
      print "Heslo není bezpečné\n";
  }

Poznámka - pokud byste chtěli, aby při zadávání hesla nezobrazovaly jeho pravé znaky, ale pouze hvězdičky

(případně vůbec nic), použijte modul Term::ReadKey.

Hladovost a sytost kvantifikátorů

Metaznaky {}, *, + a ? jsou takzvaně hladové (,,greedy"). To znamená, že spolknou co možná nejdelší část
řetězce. Dokazuje to tento kód:

  $_ = "123456789";
  /(\d*)/;
  print $1; #tiskne hodnotu, která vyhověla vzoru - více v příštím díle

Je vytisknuto 123456789 - tedy celý řetězec. To i přesto, že by vyhověl třeba také prázdný řetězec,
vyhodnocování regulárního výrazu by se mohlo úspěšně ukončit a program by běžel dál. Kvantifikátory jsou
hladové, a tak pohlcují nejdelší vyhovující podřetězec. To samé by platilo i v případě, kdybychom jako
kvantifikátor místo hvězdičky zvolili například {3,5}. Potom by předchozí kód vytiskl prvních 5 číslic a ne 3,
ačkoliv by to také stačilo.

Tato vlastnost může občas činit problémy. Tak například, pokud chceme v HTML kódu vyhledat 1 obrázek a
vypsat kód celé značky IMG:

  $html = 'abc....<IMG SRC="obrazek" ALT="obrázek">....xyz';
  $html =~ /(<((IMG)|(img))\ (.*)>)/;
  print $1;

Hvězdička vlivem hladovosti spolkne znaky do konce řetězce. Ale tam v našem případě znak > není. Proto se
začne z místa, po které hvězdička spolkla text (v našem případě konec řetězce) hledat. Najde ho, ukončí další
vyhodnocování, protože je na konci regulárního výrazu a program vytiskne přesně to, co jsme chtěli. Takže v
pořádku. Problém nastane, pokud se znak > vyskytuje ještě někde mezi místem, po které byl spolknut text a tím
"správným" znakem >. To si můžeme jednoduše ilustrovat:

  $html = '<TABLE><TR><TD><IMG SRC="obrazek" ALT="obrázek"></TD><TD>...</TD></TR></TABLE>';
  $html =~ /(<((IMG)|(img))\ (.*)>)/;
  print $1;

Začátek je stejný jako v minulém příkladu. Hledá se řetězec "<IMG ", poté hvězdička pohltí vše do konce a
odtud se opačným směrem hledá znak >. Ale takový se vyskytne už v značce </TABLE>! To znamená, že
získáme řetězec <IMG SRC="obrazek" ALT="obrázek"></TD></TR><TD>...</TD></TABLE>.
To jsme opravdu nechtěli. A může za to právě hladovost.



LINUX SOFTWARE file:///C:/Ikon/Books/Perl/LinuxSoft_PERL.html

73 of 102 3/8/2006 20:54

Možnost, jak vzniklý problém řešit, může být přes funkce. To je zbytečně složité.

  $html = '<TABLE><TR><TD><IMG SRC="obrazek" ALT="obrázek"></TD><TD>...</TD></TR></TABLE>';
  $html =~ /(<((IMG)|(img))\ (.*)>)/;
  print substr($1, 0, (index $1, ">")+1);

Jiné, o dost lepší řešení, lze aplikovat pomocí negované množiny znaků. Místo libovolného znaku (v našem
regulárním výrazu reprezentován tečkou), specifikujeme vše mimo znaku >.

  $html = '<TABLE><TR><TD><IMG SRC="obrazek" ALT="obrázek"></TD><TD>...</TD></TR></TABLE>';
  $html =~ /(<((IMG)|(img))\ ([^>]*)>)/;
  print $1;

Dalším řešení nabízí sytost. Existují další 4 kvantifikátory, které fungují úplně stejně jako ty nám dosud známé,
až na to, že nejsou hladové. Jsou to {}?, *?, +? a ??. Zapisují se stejné jako {}, *, + a ?, jen se za ně
připisuje otazník. Pohltí minimální možný počet znaků, který vyhoví vzoru. Tento vzor tedy nepohltí 5 znaků,
jak by to udělal hladový kvantifikátor, ale pouze 3:

  $_ = "123456789";
  /(\d{3,5}?)/;
  print $1;

Nakonec si ještě ukážeme řešení pro problém s vyhledáním obrázku v HTML kódu pomocí sytosti:

  $html = '<TABLE><TR><TD><IMG SRC="obrazek" ALT="obrázek"></TD><TD>...</TD></TR></TABLE>';
  $html =~ /(<((IMG)|(img))\ (.*?)>)/;
  print $1;

Příště se podíváme na funkci kulatých závorek v regulárních výrazech.

Online verze článku: http://www.linuxsoft.cz/article.php?id_article=1018

Perl (20) - Regulární výrazy - magické závorky

Dnešní díl popisuje použití kulatých závorek pro změnu priority, seskupování a pamatování.

5.12.2005 06:00 | Jiří Václavík | přečteno 1130x 

V regulárních výrazech mají závorky hned několik důležitých a jinak nenahraditelných významů. Mezi jejich
hlavní funkce patří seskupení, priorita a paměť. Rozeberme si postupně tyto tři vlastnosti.

Priorita

Stejně jako pro operátory existuje tabulka priorit i pro metaznaky. Určuje, ke kterým znakům se váže určitý
metaznak. Čím výše v tabulce, tím je metaznak prioritnější.

Metaznaky Význam

( ), (?: ) závorky

*, ?, +, {}, *?, ??, +?, {}? kvantifikátory

aa, \x, ^, $ apod. posloupnost znaků, kotvy

| alternativy

Priorita se dá měnit kulatými závorkami.

Seskupování

Kvantifikátor jsme zatím používali pouze pro znak nebo třídu znaků, která ho předcházela. Pomocí závorek
můžeme docílit toho, aby se vztahoval na nějakou delší část regulárního výrazu. Srovnejme si tyto 2 vzory:

  /(xy){3}/;
  /xy{3}/;

V 1. případě jsou znaky xy seskupeny. To znamená, že se kvantifikátor vztahuje na celý řetězec xy v
závorkách. Lze to vysvětlit také jako změnu priority - kvantifikátor má menší prioritu než závorky a proto je
aplikován na celý uzávorkovaný výraz. Takovému regulárnímu výrazu vyhoví řetězec, který obsahuje
xyxyxy. Další řádek je stejný až na to, že chybí závorky a znaky xy seskupeny nejsou. Kvantifikátor je
prioritnější než posloupnost znaků a vztahuje se pouze na znak y. Vzoru tedy vyhoví xyyy.

Pamatování obsahu závorek

Vše, co uzávorkujete kulatými závorkami si Perl, pokud mu to nezakážete, pamatuje. To je vlastnost, se kterou
lze dělat opravdu nevídané věci.

Výraz $text =~ /(...)/ vrací úsek textu, který vyhověl regulárnímu výrazu. Tímto způsobem můžeme
do pole velice pohodlně získat třeba seznam slov nebo čísel. Právě seřazený výtah čísel ze zadaného textu
tiskne následující ukázka.

  $, = ", ";
  $text = "333 stříbrných stříkaček stříkalo přes 33 stříbných střech";
  @cisla = ($text =~ /(\d+)/g);
  print sort @cisla; #tiskne 33, 333

Obsah závorky se ukládá okamžitě a takto uložený řetězec se dá použít ještě ve vzoru. To, co je v 1. závorce je
přístupné v \1, další závorka \2 a tak dále až do \99 (dále ne, protože by se takový zápis překrýval s
osmičkovým zápisem znaku. \100 je zavináč, \101 je znak A apod.). Tento mechanizmus nám otevírá
spoustu dalších možností.

Pokusme se zapsat vzor, který vyhoví řetězci o 5 stejných číslicích. Jinými slovy vzor, kterému, pokud začíná
na 1, vyhoví pouze 11111, pokud začíná na 2, vyhoví pouze 22222, atd. Pokud vzor nezačíná číslicí,
nevyhoví nikdy.

Jako 1. znak hledáme číslici. Zápis množiny číslic musí být v závorkách, abychom její hodnou neztratili. Dále
použijeme zapamatovanou hodnotu jako množinu znaků a otestujeme ji 4x. Nakonec označíme začátek a konec
řetězce.

  print "MATCHED" if "55555" =~ /^(\d)\1{4}$/; #vyhovuje
  print "MATCHED" if "555555" =~ /^(\d)\1{4}$/;#nevyhovuje
  print "MATCHED" if "11112" =~ /^(\d)\1{4}$/; #nevyhovuje
  print "MATCHED" if "xxxxx" =~ /^(\d)\1{4}$/; #nevyhovuje

Tato vlastnost se hodí zejména při nahrazování, které bude rozebráno v příštím díle.

Podobně lze pomocí závorek získávat hodnoty i mimo samotný vzor. Obsahy jsou přístupné ve speciálních
proměnných $1, $2 až $99. Nyní z řetězce opět vyseparujeme všechna čísla. Použijeme cyklus, každou iteraci
bude 1 číslo zapamatováno a v těle cyklu vytištěno.

  while ("333 stříbrných stříkaček stříkalo přes 33 stříbných střech" =~ /(\d+)/g){
      print "Číslo: $1\n";
  }

Pořadí (u zápisu \n i $n) se uděluje na základě pozice otevírací závorky. Tento systém řeší případné
víceúrovňové zanoření. Nějaké podřetězce tak mohou být uloženy i vícekrát. Nic tedy nebrání například zápisu
((\d)) - ta stejná číslice bude uložena do více proměnných. Zanořování si ilustrujeme na následujícím úseku
kódu. Ten vypíše první a poslední číslici nejméně trojmístného zadaného čísla a také toto celé číslo.

  <STDIN> =~ /^((\d)\d+(\d))$/;
  print "1. číslice: $2\n";
  print "poslední číslice: $3\n";
  print "celé číslo: $1\n";

Poznámka - Pokud řetězec nebude vyhovovat vzoru, proměnné zůstanou prázdné - obsah závorek se ukládá,

jen pokud byl test vyhodnocen pravdivě. Problémy pak mohou nastat, pokud v proměnných $1, $2 atd. zbyly

nějaké hodnoty z předchozího testu. Jde o velice nenápadnou chybu a může trvat dlouho, než je odhalena.

Poznámka - Jaký je tedy rozdíl mezi \n i $n? \n je do paměti ukládáno okamžitě s uzavírací závorkou, takže

je možné tímto způsobem zapamatovanou hodnotu použít ještě ve vzoru. Oproti tomu proměnné $n se načtou

vždy až po úspěšném srovnání řetězce se vzorem.

Ukládání do \n, resp. $n lze zabránit speciální syntaxí pro zápis závorek. Místo levé (otevírací) závorky
použijte posloupnost znaků (?:. V takovém případě se nic ukládat nebude a proměnná $n se bude chovat jako
ostatní nedefinované proměnné, což dokazuje kód:

  "123456789" =~ /^(?:\d*)$/;
  print "Hodnota uložena\n" if defined $1;

Vždy, když není pamatování potřeba, je lepší použít právě verzi bez zapamatování. Zvyšuje se tím rychlost
vyhodnocení regulárního výrazu.

Hezky lze také využít pamatování v souvislosti s kontextem. Tato vlastnost umožňuje například snadno
vyseparovat z rodného čísla den, měsíc a rok narození. Stačí jen uzavřít příslušné části regulárního výrazu do
závorek a regulární výraz přiřadit do nějakého seznamu.

  $rc = "3012013522";
  ($rok, $mesic, $den) = $rc =~ /^(\d\d)(\d\d)(\d\d)\d{4}$/;
  print "Narozen $den.$mesic.19$rok.";

Poznámka - Příklad je samozřejmě zjednodušený - nebere v úvahu ženská rodná čísla a narozené jindy než

mezi roky 1900 a 1999.

Nejdříve se vyhodnotil pravý operand přiřazení. Tak jsme získali seznam, který jsme přiřadili do už
pojmenovaného seznamu. Problém opět nastává, pokud srovnávaný řetězec nevyhoví regulárnímu výrazu. Lze
to však řešit například umístěním podmínky za příkaz.

  print "Narozen $den.$mesic.19$rok." if ($rok, $mesic, $den) = $rc =~ 
/^(\d\d)(\d\d)(\d\d)\d{4}$/;

Speciální proměnné

V proměnné $& je část testovaného řetězce, která se shoduje se vzorem. Tímto způsobem si můžeme mimo jiné
hezky ilustrovat hladovost kvantifikátorů.

  "číslo 101 je prvočíslo" =~ /\d+.{2,10}/;
  print $&;
  
  "číslo 101 je prvočíslo" =~ /\d+.{2,10}?/;
  print $&;

1. zápis je hladový. Spolkne 10 znaků za číslem 101: "101 je prvočí". Pokud za kvantifikátor přidáme
otazník, stává se sytým a pohltí pouze nejmenší možný počet znaků, které vyhoví: "101 j".

Další proměnné $` a $' tisknou tu část testovaného řetězce, kterou nevytiskla proměnná $&. $` tiskne část
před ("číslo ") a $' část za ("e prvočíslo").

  "číslo 101 je prvočíslo" =~ /\d+.{2,10}?/;
  print $`;
  print $';

V proměnné $+ je uložen poslední závorkami zapamatovaný řetězec.

Poznámka - Je dobré mít na paměti, že proměnné $&, $`, $' a $+ značně zpomalují program.

Speciální proměnná @- si pamatuje v prvku s indexem 0 první pozici v testovaném řetězci, která vyhovuje
vzoru. V dalších prvcích pak pozice začátků zapamatovaných podřetězců. Proměnná @+ dělá to samé, ale

pamatuje si místo začátků konce.

Funkce split a regulární výrazy

Oddělovač, který je parametrem této funkce, lze uvést i jako regulární výraz. To je v mnoha případech opravdu
užitečné. Mějme nějaký text, ze kterého načteme do pole seznam slov.

  $, = "\n";
  @pole = split (/\W+/, "Text, ve kterém je interpunkce!");
  print @pole;

Vše mezi slovy (tedy oddělovače - mezery a interpunkce - lépe řečeno vše, co vyhovuje vzoru) je navždy
ztraceno. Být tomu tak vždy nemusí. Proč, to si budeme demonstrovat na jiném příkladu. Budeme mít nějaký
řetězec, ve kterém jsou promíchány čísla a právě čísla zvolíme jako oddělovač.

  $, = " "; $\ = "\n";
  print split (/\d+/, "abc01def2ghijk34567l89mnop");
  print split (/(\d+)/, "abc01def2ghijk34567l89mnop");
  print split (/(?:\d+)/, "abc01def2ghijk34567l89mnop");

V 1. případě se vytiskne seznam získaných podřetězců tak, jak jsme to dosud znali. Zajímavé to začíná být až
na dalším řádku. Vzor je uveden v závorkách, které zde mají ten efekt, že se do výsledného seznamu tisknou i
oddělovače! Skutečným oddělovačem je tedy (v tomto konkrétním případě) hranice mezi podřetězcem
vyhovujícím vzoru a jiným znakem. Nic z původního řetězce tak není ztraceno. Posledním případem je
speciální druh uzávorkování, které této vlastnosti zamezuje - oddělovače se chovají jako v 1. případě.

  $ perl split.pl
  abc def ghijk l mnop
  abc 01 def 2 ghijk 34567 l 89 mnop
  abc def ghijk l mnop
  $

Je též možné ozávorkovat pouze část vzoru. V tom případě bude do výsledného seznamu přiřazena příslušná
část oddělovače.

  print split (/(\d+)45/, "***12345***12345***123***12345678***");

Podtržené podřetězce ukazují oddělovače. Ve výsledném seznamu ale bude jejich část (znaky 45) odmazána,
protože jsou mimo závorky (částí oddělovače ale samozřejmě zůstávají).

Dále můžete také vyzkoušet závorky různě vnořovat, zdvojovat apod. V takových případech budou ve
výsledném seznamu všechny uzávorkované řetězce. A to i přesto, že některé části pak budou ve výsledném
seznamu vícekrát.

Online verze článku: http://www.linuxsoft.cz/article.php?id_article=1029

Perl (21) - Regulární výrazy - nahrazování

Do této chvíle jsme regulární výrazy používali pouze k vyhledávání. To je ale jen polovina toho, co skutečně
dokáží.

13.12.2005 06:00 | Jiří Václavík | přečteno 941x 

Jak již o regulárních výrazech víme, slouží nejen pro vyhledávání vzorů, ale také pro nahrazování jejich částí.
Představme si, že máme nějaký textový řetězec a potřebujeme z něj vymazat všechna čísla. Už známe několik
způsobů, jak tento problém řešit. Celý řetězec by se dal rozdělit na znaky a každý získaný znak porovnat s
možnými číslicemi. Další možností je rozdělit řetězec funkcí split s číselným oddělovačem. Nicméně to je
všechno zbytečně pracné.

Regulární výrazy mají pro takové případy další možnou syntaxi. Chceme-li nahrazovat, použití regulárního

výrazu se s prostým vyhledáváním v několika detailech liší. V zápisu se místo úvodního m používá s a jeho
uvedení je povinné. Následuje vzor a řetězec, kterým bude úsek vyhovující vzoru nahrazen. Protože zde je o
položku více, je třeba oddělovač navíc.

  s/vzor/náhrada/

Ač je zřejmé, že tomu nemůže být jinak, podotkněme, že náhrada je vždy prostým řetězcem a nikoliv
regulárním výrazem.

V diskuzních fórech se dnes a denně setkáváme s reakcemi typu

  s/nabýdku/nabídku/

To je právě ukázka nahrazení, jen bez využití regulárních výrazů. Autor takového příspěvku dává na vědomí,
že pisatel udělal pravopisnou chybu a měla by být opravena. Tímto způsobem bychom opravu řešili v Perlu:

  $reakce = "Takovou nabýdku nebudu komentovat...";
  $reakce =~ s/nabýdku/nabídku/;
  print $reakce; #tiskne opravený text

I kdyby bylo v řetězci více stejných chyb, opraví se jen jedna z nich. Vždy se nahrazuje pouze 1. výskyt. Po
jeho nalezení totiž vyhledávání vzoru skočí úspěchem. Náš problém, kdy jsme chtěli z řetězce odstranit všechna
čísla, by se dal řešit uvedením přepínače g. Ten se, podobně jako u vyhledávání, aplikuje na všechny výskyty
vyhovující vzoru v řetězci. Výskyty tedy budeme nahrazovat prázdným řetězcem.

  $retezec = "P21E251215563R413215305711L587";
  $retezec =~ s/\d+//g;
  print $retezec; #tiskne PERL

Jednoduché nahrazování - konverze znaků

Tato problematika do regulárních výrazů nepatří, ale má s nimi některé společné rysy. Z tohoto důvodu je
zařazena právě zde.

Činnost této konstrukce není nic jiného, než nahrazování konkrétního znaku (nikoliv regulárního výrazu) za
jiný konkrétní znak.

Syntaxe je podobná jako při nahrazování u regulárních výrazů:

  $text =~ tr/abc/ABC/;
  $text !~ tr/abc/ABC/;
  $text =~ y/abc/ABC/;

y a tr jsou 2 synonymní zápisy.

Po vykonání libovolného ze zmíněných tří příkazů se obsah proměnné $text změní. Všechna a se nahradí za
A, všechna b za B a všechna c za C. Obecně se tedy ntý znak na levé straně nahrazuje ntým znakem z pravé
strany.

Rozdíl mezi použitím !~ nebo =~ je v návratové hodnotě. Výraz s !~ vrací true, pokud nedošlo k žádnému
nahrazení. Pokud ano, vrací false. !~ se u tr příliš nepoužívá. Naproti tomu =~ vrací vždy počet nahrazení.
Pokud je tento počet nenulový, návratová hodnota je tedy true.

I u tr fungují rozsahy. Není tak problémem v proměnné $text nahradit všechna malá písmena za velká, bez
toho abychom je všechny vypisovali (pomiňme použití funkce uc):

  $text =~ tr/a-z/A-Z/;

Teď si objasníme sporné případy použití tr:

Pokud je na levé straně tentýž znak uvedený více než 1×, platí pouze 1. výskyt a ostatní jsou ignorovány.

Pokud je na levé straně více znaků než na pravé, jsou znaky na pravé straně příslušně doplněny. A to tak,
že poslední znak pravé strany je naklonován, kolikrát je potřeba. Například:
  $text = "ABCDEF";
  $text =~ tr/A-F/xyz/;
  print $text;

Výsledkem bude xyzzzz. Význam tr/A-F/xyz/ je stejný jako tr/A-F/xyzzzz/, protože znak z
je příslušně nakopírován.
Pokud je na levé straně méně znaků než na pravé, je příslušný počet znaků na pravé straně zprava
ignorován.
  $text = "ABCDEF";
  $text =~ tr/AB/xyz/;
  print $text;

tr/AB/xyz/ má shodný význam se zápisem tr/AB/xy/, proto je tisknuto xyCDEF

Jako ukázku si nemohu odpustit Caesarovu šifru (v Unixu příkazy caesar, rot13). Jde o to nahradit každý
znak (písmeno) znakem, který je v abecedě (ASCII tabulce, nebo zkrátka v nějaké soustavě znaků) o určitý
počet znaků dále nebo blíže. Poprvé tuto šifru použili ke komunikaci Caesar a Cicero v době galské války.
Písmena tehdy posouvali vždy o 3 znaky v abecedě dopředu.

Vytvoříme Caesarovu šifru s posunutím +4. Na vstupu (případně v souboru jako argumentu) bude program
získávat otevřený text a následně vypisovat text šifrovaný.

K tomu potřebujeme cyklicky získávat řádky textu a zpracovávat je pomocí konstrukce tr. Posunutí +4
znamená, že a nahrazujeme za e, b za f, ..., v za z, w za a, x za b, y za c a z za d.

Takhle bude vypadat šifrovaná abeceda:

  Otevřená abeceda   abcdefghijklmnopqrstuvwxyz

  Šifrovaná abeceda (+4)   efghijklmnopqrstuvwxyzabcd

Otevřenou abecedu nahradíme šifrovanou:

  tr/abcdefghijklmnopqrstuvwxyz/efghijklmnopqrstuvwxyzabcd/;

Můžeme také využít rozsahy:

  tr/a-z/e-za-d/;

Vytvoříme cyklus, každou iteraci se načte řádek textu, provede se zašifrování a vytiskne se šifrovaný text:

  while ($radek = <>){
      $radek =~ tr/a-z/e-za-d/;
      print $radek;
  }

Také u tr lze pomocí výchozí proměnné vynechat operátor =~.

  while (<>){
      tr/a-z/e-za-d/;
      print;
  }

A abychom dovedli tuto ukázku téměř k dokonalosti, budeme nahrazovat i velká písmena:

  while (<>){
      tr/a-zA-Z/e-za-dE-ZA-D/;
      print;
  }

Přepínače

Pro tr existují dohromady 3 přepínače. V tabulce je jejich přehled.



LINUX SOFTWARE file:///C:/Ikon/Books/Perl/LinuxSoft_PERL.html

79 of 102 3/8/2006 20:54

Přepínač Význam

s Více stejných znaků za sebou, které mají být nahrazeny, jsou nahrazeny pouze 1 znakem.

c Funguje podobně jako negace hledaných znaků. Hledá a nahrazuje se jejich doplněk.

d
Není-li dostatek znaků na pravé straně tr, nenahrazují se přebytečné znaky levé strany posledním
znakem pravé strany, ale prázdným řetězcem.

Přepínač s

  $text = "xxxxyxx";
  $text =~ tr/x/X/s;
  print $text;

Podřetězce 'xxxx' a 'xx', jsou každý nahrazeny pouze jedním znakem X. Je tak vytisknuto 'XyX'.

Přepínač c

  $text = "xxxxyxXXxx";
  $text =~ tr/A-Z/ /c;
  print $text;

Bez uvedení přepínače by byla nahrazena všechna velká písmena mezerou. Protože zde ale je c, nahrazuje se
vše mimo velkých písmen a tiskne se '      XX  '. V případné kombinaci s přepínačem s by výsledkem
byl řetězec ' XX '

Přepínač d

  $text = "ABCDEFGH";
  $text =~ tr/ABC/x/d;
  print $text;

Vytisknuto je jen 'xDEFGH'. Hledáme znaky ABC, z nich pro B a C nemáme náhradu, a protože je uveden
přepínač d, budeme je nahrazovat prázdným řetězcem. V případě, že by zde přepínač d nebyl, B a C by se
nahrazovaly stejně jako A znakem x.

Počet výskytů znaku v řetězci

Protože výraz s operátorem =~ vrací počet nahrazení, vrací zároveň počet původních i nových znaků. Pokud
tedy nahrazujeme znak stejným znakem, žádné změny v řetězci nenastanou. Pouze získáme počet výskytů.
Řádek kódu přiřazuje do proměnné $vyskytu počet výskytů znaku x v obsahu proměnné $text:

  $vyskytu = $text =~ tr/x/x/;

Příště budeme pokračovat přepínači a speciálními konstrukcemi v perlových regulárních výrazech.

Online verze článku: http://www.linuxsoft.cz/article.php?id_article=1046

Perl (22) - Regulární výrazy - přepínače

Jak použít v regulárních výrazech přepínače?

21.12.2005 06:00 | Jiří Václavík | přečteno 856x 

Přepínače (volby) mění chování regulárního výrazu. Uvádějí se na konec regulárního výrazu (tedy za poslední
lomítko, je-li uvozovacím znakem). Jejich počet není omezen. Lze je definovat globálně nebo, jak poznáme v
příštím díle, lokálně.

Kompletní seznam přepínačů pro hledání a nahrazování je v tabulce:



LINUX SOFTWARE file:///C:/Ikon/Books/Perl/LinuxSoft_PERL.html

80 of 102 3/8/2006 20:54

Přepínač Význam

e vyhodnocuje náhradu jako výraz (jen u nahrazování)

g pamatuje si pozici, na které skončilo poslední vyhledávání

i nezáleží na velikosti písmen

m metaznaky ^ resp. $ jsou na začátku resp. konci všech řádků

o překládá vzor jen jednou

s množina značící se tečkou zahrnuje i znak nového řádku

x speciální syntaxe regulárních výrazů s komentáři

S některými jsme se již setkali, s některými zatím ne. Nyní jejich použití komplexně shrneme.

Velikost písma

  print "MATCHED" if "Perl" =~ /perl/i; #vyhovuje

Regulární výraz vrátí true. V řetězci se sice "perl" nikde nevyskytuje, ale protože s přepínačem i se
nerozlišuje velikost písmen, vyhoví i "Perl".

Vícenásobné prohledávání

Uvedením přepínače g si Perl zapamatuje, na které pozici byl nalezen výskyt a příště pokračuje od ní. To se dá
využít v testu cyklu. Používá se k vyhledávání nebo nahrazování všech výskytů (implicitně je hledán jen 1.
výskyt).

  $retezec = "www.linuxsoft.cz";
  while ($retezec =~ /linux/g){
      $i++;
  }
  print "Počet výskytů slova linux v řetězci je $i\n";

Existuje i speciální varianta přepínače g a to gc. Jejím použitím zabráníte resetu hodnoty v případě nezdaru při
porovnávání.

V seznamovém kontextu lze přepínač g užít, chceme-li získat seznam všech zapamatovaných řetězců nebo
řetězců, které vyhověli vzoru.

  @slova = ($retezec =~ /\w+/g);

Komentáře v regulárních výrazech

Přepínač x aktivuje speciální zápis regulárních výrazů. Čitelnost výrazu přitom rapidně roste. Jsou totiž
ignorovány mezery a znaky nového řádku v regulárních výrazech. To mimo jiné umožňuje další příjemnou věc,
kterou je možnost využití komentářů. Pozor si dejte jen na lomítko, popř. jiný zvolený uvozovací znak v
komentáři.

Připomeňme si vzor, kterému vyhoví dvanáctihodinový čas:

  /^((0\d)|(1[0-2]))[:\.][0-5]\d[:\.][0-5]\d[ ]?[pa]m$/i;

Na 1. pohled asi těžko poznáte, co by měl takový výraz vyjadřovat. S přepínačem x to bude za pár sekund
jasné:

  $cas = "12:22:11 am";
  print "MATCHED\n" if $cas =~ /
      #regulární výraz pro formát dvanáctihodinového času
      ^
      ((0\d)|(1[0-2]))   #HODINA - číslo mezi 00 a 12
      [:\.]              #oddělovač hodin a minut

      [0-5]\d            #MINUTA - číslo mezi 00 a 59
      [:\.]              #oddělovač minut a sekund
      [0-5]\d            #SEKUNDA - číslo mezi 00 a 59
      [ ]?               #nepovinná mezera
      [pa]m              #určení doby - dopoledne nebo odpoledne; přepínač i zajišťuje, že
nezáleží na velikosti písmen
      $
      /xi;

Toto, jak se dozvíte příštím díle, není jediný způsob, jak vkládat do regulárních výrazů komentáře.

Jednorázová kompilace regulárního výrazu

Zejména pro urychlení programu se používá přepínač o. Regulární výraz v nějakém cyklu s přepínačem o je
přeložen vždy pouze jednou a tento překlad je pak použit v každé iteraci. Tedy bez ohledu na hodnoty
proměnných, které, jak víme, lze do vzorů také zakomponovat. Proměnné, uvedené v regulárním výrazu, se
mohou měnit, a pak se tedy mění i samotný regulární výraz. Přepínač o tomu z výše uvedeného důvodu
zamezuje. Každou iteraci je použit regulární výraz, který vznikl kompilací v první iteraci.

Ukládání regulárních výrazů

S přepínačem o souvisí jiná věc. Existuje možnost předkompilace - použití konstrukce qr//. Pokud takový
regulární výraz přiřadíte do proměnné, lze ji používat místo onoho regulárního výrazu.

  $reg = qr/\d\d\d/;
  print "MATCHED" if "12" =~ $reg;   #nevyhovuje
  print "MATCHED" if "123" =~ $reg;  #vyhovuje
  print "MATCHED" if "77a" =~ $reg;  #nevyhovuje
  print "MATCHED" if "7744" =~ $reg; #vyhovuje

Je též možné takový regulární výraz v proměnné zařadit do jiného regulárního výrazu.

  print "MATCHED" if "7744" =~ /^$reg$/;   #nevyhovuje
  print "MATCHED" if "7744" =~ /^\d$reg$/; #vyhovuje

Zkusíte-li proměnnou, ve které je regulární výraz uložen, vytisknout, bude vypadat výstup v našem případě
takto: (?-xism:\d\d\d).

Náhrada jako výraz

S touto vlastností můžete plodit divy. U nahrazování jsme psali náhradu jako text. Uvedení přepínače e umožní
napsat náhradu jako výraz (má to mnoho společného s eval). To by nebylo nic objevného, kdyby v ní nešlo
používat zapamatované proměnné. Právě v tom tkví kouzlo.

Uvedeme si 2 příklady. Přepínač e se dobře vysvětluje pomocí funkce reverse. Převrátíme pořadí písmen ve
všech slovech řetězce:

  $retezec = "prisel jsem - videl jsem - zvitezil jsem";
  $retezec =~ s/(\w+)/reverse $1/ge;
  print $retezec; lesirp mesj - lediv mesj - lizetivz mesj

V proměnné $1 máme uložena postupně všechna (je použit také přepínač g) slova a na každé je aplikována
funkce reverse.

Pojďme dál a zkusme něco složitějšího. V řetězci, který budeme zpracovávat, se vyskytují ceny v amerických
dolarech. Upravíme tento řetězec regulárním výrazem tak, abychom dolary nahradili českými korunami a to se
vším všudy. Musíme tedy přepočítat částku v dolarech na částku v korunách a zaměnit symbol měny. Navíc je
nutné ošetřit případné desetinné ceny.

  $usd2czk = 24.133; #kurz k dolaru z 19.12.2005
  $retezec = "Cena: 20.5\$";
  $retezec =~ s/((\d+(.\d+)?)(\$|USD))/$2*$usd2czk.CZK/ge;
  print $retezec;

Předpokládali jsme, že symbol měny se píše vždy za hodnotu a mezi hodnotou a symbolem není mezera.
Tisknut je řetězec "Cena: 494.7265CZK". Je zde ponecháno zbytečně mnoho desetinných míst. Pomocí
funkce sprintf je lze oříznout tak, aby byly vytištěny vždy právě dvě, případně doplněné nulami.

  $retezec =~ s/((\d+(.\d+)?)(\$|USD))/sprintf("%.2f", $2*$usd2czk).CZK/ge;

Pokud si zkusíte nahradit vstupující text za nějaký složitější, kde se vyskytuje cena v dolarech vícekrát, jsou
nahrazeny všechny. Zajímavé by také mohlo být získávání aktuálního kurzu za běhu.

Přepínač e má ještě jednu zajímavou vlastnost. Lze ho uvést pro 1 regulární výraz vícekrát. Ilustrujme si to na
této ukázce.

  $xxx = "XXX";
  $_ = "***\$xxx***";
  s/(\$\w+)/$1/ee;
  print;

Počáteční stav s 2 přepínači e:

  s/(\$\w+)/$1/ee;

V 1. kroku je jedno e spotřebováno na nahrazení proměnné $1 svým obsahem, tedy řetězcem '$xxx'.

  s/(\$\w+)/$xxx/e;

Zbývá nám ještě poslední e, které vyhodnotí výraz $xxx - tedy nahradí proměnnou $xxx jejím obsahem.

  s/(\$\w+)/XXX/;

Příště budeme v regulárních výrazech pokračovat a podíváme se na zoubek rozšířeným vzorům.

Online verze článku: http://www.linuxsoft.cz/article.php?id_article=1053

Perl (23) - Regulární výrazy - rozšířené vzory

Rozšířené vzory jsou rozšířením tradiční syntaxe regulárních výrazů o nové konstrukce.

2.1.2006 06:00 | Jiří Václavík | přečteno 751x 

Rozšířené vzory

Jak již bylo řečeno, rozšířená syntaxe zavádí některé speciální konstrukce. Ač možná nevědomky, již jsme se s
ní setkali. Jde například o zápis závorek tak, aby jejich obsah nebyl zapamatován.

Syntaxe rozšířených vzorů vypadá obecně takto: (?znak ). Znak zde zastupuje nějaký symbol nebo
symboly, které blíže určují, o jaký rozšířený vzor se jedná.

Upozornění: Některé z rozšířených vzorů jsou zařazeny pouze experimentálně a není zaručeno, že budou

dostupné i ve vyšších verzích Perlu. Proto použití těchto konstrukcí konzultujte s manuálem (man perlre).

Komentáře

Dalším způsobem, jak dovnitř regulárního výrazu vložit komentář je použití syntaxe (?# ). Chová se stejně,
jako klasické komentáře - je ignorován. Lze ho psát na libovolné místo v regulárním výrazu. Jediným
omezením v komentáři je nemožnost použít uzavírací kulatou závorku. Většinou je přehlednější užít volnou
syntaxi a přepínač x, nicméně rozšířený vzor tu je také.

  print "MATCHED" if "12AFBB" =~ /^[a-fA-F0-9]+$(?#číslo v šestnáctkové soustavě)/;
  print "MATCHED" if "12AFBB" =~ /^[a-fA-F0-9]+(?#číslo v šestnáctkové soustavě)$/;

Lokální určení přepínače

Lze nastavit, aby byl některý z přepínačů imsx aktivní jen pro určitou část vzoru. Následkem uvedení
přepínače i se nerozlišují velká a malá písmena. To můžeme aplikovat na část vzoru:

  print "MATCHED" if "abcdef" =~ /((?i)abc)def/; #vyhovuje
  print "MATCHED" if "AbCdef" =~ /((?i)abc)def/; #vyhovuje - u podřetězce abc nezáleží na
velikosti písmen
  print "MATCHED" if "abcDef" =~ /((?i)abc)def/; #nevyhovuje - u podřetězce def záleží na
velikosti písmen

Přepínač, kterému předřadíme znak -, je výslovně zakázán. Tímto způsobem lze globální přepínač naopak
zrušit.

  print "MATCHED" if "abcdef" =~ /((?-i)abc)def/i; #vyhovuje
  print "MATCHED" if "ABCdef" =~ /((?-i)abc)def/i; #nevyhovuje - globální přepínač i, který
nerozlišuje velká a malá písmena, byl přebit lokálním, který ho ruší
  print "MATCHED" if "abcDef" =~ /((?-i)abc)def/i; #vyhovuje - u podřetězce def nezáleží na
velikosti písmen. Přepínač i je zrušen pouze pro abc

Aby nebyl obsah závorek zároveň uložen, lze použít zápis (?přepínač:řetězec), což je ekvivalentní s
(?:(?přepínač)řetězec)

Lokální přepínače umožňují ještě další věc. Co když v cyklu potřebujeme testovat výraz, u nějž předem
nevíme, jestli bude záležet na velikosti? Určitě bychom našli nějaké okliky, jak toho docílit, ale nejlepší řešení
povede právě přes lokální přepínače, které vztáhneme na celý regulární výraz.

  foreach $vzor (qw(Praha (?i)linux)) {
      print "MATCHED 1\n" if "praha" =~ /$vzor/; #nevyhovuje
      print "MATCHED 2\n" if "Praha" =~ /$vzor/; #vyhovuje
      print "MATCHED 3\n" if "Linux" =~ /$vzor/; #vyhovuje
      print "MATCHED 4\n" if "linux" =~ /$vzor/; #vyhovuje
  }

Pohled dopředu a dozadu

Pohled patří mezi další speciální vlastnosti. V určitém stadiu se testování řetězce může zastavit a lze porovnat
část řetězce za nebo před aktuální pozicí. Pozice se přitom nemění. V konečném důsledku to znamená, že ten
řetězec, který byl kontrolován z nějaké zadní nebo přední pozice, nebude součástí vyhovujícího podřetězce,
přestože byl srovnáván. Změní mimo jiné věci jako návratová hodnota, hodnoty proměnných $n nebo 
proměnná $&. Syntaxe pohledu dopředu je (?= ) a pohledu dozadu (?<= ). 

Význam těchto konstrukcí je zřejmý i z jejich přesnějších názvů - vzor (?= ) se nazývá pozitivní look-ahead
a (?<= ) je pozitivní look-behind.

  $r = "mandrivalinux linux linuxsoft";
  @presna_shoda = ($r =~ /(?<=\W)linux(?=\W)/g);#linux
  @zacinajici = ($r =~ /(?<=\W)linux\w+/g);     #linuxsoft
  @koncici = ($r =~ /\w+linux(?=\W)/g);         #mandrivalinux
  

Obzvláště z 1. případu je patrné, co pohledy dopředu a dozadu dělají. Ve výsledném seznamu je jen řetězec
"linux", přestože jsou ve skutečnosti kontrolovány i znaky před a po. Po odstranění pohledů bude tato
kontrola zrušena.

Pokud (?= ) resp. (?<= ) nahradíme za (?! ) (negativní look-ahead) resp. (?<! ) (negativní
look-behind), je význam opačný. Řetězec vyhoví pouze jestliže vzor neuvidí vepředu resp. vzadu vzor, který
jsme určili.

Nakonec ještě poznamenejme, že ze zřejmých důvodů nelze v look-behind použít kvantifikátory, jež
nevyjadřují konkrétní délku.

Provedení kódu Perlu uvnitř regulárního výrazu

Konstrukce (?{výraz}) slouží k vykonání perlového bloku kódu uvnitř vzoru, přičemž výsledek nijak
neovlivňuje vzor. To lze použít v případě, kdy chceme už v regulárním výrazu přiřadit zapamatované hodnoty
do proměnných.

  $retezec = "Cena: 120USD";
  $retezec =~ /Cena:\s(\d+)(?{$cena = $1})/;
  print $cena; #tiskne 120

Místo $1 by bylo pohodlnější ve vzoru použít speciální proměnnou $^N. Jejím obsahem je vždy posledně
zapamatovaný řetězec.

A pro úplnost zmíním ještě proměnnou $^R, která obsahuje návratovou hodnotu posledního bloku vloženého
do regulárního výrazu.

Provedení kódu Perlu uvnitř regulárního výrazu s dosazením do vzoru

Zápis (??{výraz}) má podobný význam jako ten předešlý. Liší se v tom, že výsledek po vyhodnocení
výrazu je dosazen do vzoru.

  $retezec = "Cena: 120USD";
  $kusu = 5;
  $cena_za_kus = 24;
  print "MATCHED" if $retezec =~ /(??{$cena_za_kus*$kusu})USD/;

Vypnutí backtrackingu

Backtracking znamená posouvání pozice v porovnávaném řetězci zpět. To nastává při porovnávání pomocí
hladového kvantifikátoru. Nejprve je vždy vlivem hladového kvantifikátoru spolknuta nejdelší možná část textu
a poté se pozice posunuje zpět. Právě toto je backtracking.

  print "MATCHED" if "cokoliv" =~ /.*liv/;    #vyhovuje
  print "MATCHED" if "cokoliv" =~ /(?>.*)liv/;#nevyhovuje

V prvním případě backtracking normálně funguje. Ve druhém je však vlivem hladového kvantifikátoru
spolknut celý text až do konce, backtracking neprobíhá a vzor tedy nemůže uspět.

Podmínky

A na závěr jsem si nechal řídící konstrukci. Je možné se až uvnitř regulárního výrazu rozhodnout, jak bude jeho
další část vypadat. Podmínka má dva možné zápisy. Pro podmínku typu if-else to je
(?(test)v_případě_true|v_případě_false) a pro samotné if jen
(?(test)v_případě_true). Test je výrazem, který se vyhodnocuje na true nebo false. Podle
vyhodnocení je aplikována příslušná část regulárního výrazu.

  $prospel = 0;
  $retezec = "nedostatečný";
  print "MATCHED" if $retezec =~ /
      (?(?{
                $prospel == 1;                       #pokud prospěl
          })
        (^(výborný|chvalitebný|dobrý|dostatečný)$)   #aplikuje se tento regulární výraz
        |
        ^nedostatečný$                               #jinak tento
      )
      /x;

Pokud máte zájem o podrobnější informace o rozšířených vzorech, odkazuji na manuálovou stránku perlre a
její oddíl Extended Patterns.

Další escape znaky

V předcházejících dílech jsme již pár escape znaků poznali, ale ještě mnohé další zbývají. Pojďme si představit

alespoň některé z nich.

Pomocí escape znaků lze tisknout všechny znaky ASCII tabulky. Libovolný znak se zapisuje buď osmičkově
nebo šestnáckově. V 1. případě se píše příslušné trojmístné číslo za zpětné lomítko, u šestnáctkového zápisu je
to dvojmístné číslo za \x.

  print "Perl" =~ /\x50\x65\x72\x6C/; #šestnáctkový zápis
  print "Perl" =~ /\120\145\162\154/; #osmičkový zápis

Escape znaky lze použít ke změně velikosti písmen. Převedeme řetězec s náhodnou velikostí znaků na řetězec,
který velkým písmenem pouze začíná (ostatní budou malá).

  $retezec = "náhODná VElIkOSt";
  $retezec =~ s/(.?)(.*)/\u\1\L\2\E/;
  print $retezec; #tiskne "Náhodná velikost"

Znak \u Převádí znak, který následuje, na velké písmeno. Podobně znak \l převede následující písmeno na
malé. \L změmí skupinu znaků na malá písmena. Skupina začíná ihned za \L a končí uvedením znaku \E
nebo koncem řetězce (v ukázce tedy není nezbytně nutný). Opět existuje alternativa v podobě \U pro převedení
na velká písmena, která také končí znakem \E.

Další možností escape znaků je potlačení metavýznamu části řetězce. Stačí ji jen ohraničit znaky \Q (začátek) a
\E (konec). 1. případ nevyhoví, protože otazník má význam kvantifikátoru. \Q tento význam ruší.

  print "MATCHED" if " ? " =~ /^ ? $/;
  print "MATCHED" if " ? " =~ /^\Q ? \E$/;

Další věc nesouvisí ani tak s escape znaky, jako spíše s významem speciálních znaků. Vytvoříme vzor, kterému
vyhoví jeden z těchto řetězců: "/root", "/home/xdf" nebo "/usr/local/apache2.2/xdf". V nich
se vyskytuje řada lomítek, které ale mají v regulárních výrazech speciální význam. Proto jim všem musíme
předřazovat zpětné lomítko.

  /^((\/root)|(\/home\/xdf)|(\/usr\/local\/apache2\.2\/xdf))$/

Je to správné řešení, ale právě v těchto případech (typické právě pro adresářové cesty) je užitečné použít jiný
uvozovací znak, který není ve výrazu použit. Připomínám, že je potom povinné i uvedení m před uvozujícím
znakem.

  m#^((/root)|(/home/xdf)|(/usr/local/apache2\.2/xdf))$#

Příště se už konečně podíváme na nějaké praktické užití regulárních výrazů.

Online verze článku: http://www.linuxsoft.cz/article.php?id_article=1054

Perl (24) - Regulární výrazy - příklady

Náš miniseriál se pomalu chýlí ke konci. Předposlední díl bude ryze praktický.

12.1.2006 06:00 | Jiří Václavík | přečteno 886x 

Již znáte řadu konstrukcí, které se v regulárních výrazech používají, ale jedna důležitá věc stále chybí. Dosud
zde byly regulární výrazy podány výhradně teoreticky. Dnes se to změní, protože celý díl je věnován čistě
příkladům.

Testování vstupu

To, že je třeba každý vstup z neověřeného zdroje testovat, je jasné. A právě regulární výrazy nabízejí spolehlivé
a pohodlné řešení.

Mechanizmus takového testování si ukážeme. Budeme chtít po uživateli zadat trojciferné číslo a vzápětí

zkontrolujeme, zda ho opravdu zadal. Na takové drobnosti se regulární výrazy využívají velmi často.

  print "Zadej trojciferné číslo: ";
  $cislo = <STDIN>;
  if ($cislo =~ /^[1-9]\d{2}$/){
      print "OK\n";
  }else{
      die "Chyba! Toto není trojciferné číslo!\n";
  }

Validace emailové adresy

Vyjádření emailové adresa je další regulární výraz, který patří k těm nejčastějším. Dokonalý zápis je velice
složitý. Jeho vytvoření je popsáno v knize Mastering Regular Expressions. My se spokojíme s poměrně
jednoduchým zápisem. Před zavináčem povolíme alfanumerické znaky a dále tečku a pomlčku. Část za
zavináčem se skládá ze 2 částí. Doménu lze vyjádřit stejně jako jméno před zavináčem a přípona se skládá ze
2-4 písmen. Uvádím pouze samotný regulární výraz:

  /^[\w\.-]+@[\w\.-]+\.[a-z]{2,4}$/i

Hledání odkazů v HTML souboru

Máme HTML soubor. Z něj chceme získat vše, co je mezi <a href=" a "> a mezi <a...> a </a>. Tedy 
odkaz a jeho popis. Získaná data vytiskneme.

Budeme postupně načítat řádky souboru (nebudeme brát ohled na odkazy obsahující znak nového řádku). V
každém řádku se pokusíme najít HTML odkaz, z něj zapamatujeme adresu a popis a vytiskneme. Protože
mohou existovat řádky bez odkazu, je nutné tisknout pouze v případě, že porovnání vzoru bylo úspěšné. Toho
dosáhneme podmíněnním příkazu print. Dále víme, že HTML není case sensitive. Proto přidáme přepínač i.

  open SOUBOR, "index.html" or die "Nelze otevřít soubor. $!";
  while (<SOUBOR>){
      print "$1 => $2\n" if $_ =~ /
          <A\sHREF=
          ["']       #uvozovky nebo apostrofy
          ([^"']*)   #vše mimo uvozovek nebo apostrofů
          ["']       #koncová uvozovka nebo apostrof
          >
          ([^<]*)    #vše mezi <a href...> a <\/a>
          <\/A>
      /ixg;
  }

Vyvstává nám tu několik problémů. Co když je na 1 řádku více odkazů? Na každý řádek se totiž aplikuje
regulární výraz pouze jednou, není proto šance najít více než 1 odkaz. Problém by vyřešil další cyklus. Není ale
potřeba nic zásadně upravovat. Pouze if zaměníme za while.

Dalším problémem jakým způsobem vyhoví řetězce jako <a href="xxx'...> Jako odkaz se vyseparuje 
pouze xxx. Jinými slovy musíme zajistit, aby byly oba uvozující znaky stejné. To je úloha jako šitá pro
pamatování. Uzavřeme tedy úvodní uvozovací znak do závorek (['"]) a místo dalšího ['"] pro uzavření
jen \1. Podobně upravíme část ([^"']*), kde nahradíme znaky "' za \1.

Po provedení úprav získáváme již funkční program:

  open SOUBOR, "index.html" or die "Nelze otevřít soubor. $!";
  while (<SOUBOR>){
      print "$2 => $3\n" while $_ =~ /
          <A\sHREF=
          (["'])     #uvozovky nebo apostrofy
          ([^\1]*?)  #vše mimo uvozujícího znaku
          \1         #koncová uvozovka nebo apostrof, podle toho, který znak uvozuje
          >
          ([^<]*)    #vše mezi <a href...> a <\/a>
          <\/A>
      /ixg;

  }

Zvýraznění skalárních proměnných

Vytvoříme nástroj, který přijímá jako parametr soubor, ve kterém zvýrazní všechny skalární proměnné. Bude-li
se například vyskytovat v textu řetězec '$promenna', bude nahrazen za '<font 
color="#800000">$promenna</font>'.

Každou iterací cyklu načteme řádek zdrojového kódu Perlu, v něm zvýrazníme výskyty proměnných a
vytiskneme výsledek. Jediný problém tak spočívá ve vytvoření vzoru, kterému vyhoví identifikátor skalární
proměnné. Vzor bude začínat dolarem, následuje libovolné písmeno nebo podtržítko (nebereme ohled na
speciální proměnné) a nakonec libovolný počet znaků slova. Celý vzor uzavřeme do kulatých závorek, získáme
tak proměnnou $1, kterou použijeme jako část náhrady.

  while (<>){
      s/(\$[_a-zA-Z]{1}[\w_]*)/<font color="800000">$1<\/font>/g;
      print;
  }

Získání adresáře a jména souboru z umístění

Řetězec, který je cestou k souboru, rozdělíme na 2 části. Na adresář, ve kterém je daný soubor a jeho relativní
jméno. Víme, že oddělovacím znakem je poslední lomítko. Využijeme tak hladovosti kvantifikátorů.

  $umisteni = "/boot/grub/menu.lst";
  ($adresar, $soubor) = $umisteni =~ /^(.*\/)(.*)$/;
  print "Adresář: $adresar\nSoubor: $soubor\n";

Výpočet výrazů v textu

Na vstupu přijme program textový řetězec, ve kterém se občas může vyskytnout podřetězec <<výraz>>.
Výraz vyhodnotíme a získanou hodnotu za něj nahradíme.

Budeme tedy postupně načítat řádky textu a v něm takové výrazy hledat. Na 1 řádku se může vyskytnout více
výrazů, proto použijeme přepínač g. Další přepínač, který aplikujeme, je e. Umožňuje nám přímo v regulárním
výrazu nahrazovat za výsledek nějakého výrazu. A použijeme ho hned 2krát, protože ještě potřebujeme
vyhodnotit řetězec, jako by byl částí zdrojového kódu.

  while (<>){
      s/<<([^>]+)>>/$1/gee;
      print;
  }

Spíše jen pro ukázku zde uvádím zápis stejného programu, který používá pouze jeden přepínač e. Funkce 
eval může ten druhý zastoupit.

  while (<>){
      s/<<([^>]+)>>/eval $1/ge;
      print;
  }

Pošleme programu na vstup řetězec

  Týden má <<7>> dní, <<7*24>> hodin, <<7*24*60>> minut nebo <<7*24*60*60>> sekund.

a získáme

  Týden má 7 dní, 168 hodin, 10080 minut nebo 604800 sekund.

Upozorňuji, že toto je jen ilustrační příklad na regulární výrazy. Mimo jiné je totiž také velkou bezpečnostní
dírou. Umožňuje spuštění příkazů. Například po zadání řetězce "...<<system("rm soubor")>>..."
bude proveden systémový příkaz rm soubor. Speciální techniky, kterými se lze chránit, probereme někdy v

budoucnu.

Zdvojení všech znaků slova v řetězci

Problém vyřešíme tak, že vyhledáme všechny (přepínač g) výskyty znaků slova a ty jednoduše zdvojíme.
Nejlepší řešení vede přes pamatování, ale abychom vyzkoušeli také něco dalšího, použijeme proměnnou $&.

  $_ = "LINUX & PERL";
  s/\w/$& x 2/eg;
  print;

Validace HTML tagu

HTML tagem může být <html>, <a href="http://www.linuxsoft.cz">, </b>, <img 
src="logo.png"/> nebo i <center >. Možností je celá řada. Pokusíme se je v co největší míře
obsáhnout v našem řešení.

Tag bude vždy začínat znakem < a končit znakem >. Za úvodním < následuje případné lomítko a dále samotné
jméno tagu. To je složeno z nenulového počtu znaků slova. Potom je možných ještě několik bílých znaků, dále
opět nepovinné lomítko a nakonec znak >. To je struktura tagu, který nemá žádné parametry.

Argumenty se píší za jméno tagu. Může jich být libovolné množství. Každý z parametrů začíná bílým znakem.
Tu následuje nenulový počet znaků slova. Nyní jsou v našem výrazu implementovány i přepínače. Ale stále
chybí pravé parametry. Za přepínačem může nepovinně být rovnítko (případně obalené bílými znaky) a nějaká
hodnota - v uvozovkách, apostrofech, nebo jako holé slovo.

U HTML tagů nezáleží na velikosti písmen. O to se ale starat nemusíme, protože jsme nikde konkrétní velikost
neuváděli.

Když všechny tyto poznatky aplikujeme na regulární výraz, vznikne nám toto:

/
    <                       #začátek tagu
    \/?                     #případné lomítko
    \w+                     #jméno tagu
    (                       #parametry
        \s+                 #mezera
        \w+                 #parametr
        (                   #případná hodnota parametru
            \s*
            =               #rovnítko
            \s*
            ("[^"]*")       #v uvozovkách
            |               #nebo
            ('[^\']*')      #v apostrofech
            |               #nebo
            ([^\W]*)        #holé slovo
        )?
    )*
    \s*
    \/?                     #případné lomítko
    >                       #konec tagu
/x

Ale nemůže to stále být konečné řešení, protože vyhoví i řetězec </x/>, který tagem rozhodně není. Lomítko
tedy může být maximalně jedno - buď na začátku nebo na konci. Toho docílíme rozvětvením.

/
    <                                                             #začátek tagu
    ((\/\w+(\s+\w+(\s*=\s*("[^"]*")|('[^\']*')|([^\W]*))?)*\s*)   #lomítko na začátku
    |                                                             #nebo
    (\w+(\s+\w+(\s*=\s*("[^"]*")|('[^\']*')|([^\W]*))?)*\s*\/?))  #případné lomítko na konci
    >                                                             #konec tagu
/x

Generování příkazů INSERT z dat textového souboru

Nakonec si uvedeme (po testování vstupu) asi nejpraktičtější příklad. Řekněme si, že máme data v nějakém
textovém souboru a chceme je dostat do databáze. Ve zdrojovém souboru je máme k dispozici ve formátu
sloupec\tsloupec\tsloupec... (sloupce oddělené tabulátorem), tedy například:

20050101        50000
20050102        84000
20050103        0
20050102        -20000
20050104        47000

Právě pro případ dvou sloupců si napíšeme skript, který data převede na INSERT příkazy:

INSERT INTO finance (datum, zisk) VALUES ("20050101", "50000");
INSERT INTO finance (datum, zisk) VALUES ("20050102", "84000");
INSERT INTO finance (datum, zisk) VALUES ("20050103", "0");
INSERT INTO finance (datum, zisk) VALUES ("20050104", "-20000");
INSERT INTO finance (datum, zisk) VALUES ("20050105", "47000");

Náš příklad je velmi konkrétní. Mělo by to usnadnit pochopení.

Postupně načteme každý řádek, rozdělíme na jednotlivé sloupce a vygenerujeme INSERT příkaz.

  open DATA, "data" or die "Nelze cist zdroj dat\n";
  open W, ">insert.sql" or die "Nelze zapisovat\n";
  
  while (<DATA>){
      ($datum, $zisk) = $_ =~ /(.+)\t(.+)/;
      chomp $zisk;
      print W "INSERT INTO finance (datum, zisk) VALUES (\"$datum\", \"$zisk\");\n";
  }
  
  close DATA;
  close W;

Poznámka - samozřejmě by šlo rozdělení řešit jednodušeji přes split:

  ($datum, $zisk) = split "\t", $_;

Zkusíme si ještě podobnou věc a to generování INSERT příkazů z dokumentů tabulkových editorů. Přesně toto
jsem kdysi potřeboval a někde jsem našel velice hezké řešení. Prvním krokem je uložit dokument v nějakém
textovém formátu - například CSV. CSV soubor může vypadat třeba takto:

0;0;0;0;0;0;0;0;0;0;0;0;0;0;0;0;0;0;0;0;0;0;0;0;0;0;0;0;0;0;0
0;0;0;0;0;0;0;0;0;0;0;0;0;0;1;1;1;1;1;1;1;0;0;0;0;0;0;0;0;0;0
0;0;0;0;0;0;0;0;0;1;1;1;1;1;1;1;1;1;1;1;1;0;0;0;0;0;0;0;0;0;0
0;0;0;0;0;1;1;1;1;1;1;1;1;1;1;1;1;1;1;1;1;1;0;0;0;0;0;0;0;0;0
0;0;0;0;0;1;1;1;1;1;1;1;1;1;1;1;1;1;1;1;1;1;0;0;0;0;0;0;0;0;0
0;1;1;1;1;1;1;1;1;1;1;1;1;1;1;1;1;1;1;1;1;1;0;0;0;0;0;0;0;0;0

Středníky jsou oddělovače sloupců a konce řádků oddělovače řádků. V každé buňce máme hodnotu 0 nebo 1.
Tuto dvojrozměrnou strukturu budeme ukládat do jednoho rozměru. Pro každou buňku budeme tedy chtít v
databázi uložit její souřadnice.

  my $radek;
  my @radek;
  my $cislo_radku = 0;
  my $jmeno_tabulky = "table";
  
  open DATA, "data.csv" or die "Nelze číst zdroj dat\n";
  open(W, ">insert.sql") or die "Nelze zapisovat\n";
  
  postupně načítáme jednotlivé řádky
  while ($radek = <DATA>){
      $cislo_radku++;
      @radek = $radek =~ /(?:([^;]*);)/g;#řádek (řetězec) je rozdělen na jednotlivé buňky

      #pro každou buňku aktuálního řádku vytvoříme INSERT příkaz

      for (my $cislo_sloupce=0; $cislo_sloupce<@radek; $cislo_sloupce++){
          chomp $radek[$cislo_sloupce];#je-li hodnota poslední v řádku, obsahuje i znak
konce řádku a ten je třeba odstranit
          print W "INSERT INTO $jmeno_tabulky (x, y, hodnota) VALUES ($cislo_radku,
$cislo_sloupce, $radek[$cislo_sloupce]);\n";
      }
  }
  close DATA;
  close W;

Poznámka - rozdělení by opět šlo řešit jednodušeji použitím funkce split:

      @radek = split(";", $radek);

Dnes jsem se pokusil ukázat široké využití regulárních výrazů. Doufám, že jsem vám dodal alespoň trochu
inspirace. Příští díl bude ze série o regulárních výrazech poslední. Podíváme se na měření rychlosti a
debugging.

Online verze článku: http://www.linuxsoft.cz/article.php?id_article=1061

Perl (25) - Regulární výrazy - závěr

V posledním díle věnovaném regulárním výrazům se podíváme na měření rychlosti a debugging.

25.1.2006 06:00 | Jiří Václavík | přečteno 799x 

Porovnání rychlosti několika regulárních výrazů

Pro měření času potřebného pro vykonání určitého úseku kódu se v Perlu obvykle používá modul
Benchmark. Nyní si ukážeme pouze to, jak pomocí něj porovnávat rychlosti regulárních výrazů. To ovšem
není zdaleka jediné užití. Benchmark se používá poměrně často a někdy později v seriálu si podrobněji
rozebereme i jeho jiné funkce. Modul Benchmark načteme takto:

use Benchmark qw(:all);

Nyní využijeme jen jeho jedinou funkci a to cmpthese. Ta se používá právě pro porovnávání rychlostí
jednotlivých úseků kódu. Jako parametr jí jsou předávany 2 hodnoty:

číslo - je-li záporné, znamená minimální počet sekund, po které testování poběží1.
odkaz na hash, jehož každý prvek obsahuje nějaký úsek kódu - právě rychlosti těchto úseků jsou pak
porovnávány

2.

My si zkusíme dokázat, že použití závorek s pamatováním hodnoty uvnitř nich je časově náročnější než
speciální syntaxe závorek, která si svůj obsah nepamatuje.

Nejprve se musíme konkrétně rozhodnout, které regulární výrazy budeme porovnávat. Jako ukázku si zvolíme
například tyto:

my $re1 = qr/X(\w)/o;
my $re2 = qr/X(?:\w)/o;

Dále je třeba určit s čím se budou tyto regulární výrazy porovnávat. Protože testujeme pouze rychlost závorek,
musí být jediným odlišným faktorem v regulárním výrazu právě závorky a vše ostatní musí být stejné. To platí i
o řetězcích, které budeme se vzory porovnávat. V opačném případě by to mohlo ovlivňovat rychlost.

Nastavíme tedy parametry funkce cmpthese. Předáme jí hash, jehož klíči budou nějaké unikátní řetězce
(jména porovnávaných úseků) a hodnotami anonymní podprogramy, v nichž probíhá test regulárního výrazu.

Testovaný úsek bude obsahovat určitý počet porovnání nějakého řetězce s regulárním výrazem. Proto musíme
definovat řetězec a počet porovnání (obě tyto hodnoty budou pro testování obou regulárních výrazů stejné).

Jako řetězec, který bude srovnáván, definujeme podřetězec "X5 " uvnitř dvou stejných podřetězců.

my $okoli = "zxcvasdf" x 500;
my $text = $okoli . "X5 " . $okoli;

Nastavíme počet opakování:

my $pocet_opakovani = 1000;

Ještě nám zbývá zadat 1. argument - minimální počet sekund, po které porovnávání poběží:

my $delka_testu = -3;

A můžeme volat cmpthese:

cmpthese($delka_testu, {
    "s pamatovanim"  => sub {$text =~ m/$re1/ for 1 .. $pocet_opakovani;},
    "bez pamatovani" => sub {$text =~ m/$re2/ for 1 .. $pocet_opakovani;},
});

Po chvilce čekání program vytiskne výsledky ve formě tabulky, seřazené podle rychlostí vzestupně. Sloupec
Rate udává hodnotu, kolikrát byl proveden daný úsek kódu (tj. 1000 porovnávání) za sekundu. Další sloupce
ukazují relativní rozdíl rychlostí každé dvojice úseků v procentech. V našem případě je tato dvojice jediná
(nebereme v potaz pořadí), protože jsme k porovnávání zadali pouze 2 úseky.

$ perl cmpthese
                Rate  s pamatovanim bez pamatovani
s pamatovanim  115/s             --           -31%
bez pamatovani 168/s            46%             --
$

Pokud si zkusíte pohrát se vstupním řetězcem (změna délky, pozicí apod.), zjistíte, že poměr rychlostí záleží i
na dalších okolnostech a to dost výrazně. Vždy by ale mělo být rychlejší porovnávání bez pamatování.

Zpracování regulárních výrazů

Způsob, kterým se zpracovávají regulární výrazy můžeme kontrolovat. Ne že by se debugging používal
masově, ale v případech, kdy potřebujeme opravit nějakou chybu v rozsáhlém regulárním výrazu a nemáme
tušení, kde by mohla být, může pomoci. V Perlu máme možnost zapnout sledování překladu a následného
vyhodnocování pomocí jedné z následujících direktiv. Direktiva

use re "debug";

zapíná debugging. Budou tak nalezeny a podrobně rozepsány všechny regulární výrazy. Lze užít také
přehlednější variantu se zvýrazněním:

use re "debugcolor";

Zkusíme tímto způsobem otestovat nějaký jednoduchý úsek kódu.

use re "debugcolor";
"xy123456zzz" =~ /^xy\d{6}z*(1)$/;

Program jako obyčejně spustíme:

$ perl debug.pl

Objeví se výpis. My si vysvětlíme pouze to nejdůležitější z něj. Začneme tímto úsekem:

   1: BOL(2)
   2: EXACT <xy>(4)
   4: CURLY {6,6}(7)
   6:   DIGIT(0)
   7: STAR(10)
   8:   EXACT <z>(0)

  10: OPEN1(12)
  12:   EXACT <1>(14)
  14: CLOSE1(16)
  16: EOL(17)
  17: END(0)

Každá položka nebo skupina položek (můžeme jim říkat uzly) vyjadřuje nějakou část regulárního výrazu. Číslo
na začátku každého řádku je id uzlu. Za každým uzlem je v závorce id následujícího uzlu.

BOL znamená prázdný řetězec na začátku řádku - tedy ^. EXACT je přesná shoda řetězce mezi < a > - v našem
případě xy. Kvantifikátor {6} je převeden na obecnější zápis {6,6}. 6krát se opakuje vše odsazené - v našem
případě jen uzel DIGIT, který označuje numerický znak. Pokračujeme přesným výskytem znaku s libovolným
počtem opakování. Dále máme 1. otevírací závorku. Vše mezi OPEN1 a CLOSE1 je uloženo v $1, vše mezi
OPEN2 a CLOSE2 v $2 atd. EOL označuje prázdný řetězec na konci řádku - znak $ - a konečně END je vždy
na konci.

Toto zatím nemá se srovnávaným řetězcem nic společného. Probíhá pouze překlad regulárního výrazu.
Porovnávat se bude až v další fázi.

Uvádím tabulku několika častých uzlů. Kompletní je k vidění v manuálové stránce perldebguts.

Uzel Význam

Kotvy

BOL začátek řádku

END konec regulárního výrazu

EOL konec řádku

BOUND na hranici slova

NBOUND mimo hranici slova

Znak z množiny znaků

ALNUM alfanumerický znak

NALNUM nealfanumerický znak

DIGIT číslice

DIGIT nečíslice

SPACE bílý znak

NSPACE nebílý znak

ANYOF množina definovaná hranatými závorkami

ANY libovolný znak

Kvantifikátory

STAR libovolný počet opakování

CURLY opakování definované složenými závorkami

PLUS minimálně 1 opakování

Ostatní

EXACT přesná shoda

NOTHING prázdný řetězec

OPENn otevírací kulatá závorka

CLOSEn uzavírací kulatá závorka

V tuto chvíli máme regulární výraz zkompilován a můžeme se pustit do porovnávání.

Matching REx `^xy\d{6}z*(1)$' against `xy123456zzz1'

Následují řádky, které postupně srovnávají zkompilovaný regulární výraz se vzorem. Syntaxe se liší podle toho,
zda používáte direktivu use re "debug"; nebo use re "debugcolor";. V prvním případě je
následovná:

pozice_ve_srovnávaném_řetězci <vyhovující_podřetězec> <zbývající_podřetězec> | id uzel

Konkrétně může vypadat třeba takto:

  11 <xy123456zzz> <1> | 10: OPEN1

debugcolor má tu výhodu, že spojuje <vyhovující_podřetězec> a <zbývající_podřetězec>.
Vyhovující podřetězec je bíle podbarven (v článku tučně zeleně). Varianta s debugcolor je hlavně díky této
vlastnosti o mnoho přehlednější.

Vezměme teď výstup řádek po řádku.

   0 <xy123456zzz1>    |  1:  BOL
   0 <xy123456zzz1>    |  2:  EXACT <xy>
   2 <xy123456zzz1>    |  4:  CURLY {6,6}
                       DIGIT can match 6 times out of 6...

1. řádek ukazuje vždy výchozí stav. Na dalším je již úspěšně nalezen znak ^ (ten je samozřejmě v každém
řetězci), který ale není viditelný - proto jsme stále na pozici 0. Zajímavější je to na 3. řádku. Podbarven máme
podřetězec xy. Vyhovuje totiž uzlu EXACT <xy>.

   8 <xy123456zzz1>    |  7:  STAR
                       EXACT can match 3 times out of 2147483647...

Právě jsme úspěšně nalezli 6 po sobě jdoucích číslic. Podbarveno již máme 8 znaků.

  11 <xy123456zzz1>    |  10:  OPEN1
  11 <xy123456zzz1>    |  12:  EXACT <1>
  12 <xy123456zzz1>    |  14:  CLOSE1
  12 <xy123456zzz1>    |  16:  EOL
  12 <xy123456zzz1>    |  17:  END

Dále byly nalezeny 3 znaky z a nakonec ještě znak 1. V okamžiku, kdy se regulární výraz dostane k uzlu END,
skončí porovnání úspěchem.

Match successful!

Závěr

Tímto jsme definitivně skončili rozsáhlou kapitolu o regulárních výrazech. Pokud máte zájem o další
informační zdroje, zde některé uvádím:

perlre(1) - popis syntaxe regulárních výrazů v Perlu
perlrequick(1) - vysvětluje základy regulárních výrazů
perlretut(1) - tutoriál na téma Perl a regulární výrazy
perlop(1), oddíly Regexp Quote-Like Operators a Gory details of parsing 
quoted constructs - operátory v Perlu pro práci s regulárními výrazy
perlfaq6(1) - časté otázky
perldebguts(1), oddíl Debugging regular expressions - sledování průběhu
vyhodnocování regulárních výrazů
kniha Mastering Regular Expressions, Jeffrey Friedl, O'Reilly and Associates. Česká recenze vyšla na
serveru linuxzone.cz
existují weby zaměřené přímo na regulární výrazy - namátkou www.regular-expressions.info nebo 
www.regularnivyrazy.info.
řada článků a tutoriálů na Internetu, z nichž nemohu opomenout vynikající dílo Pavla Satrapy na
www.kit.vslib.cz/~satrapa/docs/regvyr/all.html nebo práci Pavla Dařeny.
www.regexlib.com - zde je archiv regulárních výrazů. Dříve, než začnete psát složitější regulární výraz,

se podívejte sem. Možná už někdo měl stejný problém.

Online verze článku: http://www.linuxsoft.cz/article.php?id_article=1082

Perl (26) - Podprogramy

Uživatelsky definované funkce jsou základem programování složitějších aplikací. Pojďme si je přiblížit.

9.2.2006 06:00 | Jiří Václavík | přečteno 713x 

Pomocí podprogramů lze program rozložit na kratší úseky, což pomáhá hned v několika aspektech. Některé
části programu se až na pár detailů, jimiž jsou vstupní hodnoty, častěji opakují. Když je nutné tyto části nějak
změnit (například kvůli vylepšení algoritmu nebo přidání dalších vlastností), musíme jich místo jediného řádku
přepisovat třeba desítky. Toto tvrzení však neplatí, pokud použijeme podprogram.

Podprogram je uživatelsky definovanou funkcí. Vykonává nějaký úsek kódu na základě algoritmu, který je pro
jeden podprogram vždy stejný, a vstupních hodnot, jenž tento algoritmus používá. Podprogram lze zapsat ve
kterémkoliv místě programu, ale pro přehlednost bychom vždy měli zvolit nějaký systém. Vhodné je definovat
podprogramy na konci nebo začátku programu, případně v modulu. Z hlavního programu (a samozřejmě i
podprogramů) pak můžeme podprogramy volat. Jakmile je podprogram definován, práce s ním je stejná jako
práce s předdefinovanými funkcemi, což jsou konec konců také podprogramy.

Neméně významnou výhodou podprogramů je zvýšení srozumitelnosti. Pokud zvolíme vhodné členění
programu, lze poměrně rychle poznat, jak program funguje, aniž bychom museli číst kilobajty komentářů.
Obzvláště u delších programů je tak mnohem snazší hledat chyby nebo je upravovat. Každý problém je
rozložen na menší problémy a ty se řeší jednotlivě.

Další výhoda souvisí s tou prvně jmenovanou a spočívá v tom, že úseky programu, které mají být stejné,
opravdu stejné jsou, neboť je píšeme pouze jednou.

Definice podprogramu

Definice pojmenovaného podprogramu vypadá následovně:

sub jméno {
    příkazy
}

Pomocí jména pak lze podprogram volat. Jak bude podrobněji rozebráno v příštím díle, lze do definice
zahrnout i takzvaný prototyp. Pomocí něj si program může vyžádat počet, pořadí a typy argumentů.

sub jméno(prototyp) {
    příkazy
}

Volání podprogramu

Jménu podprogramu se předřazuje znak &. Seznam argumentů podprogramu se uvádí za jeho jménem v
závorkách. Použijete-li závorky, uvedení & ve většině případů nebude nutné. Pokud není podprogram použit v
souvislosti s odkazy nebo jako parametr některých funkcí, Perl sám rozezná, že jde o podprogram.

&prumer(2, 7);
prumer(2, 7);

Použití

Velice jednoduchý podprogram může vypadat takto:

$jmeno = "uzivatel";
pozdrav();

$jmeno = "root";
pozdrav();

sub pozdrav {
    print "Podprogram: Ahoj $jmeno, já jsem podprogram\n";
}

Nutno dodat, že tento podprogram nemá vůbec dobré vychování, neboť používá globální proměnné.
Přijatelným řešením tohoto nedostatku bude předání argumentů podprogramu, jakmile se to naučíme.

Příklad by fungoval stejně i kdybychom nahradili řádek

pozdrav();

za

&pozdrav;

Argumenty a lokální proměnné

Jak již bylo napsáno pod posledním příkladem, použití globálních proměnných podprogramem je více než
nešikovné. Kdyby všechny proměnné ve všech podprogramech sdílely tentýž prostor jmen, brzy bychom se
zamotali. Proto je mnohem lepší předat podprogramu hodnotu jako argument. Ten v podprogramu uložíme do
proměnné, která bude platná jen v rámci tohoto podprogramu (tedy bloku). Potom se uvnitř podprogramu
vůbec nemusíme starat o jména globálních proměnných. Obecně platí, že čím víc je proměnná lokální, tím lépe.

Jak již víme, lokální proměnná se definuje pomocí funkce my. Funkcí podobných s my je více, ale zatím je pro
jednoduchost zamlčím a někdy v blízké době si rozsahy platnosti rozebereme ve zvláštním díle.

Parametrem funkce je buď proměnná nebo seznam proměnných v kulatých závorkách. Než se pustíme do
argumentů, uveďme si ještě příklad pro lepší pochopení funkce my.

$x = 7;
a();
b();
print "g: $x\n"; #tiskne 7 - lokální proměnná v &b byla platná jen pro
                 #daný blok. Po jeho ukončení má x opět globální hodnotu

sub a {
    print "b: $x\n"; #tiskne hodnotu globální proměnné x - tedy 7
}

sub b {
    my $x = 15;
    print "a: $x\n"; #vytiskne 15 - lokální proměnná přebíjí globální
}

Nyní už k věci. V podprogramu je seznam předaných argumentů uložen ve speciálním poli @_. Na začátku
podprogramu se obvykle vytváří lokální proměnné, které z tohoto pole hodnoty přebírají. Pokud nejde o
vyloženě triviální podprogram, není zrovna přehledné uvnitř něj používat proměnné $_[0], $_[1] a jim
podobné.

Nyní, když známe teorii, pojďme si ukázat praxi. Definujeme podprogram tiskni_prumer, který přebírá
jako argumenty 3 čísla a vypisuje jejich aritmetický průměr.

tiskni_prumer(6, 9, 33); #voláme funkci tiskni_prumer
                         #se třemi argumenty 6, 9, 33

sub tiskni_prumer {
    my($a, $b, $c) = @_;
    print (($a + $b + $c) / 3);
}

Do pole @_ jsou v našem případě uloženy hodnoty 6, 9 a 33. Ty jsme na 1. řádku podprogramu přiřadili do
lokálních proměnných $a, $b a $c.

Další často užívaná možnost, jak získat hodnoty z @_ vede přes funkci shift. shift maže první prvek pole
a jeho hodnota je výsledkem vyhodnocování.

print tiskni_prumer(6, 9, 33);

sub tiskni_prumer {
    my $a = shift @_;
    my $b = shift @_;
    my $c = shift @_;
    print (($a + $b + $c) / 3);
}

Pokud podprogram voláte bez uvedení seznamu (ať už třeba prázdného), je funkci automaticky předáno pole
@_. To demonstruje následující příklad.

@_ = (1, 2, 3);
&tisk;

sub tisk {
    print @_;
}

Též můžete zkusit místo &tisk volat &tisk(). V takovém případě se @_ nepředává.

Předáváte-li podprogramu 2 pole, slijí se do jednoho. Toto lze řešit pomocí odkazů, které prozatím necháme
stranou. Je to též možné pomocí globálních názvů, ale jak již bylo několikrát řečeno, není vhodné, aby funkce
jakkoliv zasahovala do svého okolí.

Návratová hodnota podprogramu

Co když nechceme, aby podprogram výsledky vypisoval, ale chceme je přiřadit do nějaké proměnné a následně
je použít? Od toho je zde návratová hodnota - tedy hodnota, kterou bude vracet volání podprogramu při
vyhodnocování. Právě díky návratové hodnotě tak můžeme podprogram volat na pravé straně od operátoru
přiřazení.

$prumer = prumer(6, 9, 33);

sub prumer {
    my($a, $b, $c) = @_;
    my $prumer = ($a + $b + $c) / 3;
}

Jako návratová hodnota je automaticky považována poslední hodnota v podprogramu - v našem případě jí byla
proměnná $prumer.

Chcete-li návratovou hodnotu výslovně uvést, což doporučuji, použijte klíčové slovo return. Jeho 
parametrem je návratová hodnota.

$prumer = prumer(6, 9, 33);

sub prumer {
    my($a, $b, $c) = @_;
    return ($a + $b + $c) / 3;
}

Program, který bude vracet průměr 3 hodnot asi nebude mít příliš velký význam. Posuneme se proto dále.
Daleko větší uplatnění jistě má funkce, která počítá aritmetický průměr nezávisle na počtu argumentů.

sub spocitej_prumer {
    my $soucet = 0;
    $soucet += $_ for (@_);
    return ($soucet / scalar @_);
}

Deklarace

Jak už bylo řečeno, podprogram může být definován kdekoliv v programu. Pokud ale voláte podprogram dříve
než ho definujete a zároveň používáte céčkovský styl volání (bez ampérsandu), musíte argumenty uzavřít do
závorek. Spíše pro zajímavost uvádím, že pokud ještě před voláním podprogram deklarujete, lze závorky
vynechat.

sub prumer; #deklarace

prumer 1, 2, 6;

sub prumer {
    my ($a, $b, $c) = @_;
    print (($a + $b + $c) / 3);
}

Konstanty

Konstanta je speciální typ funkce, která neočekává žádný argument, ale pouze vrací nějakou hodnotu. Jelikož
má takový podprogram při všech voláních stejné podmínky, vrací konstantní hodnotu. To, že odmítáme
jakýkoliv argument, dáme na vědomí uvedením prázdného prototypu - ten zakazuje veškeré argumenty.

sub PI() { return 3.141592654 }

sub obsah_kruhu ($) {
    my($r) = @_;
    return PI*$r**2;
}

print obsah_kruhu(2);

Právě jsme poznali první prototyp. Příště si v tomto směru znalosti rozšíříme, neboť celý díl se bude věnovat
právě jim.

Online verze článku: http://www.linuxsoft.cz/article.php?id_article=1092

Perl (27) - Prototypy

Pomocí prototypů lze kontrolovat datové typy proměnných vstupujících do podprogramu.

21.2.2006 06:00 | Jiří Václavík | přečteno 549x 

Téměř všechny smysluplné podprogramy přijímají nějaké argumenty, které jsou potom v podprogramu
přístupné v podobě seznamu. Již z minulého dílu víme, jak argumenty přijímat. Nevíme však, jak si je i vynutit.
Právě pomocí prototypů lze určit kolik jakých argumentů má podprogram vyžadovat.

Prototyp obsahuje posloupnost speciálních znaků, podle které se určuje, co za parametry bude podprogram
požadovat. Je uveden vždy za názvem podprogramu v jeho deklaraci nebo definici v kulatých závorkách.

sub jméno(prototyp) {
    příkazy
}

U podprogramů s prototypem je nutné, aby byla definice nebo deklarace před samotným voláním. V opačném
případě nebude na prototyp brán zřetel.

Jednoduché prototypy

V tabulce jsou znaky, jimiž lze určit prototyp.

Znak Význam

$ skalár

@ pole

% hash

& anonymní podprogram

* typeglob

Znak $ určuje, že bude očekávána hodnota ve skalárním kontextu. Skalární kontext je vynucen - tzn. při
předání seznamu je vrácen počet jeho prvků. Pomocí * se dá předat typeglob, což se používá pro přijímání
ovladačů. Prototyp pole a hashe spolkne vše až do konce seznamu argumentů.

Pokud definujeme funkci s prototypem $$$$$$, je očekáváno 6 skalárních argumentů. Prototyp $@ zase
vyjadřuje skalár a pole v daném pořadí. Pozor na zápisy jako @@. Pole se slijí, protože 1. pole spolkne vše až do
konce seznamu a na 2. pole žádné hodnoty nezbydou.

sub f($$$) {
    print @_;
}

f(1, 2);          #chyba, funkce f požaduje 3 skalární argumenty
f(1, 2, 3);       #ok
f(1, 2, 3, 4, 5); #chyba, funkce f požaduje 3 skalární argumenty

Pokud spustíte tento úsek kódu, bude hlásit chyby. Prohoďme pořadí definice a volání funkcí tak, aby volání
bylo před definicí.

f(1, 2);          #zdánlivě ok
f(1, 2, 3);       #ok
f(1, 2, 3, 4, 5); #zdánlivě ok

sub f($$$) {
    print @_;
}

Žádné chyby nejsou hlášeny! Jakoby prototypy nebyly specifikovány. Pokud však budete spouštět program s
přepínačem -w,

$ perl -w prototyp.pl

tedy zapnutými varováními, zobrazí se pro taková volání funkcí

main::f() called too early to check prototype

Znamená to, že argumenty nebyly zkotrolovány pomocí prototypu. Zabránit se tomu dá, jak již bylo řečeno,
buď definicí podprogramů před jejich použitím, nebo dopřednou deklarací podprogramu.

Deklarace popisuje parametry podprogramu před jeho definicí.

sub podprogram;

Deklarace s prototypem pak vypadá následovně:

sub podprogram(prototyp);

Ukažme si, použití v praxi. K poslednímu úseku kódu, který hlásil varování, přidáme deklaraci.

sub f($$$);

f(1, 2);          #chyba, funkce f požaduje 3 skalární argumenty
f(1, 2, 3);       #ok
f(1, 2, 3, 4, 5); #chyba, funkce f požaduje 3 skalární argumenty

sub f($$$) {

    print @_;
}

Teď už byl prototyp normálně aplikován. Jen pozor na to, aby nedošlo ke konfliktu prototypů. V deklaraci musí
být stejný prototyp jako v definici.

Protože to, zda bude prototyp volán, záleží i na syntaxi volání podprogramu, rozlišíme si volání na 2 případy
podle toho, zda bude prototyp aplikován. Tabulka obsahuje 4 možné zápisy volání:

Zápis volání Prototyp?

&podprogram(parametry); není aplikován!

&podprogram; (jako &podprogram(@_);) není aplikován!

podprogram(parametry); ok

podprogram; (jako bez parametrů) ok

Jinak řečeno, je-li podprogram volán s ampérsandem, prototyp není nikdy aplikován.

Zde jsou další příklady využití prototypů:

sub f(@@);  #nemá smysl - pole se slijí
sub f();    #funkce nepřijímá argumenty
sub f($@);  #přijme skalár a pole
sub f(&$$$);#přijme anonymní podprogram a 3 skaláry
sub f(*);   #přijme ovladač souboru

Zajímavá situace může nastat, pokud voláme funkci příkazem podobným následujícímu:

funkce +10;

Jak je vlastně takový příkaz interpretován? Jako funkce s argumentem 10 nebo jako součet návratové hodnoty
funkce s hodnotou 10? Je-li definován prototyp, mělo by to být jasné. V případě prázdného prototypu se příkaz
bere jako součet, jinak je číslo parametrem podprogramu.

Předávání odkazů a další vlastnosti

Před symbolem, označujícím datový typ, může být znak \. To znamená, že podprogram přijímá datový typ,
který označuje symbol, ale předává odkaz na něj. V samotném volání podprogramu pak nesmí být konstantní
hodnota.

sub f(\@@);   #1. pole bude převedeno na odkaz
sub f(\%@);   #hash bude převeden na odkaz
sub f(\@\@\@);#pole budou převedena na odkazy

Na posledním řádku funkce přijímá jako parametr 3 pole, které jsou v podprogramu dostupné jako odkazy
(skaláry) a neslijí se tak. Pak je možné volat podprogram klidně takto:

sub f(\@\@\@);

@a = (100, 200);
@b = (10, 20, 30);
@c = (1, 2, 3, 4, 5);
f(@a, @b, @c);

sub f(\@\@\@){
    print "1. předané pole: ", @{$_[0]};
    print "2. předané pole: ", @{$_[1]};
    print "3. předané pole: ", @{$_[2]};
}

Nepovinné parametry se od povinných oddělují středníkem. Můžeme tak definovat podprogram, který bude
přijímat právě 1 nebo 2 skalární hodnoty.

sub f($;$){print @_;}

f(10);         # ok
f(10, 20);     # ok
f(10, 20, 30); # chyba

Další možnost, kterou prototypy nabízejí, je specifikace více možných datových typů na tutéž pozici.
Podprogram s touto deklarací přijme buď skalár nebo pole:

sub f(\[$@]);

Pojmenované argumenty

V případech, kdy funkce přijímá velké množství argumentů, z nichž jsou navíc některé povinné a některé ne, je
těžké udržet si v nich pořádek. Právě s tímto mohou pomoci pojmenované argumenty. Nejde syntakticky o nic
nového, jen použijeme novou filozofii předávání argumentů.

Předáme-li funkci hash s pevně danou strukturou, vytváříme vlastní systém správy argumentů. Je pak už na nás,
jak si je ošetříme a zpracujeme.

Argumenty se často uvozují znakem -. Uvedeme si jednoduchý příklad.

vypis_jmeno("-jmeno" => "Josef", "-prijmeni" => "Adamec");

sub vypis_jmeno(%){
    my %args = @_;
    print "Jméno: ", $args{"-jmeno"}, "\n";
    print "Příjmení: ", $args{"-prijmeni"}, "\n";
}

Příští díl se bude zabývat rozsahy platnosti proměnných.

Online verze článku: http://www.linuxsoft.cz/article.php?id_article=1102

Perl (28) - Rozsahy platnosti proměnných

Jak v Perlu vytvářet lokální a globální proměnné?

7.3.2006 06:00 | Jiří Václavík | přečteno 242x 

Definici proměnných lze provádět třemi funkcemi. V minulých dílech jsme používali pouze funkci my, aniž
bychom hlouběji pátrali po jejích účincích. Mimo my existují s podobným významem další 2 funkce - local a 
our.

S definicemi proměnných souvisí to, že Perl umožňuje užívat pro různé datové typy proměnné stejného názvu.
Podprogram &prog, skalární proměnná $prog, hash %prog, pole @prog, formát prog nebo ovladač prog.
Všechno to jsou identifikátory se stejným názvem. Existují tzv. typegloby, které sdružují identifikátory se
stejným názvem.

Lexikální platnost proměnné

Funkce my lexikálně deklaruje lokální proměnnou. Její rozsah platnosti je omezen jen na blok. To znamená, že
je přístupná pouze zde a z jiného bloku není její hodnota žádným způsobem zjistitelná. Hodnota takové
proměnné není dostupná ani z podprogramů volaných ze stejného bloku (jde však spíše o podstatu než o
praktické využití takových podprogramů).

Proměnná pozbývá platnosti s ukončovací složenou závorkou aktuálního bloku. Stejně jako pro bloky za
podmínkou nebo v podprogramu to platí i pro holé bloky. Důkazem toho je následující ukázka.

{

    my $p = 1;
}
print $p;

Vytisknuto není nic, protože platnost lokální proměnné $p je omezena jen na blok.

Plně kvalifikované jméno proměnné

Lokální proměnná pro daný blok vždy překryje globální proměnnou stejného názvu a datového typu. Přestože
je proměnná překrytá, můžeme k ní přistupovat. Doteď jsme, aniž bychom si to uvědomovali, proměnné
nepojmenovávali celými názvy, ale pouze jejich částmi. Plné jméno globální proměnné se skládá nejen z názvu,
kterým jsme dosud vždy proměnné identifikovali, ale i z názvu balíčku, ve kterém se nachází.

Pokud příkazem package balíček neměníme, je implicitním balíčkem main. Z libovolného místa pomocí něj
můžeme získat hodnotu proměnné $x, která je v tomto balíčku definována - a to zápisem $main::x.
Proměnnou balíčku nelze deklarovat pomocí my, protože my deklaruje proměnné lexikálně. To znamená mimo
jiné to, že taková proměnná pouze překrývá proměnnou globální. Pokud ve stejném bloku definujeme
proměnnou balíčku a následně lexikální proměnnou se stejným názvem, vytvoříme tím 2 různé proměnné! K té
lexikálně deklarované přistupujeme stejně jako dosud, ale chceme-li použít globální proměnnou stejného
názvu, je třeba ji plně kvalifikovat.

Dynamická platnost proměnné

Další funkcí pro deklaraci proměnných je local. Syntaxe je shodná se syntaxí funkce my. Proměnné takto
deklarované mají tzv. dynamicky vymezenou platnost. Při deklaraci funkcí local je případná hodnota
proměnné stejného názvu jen odložena a nahrazena do konce bloku jinou hodnotou. Poté se opět vrátí na
původní globální hodnotu. Takže i po deklaraci funkcí local jde o globální proměnnou! Jen je dočasně
změněna její hodnota. Z tohoto důvodu je vidět i v podprogramech, volaných z bloku s deklarací.

local se užívá jen málo, ale přesto je pro některé úkoly účinným pomocníkem. Kupříkladu při práci s
vestavěnými proměnnými. Potřebujeme na chvíli nějakou vestavěnou proměnnou změnit? To je úkol přesně na
míru šitý pro funkci local. 

{
    local $, = ", ";
    print @pole;
}

Vestavěná proměnná je do konce bloku nahrazena novou hodnotou. Dále se o nic nemusíme starat, protože za
blokem se obsah vestavěné proměnné opět nahradí původní hodnotou.

Srovnání my a local

Stručně tedy shrňme rozdíl mezi my a local. local nedeklaruje žádnou novou proměnnou, pouze dočasně
pozmění hodnotu globální proměnné stejného názvu. Oproti tomu my deklaruje skutečnou lokální proměnnou,
která v daném bloku překrývá globální proměnnou.

To byla teorie. Následující příklad se dvěma podprogramy je typickou ukázkou činnosti zmiňovaných funkcí.

$my = 1;
$local = 1;
a();

sub a {
    local $local = 2;
    my $my = 2;
    b();
}

sub b {
    print "LOCAL:", $local, "\n";#vytiskne 2

    print "MY:", $my, "\n"; #vytiskne 1
}

Voláme podprogram &a. V něm je pomocí my vytvořena lokální proměnná $my a zároveň je pozměněna
hodnota globální proměnné $local. Dále je v tomto podprogramu volán jiný podprogram &b, v němž jsou
tisknuty proměnné $my a $local. Protože žádné lokální proměnné takových jmen neexistují, tisknou se
globální proměnné. V případě my je vytisknuto 1, protože tato globální hodnota byla určena už na 1. řádku. V
případě local se tiskne hodnota 2. To je také hodnota globální proměnné, která však byla v podprogramu &a
dočasně pozměněna.

Stejně to funguje například s hashi. Nelze napsat

my $hash{"x"} = 1;

ale pouze překrýt prvek $hash{"x"} lze. Použijeme-li local, bude jen odložena globální hodnota prvku a
dočasně nahrazena novou.

local $hash{"x"} = 1;

Definice globální proměnné

Funkce our deklaruje globální proměnné. A to úplně kdekoliv. Nezáleží na hloubce zanoření. Jedinou
podmínkou je, že deklarace musí proběhnout před použitím proměnné. Proměnné deklarované pomocí our
jsou obsaženy v balíku (jsou tedy skutečnými globálními proměnnými), čehož lze využít pro vytváření
globálních proměnných v režimu strict.

{
    our $x = 1;
}
print $x;

Příklad tiskne 1, protože funkce our deklarovala globální proměnnou. Ta je globálně platná od okamžiku
deklarace. Kdybychom místo our použili my, stala by se proměnná $x naopak lokální.

Zajímavostí je, že lze proměnnou deklarovat pomocí our i local zároveň. Vytváříme tak globální
proměnnou, která však má hodnotu pouze v bloku.

{
    local our $p = 1;
    print "V BLOKU: $p\n";
}
print "MIMO BLOK: $main::p\n";

V příštím díle se zaměříme na práci se soubory.

Online verze článku: http://www.linuxsoft.cz/article.php?id_article=1118


